

34 MORE Tested, Ready-To-Run Game Programs in BASIC



by Delton T. Horn

34 More Tested, Ready-To-Run Game Programs in BASIC

Other TAB books by the author:

No. 1167 *Electronic Music Synthesizers*

No. 1338 *Basic Electronics Theory—with projects & experiments*

No. 1438 *The Beginner's Book of Electronic Music*

34 More Tested, Ready-To-Run Game Programs in BASIC

by Delton T. Horn



TAB BOOKS Inc.

BLUE RIDGE SUMMIT, PA. 17214

FIRST EDITION

FIFTH PRINTING

Printed in the United States of America

Reproduction or publication of the content in any manner, without express permission of the publisher, is prohibited. No liability is assumed with respect to the use of the information herein.

Copyright © 1981 by TAB BOOKS Inc.

Library of Congress Cataloging in Publication Data

Horn, Delton, T

**34 more tested, ready-to-run game programs in
BASIC.**

Includes index.

**1. Games—Computer programs. 2. Basic (Computer
program language) I. Title.**

GV1469.2.H67 794 80-28640


ISBN 0-8306-9654-7

ISBN 0-8306-1228-9 (pbk.)

Contents

Preface	6
1 One Player Games	7
Craps 8	
High/Low 12	
Guess The Variables 16	
Hangman 23	
Freebish! 34	
Crops 41	
There's Gold In Them There Skyscrapers 49	
Lost In The Jungle 56	
Galactic Search 72	
High Bid 90	
High Bid II 95	
Balancing The Scales 99	
Memory Tes 104	
What Comes Next 109	
Go Fish 113	
Quiz Whiz 122	
Tic Tac Toe 127	
2 Two Player Games	135
To And Fro 136	
Passing Points 141	
Money Mad 147	
Gambling Boxes 155	
Stack Cut 161	
3 Frustration	169
Main Game Program 171	
Two Man Frustration 176	
Frustration vs the Computer 181	
The Computer vs. Itself 184	
4 Non Games	188
Coin Flipper 189	
Surprise Poem 192	
Favorite Song 192	
Tunesmith 198	
5 Helpful Programs	202
Binary/Decimal 202	
Day of the Week 207	
Ohm's Law 210	
Fahrenheit/Celsius 214	
Appendix Notes On Programming In BASIC	217
Index	224

Preface



Computers have countless practical applications in business, science, education, home finance and many other fields. But there's no reason why they can't be used for fun and games too.

This book is a collection of fun and helpful programs. They've provided me with hours of entertainment, and I hope they can do the same for you. Some (*Coin Flipper*, *Craps*, *Hangman*, for example) are fairly standard games. Others (*Money Mad*, *Crops*, *Freebush!*) are more unique. Some (*Favorite Song*, *Surprise Poem*) aren't even games, but computerized gags: you'll just have to run the programs to find out what they do!

For each program, I've given two complete listings. One is in standard BASIC, which can be run on most home computers either directly or with only minor changes. The second set of listings is in the abbreviated BASIC used in Radio Shack's Level I TRS-80 computer.


Unfortunately, there isn't much standardization in graphics generation. For this reason none of the games in this book use graphics. You can, however always program in your own graphics.

Some of the programs in this book are quite simple, or based on luck. Others can be quite challenging to play. Some are designed to help you. All are designed for fun. I hope you enjoy them.

Delton T. Horn

Chapter 1

One-Player Games



This chapter contains an assortment of games which pit a single player against the computer. They range from the familiar (*Craps* and *Hangman*) to the bizarre (*Freebish!*), and from games of luck (*Craps*), through games of simple logic (*High/Low*) and games of strategy and choices (*Crops* and *Galactic Search*). I hope you find them fun and challenging. Feel free to program in any variations you might think of.

Craps

Just in case you're not familiar with the game of Craps, let's discuss how it's played. You make a bet (step 15) and roll a pair of dice (steps 85 - 120). If you roll two one's (Snake eyes, steps 25 & 160) or two six's (Boxcars, steps 30 & 175) on your first roll, you lose. On the other hand, if the total of the dice on your first roll equals 7 or 11 (steps 35, 40 & 125), you win. Any other total is recorded as your point and you roll again (steps 45 & 50).

You keep rolling until you match your point (steps 65 & 125) or get 7 or 11 for a total (steps 70, 75 & 185). Matching your point wins, but getting a 7 or 11 on any but the first roll loses.

The computer uses a random number generator to simulate rolling the dice (steps 95 - 110): each of the dice can be from 1 to 6, so the totals can be from 2 to 12. The computer checks for wins and losses, and keeps tabs on your bets if you play more than one round.

Of course this is a game of pure chance so it does not take skill or strategy to play. Consequently, it won't be as entertaining as many of the other games in this book. But the program is simple and fun, and it's worth running a few times just for kicks. See Fig. 1-1 for the flowchart.

Standard BASIC

```
10 PRINT"LET'S SHOOT SOME CRAPS, PAL!":PRINT
12 LET F=0
14 PRINT"WHAT IS YOUR BET":
16 INPUT E
20 GOSUB 85
25 IF C=2 THEN GOTO 160
30 IF C=12 THEN GOTO 175
35 IF C=7 THEN GOTO 125
40 IF C=11 THEN GOTO 125
45 LET D=C
50 PRINT"YOUR POINT IS";D
55 GOSUB 85
60 PRINT "YOUR POINT WAS";D
63 GOSUB 120
67 IF C=D THEN GOTO 125
70 IF C=7 THEN GOTO 185
75 IF C=11 THEN GOTO 185
80 GOTO 55
85 PRINT"PRESS 'ENTER' TO ROLL"
87 INPUT A$
```




Fig. 1-1. Craps flowchart.

```

90 FOR X=1TO200:NEXT X
95 LET A=INT(RND(0)*6)+1
100 LET B=INT(RND(0)*6)+1
105 PRINT:PRINT" ", "*** ";A," YOUR ROLL",B;" ***"
110 LET C=A+B
115 PRINT:PRINT:PRINT"YOU JUST ROLLED ";C
120 FOR X=1TO333:NEXT X:RETURN
125 PRINT" ","YOU WIN!"
130 LET F=F+E
135 IF F<0 THEN GOTO 195
140 PRINT"SO FAR YOU'VE WON $";F
145 LET Y=1:LET N=0
147 PRINT"WANT TO PLAY AGAIN";
149 INPUT G
152 IF G=1 THEN GOTO 15
155 END
160 PRINT" SNAKE EYES!"
162 GOSUB 120
165 PRINT" ","YOU LOSE!"
170 LET F=F-E
172 GOTO 135
175 PRINT"BOX CARS!"
180 GOTO 162
185 PRINT"YOU CRAPPED OUT!"
190 GOTO 162
195 PRINT"SO FAR YOU'VE LOST $";F
200 GOTO 145

```

TRS-80 BASIC

```

10 P."LET'S SHOOT SOME CRAPS, PAL!":P.:F=0
15 IN."WHAT'S YOUR BET";E:GOS.70:IF C=2 G.135
20 IF C=12 G.145
25 IF C=7 G.105
30 IF C=11 G.105
35 D=C:P."YOUR POINT IS ";D
40 GOS.70:P."YOUR POINT WAS ";D:GOS.100:IF C=D G. 105
45 IF C=7 G. 150
50 IF C=11 G. 150
55 G.40
70 IN."PRESS 'ENTER' TO ROLL ";A$:F.X=1TO200:N.X
75 G=RND(30)+10:F.H=1TOG:A=RND(6):B=RND(6):CLS
80 P.:"P. *** ";A," YOUR ROLL",B;" ***"
85 F.X=1TO25:N.X:N.H:C=A+B:P."YOU JUST ROLLED ";C
90 F.X=1TO333:N.X:RET..
105 P. ","YOU WIN!":F=F+E
110 IF F<0 G.155

```

```
115 P."SO FAR YOU'VE WON $";F
120 Y=1:N=0:IN."WANT TO PLAY AGAIN?";G:IFG=1 G.15
125 END
135 P."SNAKE EYES!"
140 P." ", "YOU LOSE!":F=F-E:G.110
145 P." BOX CARS!":G.140
150 P."YOU CRAPPED OUT!":G.140
155 P."SO FAR YOU'VE LOST $";F:G.120
```

Sample Run

```
LET'S SHOOT SOME CRAPS, PAL!
WHAT'S YOUR BET? 100
PRESS 'ENTER' TO ROLL ?_____
*** 5   YOUR ROLL    3 ***
YOU JUST ROLLED 8
YOUR POINT IS 8
PRESS 'ENTER' TO ROLL ?_____
*** 4   YOUR ROLL    2 ***
YOU JUST ROLLED 6
YOUR POINT WAS 8
PRESS 'ENTER' TO ROLL ?_____
*** 6   YOUR ROLL    1 ***
YOU JUST ROLLED 7
YOU CRAPPED OUT
                                         YOU LOSE!
SO FAR YOU'VE LOST $100
WANT TO PLAY AGAIN? YES
WHAT'S YOUR BET? 250
PRESS 'ENTER' TO ROLL ?_____
*** 6   YOUR ROLL    5 ***
YOU JUST ROLLED 11
                                         YOU WIN!
SO FAR YOU'VE WON $150
WANT TO PLAY AGAIN? NO
```

Summary of Variables Used

- A Die #1
- B Die #2
- C Total of last roll
- D Point
- E Current bet
- F Record of winnings (or losses)
- G Timing
- H Timing
- N No (to "PLAY AGAIN?")
- X Timing
- Y Yes (to "PLAY AGAIN?")

High/Low

High/Low is probably the simplest game in this book. The computer randomly selects a number which you try to guess. You get seven tries, and each time the computer tells you if you are too high or too low. In the easy version the number is between 1 and 25. The medium version is between 1 and 40, and the hard version is between 1 and 55. If you prefer you can enter any upper limit you like when the computer asks you if you want an easy, medium or hard game (step 20). The computer only looks at the first letter of any word you type: entering "EASY", "EGG", or just "E" will all be the same to the machine. The values for E, M and H are set in line 15, but you can just enter a number instead of one of the variables. Entering "8" would make for a pathetically easy game, and entering "888888" would be practically impossible.

The computer asks you for a guess seven times (steps 45 to 75). Your guess (called F in the program) is compared to the computer's number (X). If there is a match, the program will be sent to steps 105 to 145 for a win statement. Otherwise the computer determines if your guess is too high or too low and prints the appropriate statement.

If you haven't found the correct number by your seventh guess, the computer tells you the correct number and ends the game. See Fig. 1-2 for the flowchart.

Standard BASIC

```
10 PRINT:PRINT" ","HIGH/LOW":PRINT
15 LET E=25:LET M=40:LET H=55
20 INPUT"EASY, MEDIUM, OR HARD GAME";G
25 LET X=INT(RND(1)*G)+1
30 FOR Z=1 TO 555:NEXT Z
35 PRINT"I AM THINKING OF A NUMBER FROM 1 TO ";G
40 PRINT"YOU GET 7 GUESSES":PRINT
42 FOR Z=1 TO 444:NEXT Z
45 FOR Y=1 TO 7
50 INPUT "YOUR GUESS";F
55 FOR Z=1 TO 333:NEXT Z
60 IF F=X GOTO 105
65 IF F > X PRINT "TOO HIGH!"
70 IF F < X PRINT "TOO LOW!"
75 NEXT Y
80 PRINT"SORRY. TIME'S UP."
```




Fig. 1-2. High-Low flowchart.

```
85 PRINT "THE NUMBER WAS ";
90 FOR Z=1 TO 333:NEXT Z
95 PRINT X
100 END
105 FOR L=1 TO 20
110 FOR Z=1 TO 20
115 PRINT
120 NEXT Z
125 PRINT" ", "WINNER!"
130 FOR Z=1 TO 50:NEXT Z
135 NEXT L
140 PRINT:PRINT"THE NUMBER WAS ";X
145 PRINT"YOU GOT IT ON GUESS #";Y
150 END
```

TRS-80 BASIC

```
10 P.:P." ", "HIGH/LOW":P.:E=25:M=40:H=55
15 IN."EASY, MEDIUM OR HARD GAME";G:X=RND(G)
20 F.Z=1 TO 555:N.Z:P."I AM THINKING OF A NUMBER FROM 1
TO ";G
25 P."YOU GET 7 GUESSES.":P.:F.Z=1 TO 444:N.Z
30 F.Y=1 TO 7:IN."YOUR GUESS";F:F.Z=1 TO 333:N.Z
35 IF F=X G.60
40 IF F > X P."TOO HIGH!"
45 IF F < X P."TOO LOW!"
50 N.Y:P."SORRY. TIME'S UP.":P."THE NUMBER WAS ";
55 F.Z=1TO333:N.Z:P.X:END
60 F.L=1TO20:CLS:P.:P.:P.:F.Z=1TO50:N.Z
65 P." ", "WINNER!":F.Z=1TO50:N.Z:N.L:P.
70 P."THE NUMBER WAS ";X:P."YOU GOT IT ON GUESS #";Y
75 END
```

Sample Run

```
HIGH/LOW
EASY, MEDIUM OR HARD GAME? EASY
I AM THINKING OF A NUMBER FROM 1 TO 25
YOU GET 7 GUESSES
YOUR GUESS? 10
TOO LOW!
YOUR GUESS? 20
TOO HIGH!
YOUR GUESS? 15
TOO LOW!
YOUR GUESS? 17
WINNER!
THE NUMBER WAS 17
YOU GOT IT ON GUESS #4
```

Summary Of Variables Used

- E Easy game. E=25
- F Guess
- F Game range
- H Hard game. H=55
- L Timing variable
- M Medium game. M=40
- X Secret number
- Y Guess count
- Z Timing variable

Guess The Variables

This game is simple enough in concept: it's another number guessing game. But playing this one just might tax your mathematical abilities somewhat. If you always hated algebra, you might want to skip ahead to the next chapter, but if you enjoy solving a good puzzle, read on.

The object of the game is to guess the value of the four variables (A, B, C & D). In an easy game they are between 1 & 10, in a medium game they are between 1 & 20, and in a hard game they are between 1 & 40.

If you request a clue (steps 45, 50, & 225 - 485) the computer will give you an algebraic equation using two or more of the variables, and tell you the result. For example, the computer might give you the following clue; " $C + D/A = 7$ ". By combining several of these clues, you try to determine what the individual variables are.

If you decide to guess the variables (steps 45, 55, & 115 - 170) the computer will ask for each variable in turn, comparing your guesses with the correct answers. If you get all four right, you win, of course (steps 165 & 175 - 210). If you miss any, the computer will tell you how many you had right, but not which ones, and count it as half a clue.

You get up to 20 clues (step 60), then you must take one final stab at guessing the variables (steps 70-80 & 115 - 170). If you miss this time, the computer prints out the correct answers, (steps 85 - 110) and you lose.

The first clue is always the sum of the four variables ($A + B + C + D$) (steps 30 & 35). The other clues are randomly chosen from a list of 25 equations (steps 225 - 485). The computer keeps track of which equations have already been used, so no clue will be used twice in one game (steps 230 & 235).

You might want to use a calculator to help you play this game, or at least a pencil and paper.

This is a good game to teach children math, but let them have fun with it. If you turn it into an extra chore nobody will get much out of it.

It's also a good puzzler for adults. You might want to add some additional clues yourself. I think the program makes it quite obvious how this is done.

Standard BASIC

```
5 FOR Z=1TO25:PRINT:NEXT Z
10 PRINT“GUESS THE VARIABLES”:PRINT
15 LET E=10:LET M=20
17 LET H=40:LET Q=1
20 PRINT“EASY, MEDIUM OR HARD”;:INPUT V
22 LET A=INT(RND(1)*V)+1
24 LET B=INT(RND(1)*V)+1
26 LET C=INT(RND(1)*V)+1
28 LET D=INT(RND(1)*V)+1
30 LET X=A+B+C+D
35 PRINT“A+B+C+D=”;X
40 FOR Z=1TO25:LET A(Z)=0:NEXT Z
45 PRINT“ENTER 1 FOR CLUE OR 2 TO GUESS ”;
47 INPUT W
50 IF W=1 GOSUB 225
55 IF W=2 GOSUB 115
60 IF Q>19.5 GOTO 70
65 GOTO 45
70 PRINT“SORRY. YOU’VE HAD 20 CLUES.””
75 PRINT“HERE’S YOUR LAST CHANCE TO GUESS.””
80 GOSUB 115
85 PRINT“HERE ARE THE CORRECT ANSWERS”;:PRINT
90 PRINT“A =”;:GOSUB 220:PRINT A
95 PRINT“B =”;:GOSUB 220:PRINT B
100 PRINT“C =”;:GOSUB 220:PRINT C
105 PRINT“D =”;:GOSUB 220:PRINT D
110 END
115 PRINT“A =”;:INPUT E
120 PRINT“B =”;:INPUT F
125 PRINT“C =”;:INPUT G
130 PRINT“D =”;:INPUT H
135 LET T=0: IF E=A THEN LET T=T+1
140 IF F=B THEN LET T=T+1
145 IF G=C THEN LET T=T+1
150 IF H=D THEN LET T=T+1
155 FOR Z=1TO444:NEXT Z
160 PRINT “YOU GOT ”;T;“VARIABLES CORRECT.””
165 IF T=4 GOTO 175
170 LET Q=Q+0.5:RETURN
175 PRINT“YOU USED ”;Q;“CLUES.””
180 IF Q<10 PRINT“FANTASTIC!”
185 IF Q<4 PRINT“DID YOU CHEAT?”
190 IF Q>16 PRINT“IM NOT PARTICULARLY IMPRESSED.””
195 END
220 FOR Z=1TO333:NEXT Z;RETURN
```

```
225 LET S=INT(RND(0)*25)+1
230 IF A(S)=1 GOTO 225
235 LET A(S)=1:LET Q=Q+1
240 IF S=1 GOTO 370
245 IF S=2 GOTO 380
250 IF S=3 GOTO 390
255 IF S=4 GOTO 400
260 IF S=5 GOTO 410
265 IF S=6 GOTO 420
270 IF S=7 GOTO 430
275 IF S=8 GOTO 440
280 IF S=9 GOTO 450
285 IF S=10 GOTO 460
290 IF S=11 GOTO 470
295 IF S=12 GOTO 480
300 IF S=13 GOTO 490
305 IF S=14 GOTO 500
310 IF S=15 GOTO 510
315 IF S=16 GOTO 520
320 IF S=17 GOTO 530
325 IF S=18 GOTO 540
330 IF S=19 GOTO 550
335 IF S=20 GOTO 560
340 IF S=21 GOTO 570
345 IF S=22 GOTO 580
350 IF S=23 GOTO 590
355 IF S=24 GOTO 600
360 LET X=A+B
365 PRINT"A+B=";X:RETURN
370 LET X=A+C
375 PRINT"A+C=";X:RETURN
380 LET X=A+D
385 PRINT"A+D=";X:RETURN
390 LET X=B+C
395 PRINT"B+C=";X:RETURN
400 LET X=B+D
405 PRINT"B+D=";X:RETURN
410 LET X=C+D
415 PRINT"C+D=";X:RETURN
420 LET X=A*B*C*D*
425 PRINT"A X B X C X D =";X:RETURN
430 LET X=A*B
435 PRINT"A X B =";X:RETURN
440 LET X=A*C
445 PRINT"A X C =";X:RETURN
450 LET X=A*D
455 PRINT"A X D =";X:RETURN
```

```

460 LET X=B*C
465 PRINT"B X C =";X:RETURN
470 LET X=B*D
475 PRINT"B X D =";X:RETURN
480 LET X=C*D
485 PRINT"C X D =";X:RETURN
490 LET X=A+B/C
495 PRINT"A + B/C =";X:RETURN
500 LET X=B+C/D
505 PRINT"B + C/D =";X:RETURN
510 LET X=C+D/A
515 PRINT"C + D/A =";X:RETURN
520 LET X=(A+B)*(C-D)
525 PRINT"(A+B)(C-D) =";X:RETURN
530 LET X=(B-A)*(C+D)
535 PRINT"(B-A)(C+D) =";X:RETURN
540 LET X=A*B-C
545 PRINT"A X B - C =";X:RETURN
550 LET X=A*(B-C)
555 PRINT"A X (B-C) =";X:RETURN
560 LET X=A*A-B
565 PRINT"A SQUARED - B =";X:RETURN
570 LET X=A*A-C
575 PRINT"A SQUARED - C =";X:RETURN
580 LET X=B*B-D
585 PRINT"B SQUARED - D =";X:RETURN
590 LET X=D*D-A
595 PRINT"D SQUARED - A =";X:RETURN
600 LET X=C*C-(A/(B*B))+D
605 PRINT"C SQUARED - A/B SQUARED + D ="; X
610 RETURN

```

TRS-80 BASIC

```

10 CLS:P.:P."GUESS THE VARIABLES":E=10:M=20:H=40
15 Q=1:IN."EASY, MEDIUM OR HARD ";V
20 A=RND(V):B=RND(V):C=RND(V):D=RND(V):X=A+B+C+D
25 P."A+B+C+D =";X:F.Z=1TO25:A(25)=0:N.Z
30 IN."ENTER 1 FOR CLUE OR 2 TO GUESS ";W:IF W=1 GOS.225
35 IFW=2 GOS.75
40 IF Q>19.5 G.50
45 G.30
50 P."SORRY. YOU'VE USED 20 CLUES."
55 P."HERE'S YOUR LAST CHANCE TO GUESS THE
VARIABLES":GOS.75
60 P."HERE ARE THE CORRECT ANSWERS":P.:P."A = ";
65 GOS.220:P.A:P."B = ";:GOS.220:P.B:P."C = ";

```

```

70 GOS.220:P.C:P."D =";:GOS.220:P.D:END
75 T=0:IN."A =";E:IN."B =";F:IN."C =";G
80 IN."D =";H:IF E=A THEN T=T+1
85 IF F=B THEN T=T+1
90 IF G=C THEN T=T+1
95 IF H=D THEN T=T+1
100 F.Z=1TO444:N.Z:P."YOU GOT ";T;" VARIABLES CORRECT."
105 IF T=4 G.120
110 Q=A+0.5:RET.
120 P."YOU USED ";Q;" CLUES.":IF Q<10 P."FANTASTIC!"
125 IF Q<4 P."DID YOU CHEAT?"
130 IF>16 P."I'M NOT PARTICULARLY IMPRESSED."
135 END
220 F.Z=1TO333:N.Z:RET.
225 S=RND(25)
230 IF A(S)=1G.225
235 A(S)=1:Q=Q+1:IF S=1G.365
240 IF S=2G.370
245 IF S=3G.375
250 IF S=4G.380
255 IF S=5G.385
260 IF S=6G.390
265 IF S=7G.395
270 IF S=8G.400
275 IF S=9G.405
280 IF S=10G.410
285 IF S=11G.415
290 IF S=12G.420
295 IF S=13G.425
300 IF S=14G.430
305 IF S=15G.435
310 IF S=16G.440
315 IF S=17G.445
320 IF S=18G.450
325 IF S=19G.455
330 IF S=20G.460
335 IF S=21G.465
340 IF S=22G.470
345 IF S=23G.475
350 IF S=24G.480
355 X=A+B:P."A+B=";X:RET.
365 X=A+C:P."A+C=";X:RET.
370 X=A+D:P."A+D=";X:RET.
375 X=B+C:P."B+C=";X:RET.
380 X=B+D:P."B+D=";X:RET.
385 X=C+D:P."C+D=";X:RET.
390 X=A*B*C*D:P."A X B X C X D =";X:RET.

```

```

395 X=A*B:P.“A X B =”;X:RET.
400 X=A*C:P.“A X C =”;X:RET.
405 X=A*D:P.“A X D =”;X:RET.
410 X=B*C:P.“B X C =”;X:RET.
415 X=C*D:P.“C X D =”;X:RET.
420 X=B*D:P.“B X D =”;X:RET.
425 X=A+B/C:P.“A+ B/C=”;X:RET.
430 X=B+C/D:P.“B + C/D=”;X:RET.
435 X=C+D/A:P.“C + D/A=”;X:RET.
440 X=(A+B)*(C-D):P.“(A+B)(C-D)=”;X:RET.
445 X=(B-A)*(C+D):P.“(B-A)(C+D)=”;X:RET.
450 X=A*B-C:P.“A X B - C =”;X:RET.
455 X=A*(B-C):P.“A X (B-C)=”;X:RET.
460 X=A*A-B:P.“A SQUARED - B =”;X:RET.
465 X=A*A-C:P.“A SQUARED - C =”;X:RET.
470 X=B*B-D:P.“B SQUARED - D =”;X:RET.
475 X=D*D-A:P.“D SQUARED - A =”;X:RET.
480 X=C*C-(A/(B*B))+D:P.“C SQUARED - A/B SQUARED +
D=”;X
485 RET.

```

Sample Run

GUESS THE VARIABLES
EASY, MEDIUM, OR HARD ?EASY
A+B+C+D=20
ENTER 1 FOR CLUE OR 2 TO GUESS ?1
B+D=12
ENTER 1 FOR CLUE OR 2 TO GUESS ?1
A X B - C =32
ENTER 1 FOR CLUE OR 2 TO GUESS ?1
A SQUARED - B =18
ENTER 1 FOR CLUE OR 2 TO GUESS ?1
C X D =15
ENTER 1 FOR CLUE OR 2 TO GUESS ?2
A = ?6
B = ?5
C = ?5
D = ?3
YOU GOT 0 VARIABLES CORRECT.
ENTER 1 FOR CLUE OR 2 TO GUESS ?1
C SQUARED - A/B SQUARED + D =13.89796
ENTER 1 FOR CLUE OR 2 TO GUESS ?1
B X D =35
ENTER 1 FOR CLUE OR 2 TO GUESS ?2
A = ?4
B = ?7

C = ?3

D = ?5

YOU GOT 3 VARIABLES CORRECT.

ENTER 1 FOR CLUE OR 2 TO GUESS ?1

A+D=10

ENTER 1 FOR CLUE OR 2 TO GUESS ?2

A = ?5

B = ?7

C = ?3

D = ?5

YOU GOT 4 VARIABLES CORRECT.

YOU USED 8 CLUES.

FANTASTIC!

>READY

Summary of Variables Used

- A unknown variable #1
- B unknown variable #2
- C unknown variable #3
- D unknown variable #4
- E Easy — E=10 / Guess variable A
- F Guess variable B
- G Guess variable C
- H Hard — H=40 / Guess variable D
- M Medium — M=20
- Q Clue count
- S Clue selection
- W Clue or guess?
- V Game level
- X Result of clue
- Z Timing variable

Hangman

Hangman is a popular old pencil and paper game. In this version the computer selects a five letter word (out of 50 pre-programmed possibilities) and you have to guess what it is, letter by letter. If you correctly guess a letter it is printed in the proper position in the secret word. If it is used in the word more than once it will be printed wherever it occurs. An incorrect guess adds another letter to the word HANGMAN. If you spell HANGMAN before completing the secret word, you lose. Guessing the secret word, of course, is a win. You get a maximum of 11 letter guesses for each word, and 13 different letters are used in the various secret words.

On each play you're reminded that the letters B,D,G,J,M,Q,U,V,X,Y, and Z are not to be used. These letters are used as variables to run the program. V,W,X,Y, and Z are especially to be avoided since these are the variables that contain the secret word. To preserve program simplicity you are left on your honor not to use these letters, since the computer cannot distinguish between them and a correct guess. See Fig. 1-3 for the flowchart.

If you lose, the computer will tell you what the secret word was.

Of course you can program in your own words. For example, if you want to add the word "HORSE". Change line 30 to LET U=INT(RND(0)*51)+1. Then add the following steps:

```
121 IF U>50 THEN GOTO 1025  
1025 LET V=15:LET W=19  
1030 LET X=20:LET Y=21  
1035 LET Z=13  
1040 GOTO 140  
OR you could substitute  
1025 LET V=H:LET W=O  
1030 LET X=R:LET Y=S  
1035 LET Z=E
```

The numerical values of the letters are given in steps 07 to 14. Since this program is largely repetitive, only a portion of a sample run is given.

Standard BASIC

```
05 REM*SET LETTER VALUES*  
07 LET A=11:LET C=12:LET E=13:LET F=14
```

```
10 LET H=15:LET I=16:LET K=17:LET L=18
12 LET N=23:LET O=19:LET R=20:LET S=21
14 LET T=22:LET D=0:LET G=0:LET J=0
16 LET M=0:LET Q=0
18 REM*CLEAR WRONG GUESS COUNTER*
20 FOR X=1TO20:LET A(X)=0:NEXT X
25 REM*RANDOM WORD SELECTION*
30 LET U=INT(RND(0)*50)+1
35 IF U<4 THEN GOTO 500
40 IF U<9 THEN GOTO 525
45 IF U<14 THEN GOTO 565
50 IF U<19 THEN GOTO 600
55 IF U<23 THEN GOTO 635
60 IF U<26 THEN GOTO 665
65 IF U<29 THEN GOTO 690
70 IF U<33 THEN GOTO 715
75 IF U<36 THEN GOTO 755
80 IF U<38 THEN GOTO 775
85 IF U=38 THEN GOTO 795
90 IF U<41 THEN GOTO 800
95 IF U=41 THEN GOTO 820
100 IF U=42 THEN GOTO 825
105 IF U<45 THEN GOTO 830
110 IF U=45 THEN GOTO 845
115 IF U=46 THEN GOTO 850
120 IF U<49 THEN GOTO 855
122 LET V=22
125 IF U=49 THEN GOTO 870
130 LET W=13:LET X=13
135 LET Y=22:LET Z=15
140 REM*THE PLAY*
145 CLS:PRINT " ", "HANGMAN"
147 PRINT "THESE LETTERS ARE NOT USED — B,D,G,J,"
150 PRINT "M,Q,U,V,W,X,Y, & Z."
152 LET B=A(15)
154 IF B=0 THEN GOTO 165
156 PRINT "LETTERS USED SO FAR —";
158 FOR U=1TO B:LET A(19)=A(U)
160 GOSUB 880
162 NEXT U
165 PRINT:PRINT
170 IF A(17)=0 THEN PRINT "____"
175 IF A(17)=1 THEN PRINT "H____"
180 IF A(17)=2 THEN PRINT "HA____"
185 IF A(17)=3 THEN PRINT "HAN____"
190 IF A(17)=4 THEN PRINT "HANG____"
```




Fig. 1-3. Hangman flowchart.

```

195 IF A(17)=5 THEN PRINT“HANGM—”
200 IF A(17)=6 THEN PRINT“HANGMA—”
205 IF A(17)=7 THEN GOTO 950
210 PRINT:PRINT:PRINT“ ”,”*”;
212 LET A(18)=0
215 LET A(19)=D:GOSUB 875
220 LET A(19)=G:GOSUB 875
225 LET A(19)=J:GOSUB 875
230 LET A(19)=M:GOSUB 875
235 LET A(19)=Q:GOSUB 875
237 REM*WIN TEST*
238 IF A (18)=5 THEN GOTO 1000
240 PRINT“*”:PRINT
242 PRINT“YOUR GUESS”;
245 INPUT U

```

247 LET A(18)=0
248 REM* CHECK FOR CORRECT GUESS *
250 IF U=V THEN GOSUB 975
255 IF U=W THEN GOSUB 980
260 IF U=X THEN GOSUB 985
265 IF U=Y THEN GOSUB 990
270 IF U=Z THEN GOSUB 995
275 IF A(18)=0 THEN GOTO 290
280 LET B=A(15)
282 LET B=B+1
284 LET A(15)=B
286 A(B)=U
288 GOTO 140
290 LET B=A (17)
295 LET B=B+1
300 LET A (17)=B
305 GOTO 280
500 LET X=11:LET Y=12
502 LET Z=17:LET V=21
505 IF U=1 THEN LET W=15
510 IF U=2 THEN LET W=22
515 IF U=3 THEN LET W=18
520 GOTO 140
525 LET W =11:LET X=17
527 LET Y=13:LET Z=21
530 IF U=4 THEN LET V=14
535 IF U=5 THEN LET V=20
540 IF U=6 THEN LET V=22
545 IF U=7 THEN LET V=18
550 IF U=8 THEN LET V=12
555 GOTO 140
565 LET W=11: LET X=12
570 LET Y=17:LET Z=21
572 IF U=9 THEN LET V=15
575 IF U=10 THEN LET V=18
580 IF U=11 THEN LET V=20
585 IF U=12 THEN LET V=21
590 IF U=13 THEN LET V=22
595 GOTO 140
600 LET W=13:LET Z=11
605 LET Y=20:LET Z=21
607 IF U=14 THEN LET V=14
610 IF U=15 THEN LET V=15
615 IF U=16 THEN LET V=20
620 IF U=17 THEN LET V=21
625 IF U=18 THEN LET V=22
630 GOTO 140

635 LET W=16:LET X=23
637 LET Y=13:LET Z=21
640 IF U=19 THEN LET V=23
645 IF U=20 THEN LET V=22
650 IF U=21 THEN LET V=18
655 IF U=22 THEN LET V=14
660 GOTO 140
665 LET W=16:LET X=12
670 LET Y=17:LET Z=21
672 IF U=23 THEN LET V=17
675 IF U=24 THEN LET V=18
680 IF U=25 THEN LET V=22
685 GOTO 140
690 LET W=19:LET X=19
695 LET Y=17:LET Z=21
700 IF U=26 THEN LET V=12
702 IF U=27 THEN LET V=15
705 IF U=28 THEN LET V=18
710 GOTO 140
715 LET X=16: LET Y=12:LET Z=17
720 IF U>30 THEN GOTO 740
725 LET V=21:LET W=22
730 IF U=30 THEN LET W=18
735 GOTO 140
740 LET V=12:LET W=20
745 IF U=32 THEN LET V=22
750 GOTO 140
755 LET W=20:LET X=11
760 LET Y=12:LET Z=17
762 IF U=34 THEN LET V=12
765 IF U=35 THEN LET V=22
770 GOTO 140
775 LET X=11:LET Y=20:LET Z=21
780 IF U=36 THEN GOTO 790
785 LET V=21:LET W=22
787 GOTO 140
790 LET V=22:LET W=21
792 GOTO 140
795 LET V=21:LET W=22:LET X=11
797 LET Y=20:LET Z=22:GOTO 140
800 LET V=21:LET W=15
802 LET X=11:LET Y=14:LET Z=22
805 IF U=40 THEN LET X=16
810 GOTO 140
820 LET V=12:LET W=11:LET X=20
822 LET Y=13:LET Z=21:GOTO 140

```
825 LET V=13:LET W=20:LET X=20
827 LET Y=14:LET Z=20:GOTO 140
830 LET V=14:LET Y=22
832 IF U=43 GOTO 840
835 LET W=11:LET X=12
837 LET Z=12:GOTO 140
840 LET W=16:LET X=14
842 LET Z=15:GOTO 140
845 LET V=17:LET W=23:LET X=13
847 LET Y=13:LET Z=18:GOTO 140
850 LET V=18:LET W=16:LET X=17
852 LET Y=13:LET Z=21:GOTO 140
855 LET V=21
857 IF U=47 GOTO 865
860 LET W=17:LET X=16:LET Y=18
862 LET Z=18:GOTO 140
865 LET W=15:LET X=13:LET Y=11
867 LET Z=14:GOTO 140
870 LET W=15:LET X=13:LET Y=13
872 LET Z=14:GOTO 140
875 REM*CONVERSION TO LETTERS*
877 IF A(19)=0 GOTO 1015
880 IF A(19)=11 THEN PRINT"A";
885 IF A(19)=12 THEN PRINT"C";
890 IF A(19)=13 THEN PRINT"E";
895 IF A(19)=14 THEN PRINT"F";
900 IF A(19)=15 THEN PRINT"H";
905 IF A(19)=16 THEN PRINT"I";
910 IF A(19)=17 THEN PRINT"K";
915 IF A(19)=18 THEN PRINT"L";
920 IF A(19)=19 THEN PRINT"O";
925 IF A(19)=20 THEN PRINT"R";
930 IF A(19)=21 THEN PRINT"S";
935 IF A(19)=22 THEN PRINT"T";
940 IF A(19)=23 THEN PRINT"N";
942 LET A(18)=A(18)+1
945 RETURN
950 PRINT"HANGMAN":PRINT
955 PRINT",","YOU LOSE"
957 PRINT:PRINT"THE WORD WAS",
960 LET A(19)=V:GOSUB 875
962 LET A(19)=W:GOSUB 875
964 LET A(19)=X:GOSUB 875
966 LET A(19)=Y:GOSUB 875
968 LET A(19)=Z:GOSUB 875
```

```
970 END
975 LET D=V:LET A(18)=A(18)+1
977 RETURN
980 LET G=W:LET A(18)=A(18)+1
982 RETURN
985 LET J=X:LET A(18)=A(18)+1
987 RETURN
990 LET M=Y:LET A(18)=A(18)+1
992 RETURN
995 LET Q=Z:LET A(18)=A(18)+1
997 RETURN
1000 PRINT“ *”:PRINT:PRINT
1005 PRINT“YOU WIN!”
1010 END
1015 PRINT“?”,;
1020 RETURN
```

TRS-80 BASIC

```
10 A=11:C=12:E=13:F=14:H=15:I=16:K=17
15 L=18:N=23:O=19:R=20:S=21:T=22:D=0
20 G=0:J=0:M=0:Q=0
25 F,X=1TO20:A(X)=0:N,X
30 U=RND(50):IF U<4G.500
35 IF U<9G.525
40 IF U<14G.565
45 IF U<19G.600
50 IF U<23G.635
55 IF U<26G.665
60 IF U<29G.690
65 IF U<33G.715
70 IF U<36G.775
80 IF U=38 G.795
85 IF U<41G.800
90 IF U=41G.820
95 IF U=42G.825
100 IF U<45G.830
105 IF U=45G.845
110 IF U=46G.850
115 IF U<49G.855
120 V=22:IF U=49G.870
125 W=13:X=13:Y=22:Z=15
140 CLS:P.“ ”,“HANGMAN”
145 P.“THESE LETTERS ARE NOT USED—B,D,G,J,M,Q,U,
    V,W,X,Y, & Z”
150 B=A(15):IF B=0G.165
155 P.“LETTERS USED SO FAR—”;
```

160 F.U=1TOB:A(19)=A(U):GOS.880:N.U
165 P.:P.:IF A(17)=0P.“____”
170 IF A(17)=1P.“H____”
175 IF A(17)=2P.“HA____”
180 IF A(17)=3P.“HAN____”
185 IF A(17)=4P.“HANG____”
190 IF A(17)=5P.“HANGM____”
195 IF A(17)=6P.“HANGMA____”
200 IFA(17)=7G.950
210 P.:P.:P.“”,“*”;:A(18)=0
215 A(19)=D:GOS.875:A(19)=G:GOS.875
220 A(19)=J:GOS.875:A(19)=M:GOS.875
225 A(19)=Q:GOS.875:IFA(18)=5G.1000
240 P.“ *”:P.
245 IN.“YOUR GUESS”;U:A(18)=0:IFU=VGOS.975
250 IF U=WGOS.980
255 IF U=XGOS.985
260 IF U=YGOS.990
265 IF U=ZGOS.995
270 IF A(18)=0G.290
275 B=A(15):B=B+1:A(15)=B:A(B)=U:G.140
290 A(17)=A(17)+1:G.275
500 X=11:Y=12:Z=17:V=21:IF U=1 THEN W=15
505 IF U=2 THEN W=22
510 IF U=3 THEN W=18
515 G.140
525 W=11:X=17:Y=13:Z=21:IF U=4 THEN V=14
530 IF U=5 THEN V=20
535 IF U=6 THEN V=22
540 IF U=7 THEN V=18
545 IF U=8 THEN V=12
550 G.140
565 W=11:X=12:Y=17:Z=21:IF U=9 THEN V=15
570 IF U=10 THEN V=18
575 IF U=11 THEN V=20
580 IF U=12 THEN V=21
585 IF U=13 THEN V=22
590 G.140
600 W=13:X=11:Y=20:Z=21:IF U=14 THEN V=14
605 IF U=15 THEN V=15
610 IF U=16 THEN V=20
615 IF U=17 THEN V=21
620 IF U=18 THEN V=22
625 G.140
635 W=16:X=23:Y=13:Z=21:IF U=19 THEN V=23
640 IF U=20 THEN V=22

645 IF U=21 THEN V=18
 650 IF U=22 THEN V=14
 655 G.140
 665 W=16:X=12:Y=17:Z=21:V=22:IF U=23 THEN V=17
 670 IF U=24 THEN V=18
 680 G.140
 690 V=12:W=19:X=19:Y=17:Z=21:IF U=27 THEN V=15
 700 IF U=28 THEN V=18
 710 G.140
 715 X=16:Y=12:Z=17:IF U>30 G.740
 720 V=21:W=22:IF U=30 THEN W=18
 730 G.140
 740 V=12:W=20:IF U=32 THEN V=22
 750 G.140
 755 V=12:W=20:X=11:Y=12:Z=17:IF U=35 THEN V=22
 760 G.140
 775 X=11:Y=20:Z=21:IF U=36 G.790
 780 V=21:W=22:G.140
 790 V=22:W=21:G.140
 795 V=21:W=22:X=11:Y=20:Z=22:G.140
 800 V=21:W=15:X=11:Y=14:Z=22:IF U=40 THEN X=16
 810 G.140
 820 V=12:W=11:X=20:Y=13:Z=21:G.140
 825 V=13:W=20:X=20:Y=14:Z=20:G.140
 830 V=14:Y=22:IF U=43 G.840
 835 W=11:X=12:Z=21:G.140
 840 W=16:X=14:Z=15:G.140
 845 V=17:W=23:X=13:Y=13:Z=18:G.140
 850 V=18:W=16:X=17:Y=13:Z=21:G.140
 855 V=21:IF U=47 G.865
 860 W=17:X=16:Y=18:Z=18:G.140
 865 W=15:X=13:Y=11:Z=14:G.140
 870 W=15:X=16:Y=13:Z=14:G.140
 875 IF A(19)=0 G.1015
 880 IF A(19)=11P.“A”;
 885 IF A(19)=12 P.“C”;
 890 IF A(19)=13 P.“E”;
 895 IF A(19)=14P.“F”;
 900 IF A(19)=15 P.“H”;
 905 IF A(19)=16P.“T”;
 910 IF A(19)=17P.“K”;
 915 IF A(19)=18P.“L”;
 920 IF A(19)=19P.“O”;
 925 IF A(19)=20P.“R”;
 930 IF A(19)=21P.“S”;
 935 IF A(19)=22P.“T”;

940 IF A(19)=23P.“N”;
945 A(18)=A(18)+1:RET.
950 P.“HANGMAN”:P.:P.“YOU LOSE”:P.
955 P.“THE WORD WAS ”;:A(19)=V:GOS.875
960 A(19)=W:GOS.875:A(19)=X:GOS.875
965 A(19)=Y:GOS.875:A(19)=Z:GOS.875
970 END
975 D=V:A(18)=A(18)+1:RET.
980 G=W:A(18)=A(18)+1:RET.
985 J=X:A(18)=A(18)+1:RET.
990 M=Y:A(18)=A(18)+1:RET.
995 Q=Z:A(18)=A(18)+1:RET
1000 P.“ *”:P.:P.“YOU WIN!:END
1015 P.“?”;:RET.

Summary Of Variables Used

A 11 * POSSIBLE LETTER
B Letter count
C 12 * POSSIBLE LETTER
D CORRECT GUESS * 1st POSITION
E 13 * POSSIBLE LETTER
F 14 * POSSIBLE LETTER
G CORRECT GUESS * 2nd POSITION
H 15 * POSSIBLE LETTER
I 16 * POSSIBLE LETTER
J CORRECT GUESS * 3rd POSITION
K 17 * POSSIBLE LETTER
L 18 * POSSIBLE LETTER
M CORRECT GUESS * 4th POSITION
N 23 * POSSIBLE LETTER
O 19 * POSSIBLE LETTER
P NOT USED
Q CORRECT GUESS * 5th POSITION
R 20 * POSSIBLE LETTER
S 21 * POSSIBLE LETTER
T 22 * POSSIBLE LETTER
U WORD SELECTION/CURRENT GUESS
V SECRET WORD * 1st LETTER
W SECRET WORD * 2nd LETTER
X SECRET WORD * 3rd LETTER
Y SECRET WORD * 4th LETTER
Z SECRET WORD * 5th LETTER
A(1)–A(11) LETTERS USED SO FAR
A(15) LETTER COUNT
A(17)–A(19) VARIABLES

Sample Run (excerpt)

HANGMAN
THESE LETTERS ARE NOT USED—B,D,G,J,M,Q,U,V,W,X,Y, & Z
LETTERS USED SO FAR—E A I S T
HAN

* ? A ? I ? *
YOUR GUESS? R

HANGMAN
THESE LETTERS ARE NOT USED—B,D,G,J,M,Q,U,V,W,X,Y, & Z
LETTERS USED SO FAR—E A I S T R
HAN

* R A ? I ? *
YOUR GUESS? N

HANGMAN
THESE LETTERS ARE NOT USED—B,D,G,J,M,Q,U,V,W,X,Y, & Z
LETTERS USED SO FAR—E A I S T R N
HANG

* R A ? I ? *
YOUR GUESS? O

Freebish!

The Game With Unknown Rules

Freebish! is probably the strangest game in this book, since part of the fun is the challenge of figuring out the rules. I recommend running the program a few times before reading the explanation (which is right after the program listings).

There are rules, even though it might not seem that way. Have fun, and if you get a little frustrated, try not to take it out on your computer. Just think of how much fun it will be when you get to watch someone else try to figure it out.

A sample run might give it away, so I'm not including one.

Standard BASIC

```
2 FOR X=1 TO 40:PRINT
4 NEXT X
6 LET A=INT (RND(0)*21)-10
8 LET B=INT(RND(0)*21)-10
10 LET C=INT(RND(0)*21)-10
12 LET D=INT(RND(0)*21)-10
14 LET E=INT(RND(0)*21)-10
16 LET F=INT(RND(0)*21)-10
18 LET G=INT(RND(0)*21)-10
20 LET H=INT(RND(0)*21)-10
22 LET I=INT(RND(0)*21)-10
24 LET J=INT(RND(0)*21)-10
26 LET K=INT(RND(0)*21)-10
28 LET L=INT(RND(0)*21)-10
30 LET M=INT(RND(0)*21)-10
32 LET N=INT(RND(0)*21)-10
34 LET Q=INT(RND(0)*21)-10
36 LET R=INT(RND(0)*21)-10
38 LET S=INT(RND(0)*21)-10
40 LET T=INT(RND(0)*21)-0
42 LET U=INT(RND(0)*21)-10
44 LET V=INT(RND(0)*21)-10
46 LET O=0:LET A(1)=0
48 LET W=0:LET A(2)=0
50 LET X=0:LET Z=0
52 LET A(3)=-50
54 REM* THE GAME BEGINS *
56 PRINT“FREEBISH!”
58 LET Y=F
60 LET A(2)=F
62 INPUT X
```

```
64 GOSUB 180
66 LET W=INT(RND(0)*20)+1
68 IF W=1 THEN GOTO 225
70 IF W=2 THEN GOTO 255
72 IF W=3 THEN GOTO 270
74 IF W=4 THEN GOTO 285
76 IF W=5 THEN GOTO 300
78 IF W=6 THEN GOTO 315
80 IF W=7 THEN GOTO 330
82 IF W=8 THEN GOTO 345
84 IF W=9 THEN GOTO 360
86 IF W=10 THEN GOTO 375
88 IF W=11 THEN GOTO 390
90 IF W=12 THEN GOTO 405
92 IF W=13 THEN GOTO 420
94 IF W=14 THEN GOTO 435
96 IF W=15 THEN GOTO 450
98 IF W=16 THEN GOTO 465
100 IF W=17 THEN GOTO 480
102 IF W=18 THEN GOTO 495
104 IF W=19 THEN GOTO 510
110 GOTO 525
160 LET A(2)=A(2)+Y
162 LET Z=0:LET W=0
164 LET A(1)=A(1)-(Y/2)
170 INPUT X
175 GOTO 64
180 IF X=A(3) THEN GOTO 220
185 IF X=Y THEN GOTO 220
187 LET A(1)=A(1)+X
190 LET A(3)=X
192 FOR Z=1 TO 470:NEXT Z
195 PRINT" SCORE"
197 PRINT"THE BRAIN",A(2)
200 PRINT"THE CLOD",A(1)
205 IF A(1)>35 THEN GOTO 550
208 IF A(2)>35 THEN GOTO 540
210 PRINT:PRINT
212 FOR Z=1TO666:NEXT Z
215 LET Z=0
217 RETURN
220 LET A(1)=A(1)-10:FOR Z=1 TO 470:NEXT Z
222 GOTO 195
225 LET Z=A:GOSUB 240
227 IF Z=50 THEN GOTO 66
230 PRINT"ALBOOKISH!"
```

```
232 LET Y=A
235 GOTO 160
240 IF Z=Y THEN LET Z=50
245 IF Z=X THEN LET Z=50
250 RETURN
255 LET Z=B:GOSUB 240
257 IF Z=50 THEN GOTO 66
260 PRINT“BREEP!”
262 LET Y=B
265 GOTO 160
270 LET Z=C:GOSUB 240
272 IF Z=50 THEN GOTO 66
275 PRINT“CRIPKLUTS!”
277 LET Y=C
280 GOTO 160
285 LET Z=D:GOSUB 240
287 IF Z=50 THEN GOTO 66
290 PRINT“DUFFLEEM!”
292 LET Y=D
295 GOTO 160
300 LET Z=E:GOSUB 240
302 IF Z=50 THEN GOTO 66
305 PRINT“EXQUIMBLE!”
307 LET Y=E
310 GOTO 160
315 LET Z=F:GOSUB 240
317 IF Z=50 THEN GOTO 66
320 PRINT“FLIBBINK!”
322 LET Y=F
325 GOTO 160
330 LET Z=G:GOSUB 240
332 IF Z=50 THEN GOTO 66
335 PRINT“GRUNKITT!”
337 LET Y=G
340 GOTO 160
345 LET Z=H:GOSUB 240
347 IF Z=50 THEN GOTO 66
350 PRINT“HEEJAMBOONKIE!”
352 LET Y=H
355 GOTO 160
360 LET Z=I:GOSUB 240
362 IF Z=50 THEN GOTO 66
365 PRINT“ISTRIM!”
367 LET Y=I
370 GOTO 160
375 LET Z=J:GOSUB 240
```

```
377 IF Z=50 THEN GOTO 66
380 PRINT"JACQUELPHLOOM!"
382 LET Y=J
385 GOTO 160
390 LET Z=K:GOSUB 240
392 IF Z=50 THEN GOTO 66
395 PRINT"KREECK!"
397 LET Y=K
400 GOTO 160
405 LET Z=L:GOSUB 240
407 IF Z=50 THEN GOTO 66
410 PRINT"LORKE!"
412 LET Y=L
415 GOTO 160
420 LET Z=M:GOSUB 240
422 IF Z=50 THEN GOTO 66
425 PRINT"MUCSTILE!"
427 LET Y=M
430 GOTO 160
435 LET Z=N:GOSUB 240
437 IF Z=50 THEN GOTO 66
440 PRINT"NEFLOM!"
442 LET Y=N
445 GOTO 160
450 LET Z=Q:GOSUB 240
452 IF Z=50 THEN GOTO 66
455 PRINT"QUIGGGLESBY!"
457 LET Y=Q
460 GOTO 160
465 LET Z=R:GOSUB 240
467 IF Z=50 THEN GOTO 66
470 PRINT"RECKLEBOP!"
472 LET Y=R
475 GOTO 160
480 LET Z=S:GOSUB 240
482 IF Z=50 THEN GOTO 66
485 PRINT"SNORK!"
487 LET Y=S
490 GOTO 160
495 LET Z=T:GOSUB 240
497 IF Z=50 THEN GOTO 66
500 PRINT"THUBBLE!"
502 LET Y=T
505 GOTO 160
510 LET Z=U:GOSUB 240
512 IF Z=50 THEN GOTO 66
```

```
515 PRINT“UBENZZERT!”  
517 LET Y=U  
520 GOTO 160  
525 LET Z=V:GOSUB 240  
527 IF Z=50 THEN GOTO 66  
530 PRINT“VELK!”  
532 LET Y=V  
535 GOTO 160  
540 PRINT“I WIN, ”  
542 GOSUB 570  
545 PRINT“OF COURSE.”  
547 END  
550 PRINT“ HEY! ”;  
552 GOSUB 570  
555 PRINT“YOU WON!”  
557 GOSUB 570  
560 PRINT“THAT WASN’T SUPPOSED TO HAPPEN!”  
565 END  
570 FOR Z=1 TO 555  
575 NEXT Z  
580 RETURN
```

TRS-80 BASIC

```
5 CLS:P.:P.:P.:A=RND(21)-11:B=RND(21)-11:C=RND(21)-11  
10 D=RND(21)-11:E=RND(21)-11:F=RND(21)-11:G=RND(21)-11  
15 H=RND(21)-11:I=RND(21)-11:J=RND(21)-11:K=RND(21)-11  
20 L=RND(21)-11:M=RND(21)-11:N=RND(21)-11:Q=RND(21)-11  
25 R=RND(21)-11:S=RND(21)-11:T=RND(21)-11:T=RND(21)-11  
30 U=RND(21)-11:V=RND(21)-11:O=0:W=0:X=0:Z=0  
35 A(1)=0:A(2)=0:A(3)=-50  
40 P.“FREEBISH!”:Y=F:A(2)=F  
45 IN.X:GOS.180:W=RND(20):IF W=1 G.225  
50 IF W=2 G.255  
55 IF W=3 G.270  
60 IF W=4 G.285  
65 IF W=5 G.300  
70 IF W=6 G.315  
72 IF W=7 G.330  
74 IF W=8 G.345  
76 IF W=9 G.360  
78 IF W=10 G.375  
80 IF W=11 G.390  
82 IF W=12 G.405  
84 IF W=13 G.420  
86 IF W=14 G.435  
88 IF W=15 G.450
```

90 IF W=16 G.465
92 IF W=17 G.480
94 IF W=18 G.495
96 IF W=19 G.510
100 G.525
160 A(2)=A(2)+Y:Z=0:W=0:A(1)=A(1)-(Y/2)
165 IN.X:G.50
180 IF X=A(3) G.220
185 IF X=Y G.220
190 A(1)=A(1)+X:A(3)=X:F.Z=1TO470:N.Z
195 P.“SCORE”:P.“THE BRAIN”,A(2):P.“THE CLOD”,A(1)
200 IF A(1)>35 G.550
205 IF A(2)>35 G.540
210 P.:P.:F.Z=1TO666:N.Z:Z=0:RET.
220 A(1)=A(1)-10:F.Z=1TO470:N.Z:G.195
225 Z=A:GOS.240:IF Z=50 G.55
230 P.“ALBOOKISH!”:Y=A:G.160
240 IF Z=Y THEN Z=50
245 IF Z=X THEN Z=50
250 RET.
255 Z=B:IF Z=50 G.55
260 Y=B:P.“BREEPI!”:G.160
270 Z=C:GOS.240:IE Z=50 G.55
275 P.“CRIPKLUTS!”:Y=C:G.160
285 Z=D:GOS.240:IF Z=50 G.55
290 P.“DUFFLEEM!”:Y=D:G.160
300 Z=E:GOS.240:IF Z=50 G.55
305 P.“EXQUIMBLE!”:Y=E:G.160
315 Z=F:GOS.240:IF Z=50 G.55
320 P.“FLIBBINK!”:Y=F:G.160
330 Z=G:GOS.240:IF Z=50 G.55
335 P.“GRUNKITT!”:Y=G:G.160
345 Z=H:GOS.240:IF Z=50 G.55
350 P.“HEEJAMBOONKLE!”:Y=H:G.160
360 Z=I:GOS.240:IF Z=50 G.55
365 P.“ISTHRIM!”:Y=I:G.160
375 Z=J:GOS.240:IF Z=50 G.55
380 P.“JACQUELPHLOOM!”:Y=J:G.160
390 Z=K:GOS.240:IF Z=50 G.55
400 P.“KREECK!”:Y=K:G.160
405 Z=L:GOS.240:IF Z=50 G.55
410 P.“LORKE!”:Y=L:G.160
420 Z=M:GOS.240:IF Z=50 G.55
425 P.“MUCSTILE!”:Y=M:G.160
435 Z=N:GOS.240:IF Z=50 G.55
440 P.“NEFLOM!”:Y=N:G.160

```
450 Z=Q:GOS.240:IF Z=50 G.55
455 P.“QUIGGLESBY!”:Y=Q:G.160
465 Z=R:GOS.240:IF Z=50 G.55
470 P.“RECKLEBOP!”:Y=R:G.160
480 Z=S:GOS.240:IF Z=50 G.55
490 P.“SNORK!”:Y=S:G.160
495 Z=T:GOS.240:IF Z=50 G.55
500 P.“THUBBLE!”:Y=T:G.160
510 Z=U:GOS.240:IF Z=50 G.55
520 P.“UBENZZERT!”:Y=U:G.160
525 Z=V:GOS.240:IF Z=50 G.55
530 P.“VELK!”:Y=V:G.160
540 P.“I WIN, ”:GOS.570
545 P.“OF COURSE.”
547 END
550 P.“HEY! ”:GOS.570
555 P.“YOU WON!”:GOS.570
560 P.“THAT WASN’T SUPPOSED TO HAPPEN!”
565 END
570 F.Z=1TO555:N.Z
575 RET.
```

The Rules for Freebish!

Actually the rules for Freebish! are really quite simple. Various letters of the alphabet (except O, W, X and Z, which are used by the computer to keep track of things) are randomly assigned values from -10 to +10. The first letter of any entry determines the amount to be added to or subtracted from your score (any additional letters are ignored; they’re just for show). If you play the same letter twice in a row, or immediately after the computer has played that letter, you lose 10 points regardless of the ordinary point value of that letter. The computer chooses its plays randomly, but it won’t break the repetition rule mentioned above, and its first play is always “FREEBISH!” (the letter F). Whoever manages to rack up a score of more than 35 points wins the game. Even when you know the rules, it’s harder than it looks because the letters have different values each time you play. (NOTE: If two letters have the same point value, they are counted as the same letter for the repetition rule.)

Crops

In this game you are a farmer growing two crops—dreckmelons (which are delicious with rutabagas) and treephules, from which valuable treephule fibers are made. The object is to have a good harvest.

Dreckmelon plants will die if exposed to below 40° temperatures for too long. They can be kept warm by releasing strunkflies, which nest over dreckmelon plants in great droves, thereby keeping them warm. Unfortunately, strunkflies love to eat treephule bushes.

Nellum spiders eat strunkflies. Unfortunately, they also eat the frubees needed to pollenate the treephule bushes.

In case of drought you must open up irrigation ditches. But this will also bring in guntherbugs which destroy both crops (and bite people too). Guntherbugs serve no useful purpose.

To help you control the insect population, three different insecticides are available:

DDS: 13% effective against insects; 2½% effective against plants

RQL: 38% effective against insects; 27% effective against plants

MPN: 86% effective against insects; 58% effective against plants

You must balance the populations of these various insects for an optimum harvest. There is no all-purpose solution to this problem, because day-to-day weather will always have a changing effect. This means the game can be played over and over again, rather than being a repeatable puzzle. See Fig. 1-4 for the flowchart.

Standard BASIC

```
5 PRINT:PRINT" ", "CROPS":PRINT
7 LET A=0:LET B=0:LET C=0
10 REM* CROPS PLANTED *
12 LET D=500:LET T=500
14 REM* INITIAL INSECT POPULATION *
16 LET F=INT(RND(0)*500)+1:LET G=0
18 LET N=INT(RND(0)*500)+1:LET I=0
19 LET S=INT(RND(0)*500)+1:LET E=INT(RND(0)*100)-9
20 REM* DAY'S REPORT *
22 LET C=C+1
24 LET W=INT(RND(0)*10)+1
```

```

26 PRINT"YOU HAVE ";D;" DRECKMELON PLANTS, ";T;
28 PRINT"TREPHEULE BUSHES",I;" IRRIGATION DITCHES"
30 PRINT:PRINT F;" FRUBEES",N;" NELLUM SPIDERS",
32 PRINT S;" STRUNKFLIES",G;" GUNTHERBUGS"
34 PRINT"DAY #";C,"TEMPERATURE ";E;" DEGREES",
36 IF E<40 THEN GOSUB 235
40 IF A>5 THEN GOTO 200
45 IF W>4 THEN GOTO 205
50 LET A=A+1
52 PRINT"DRY DAY":PRINT
54 LET E=E+INT(RND(0)*25)-11
56 LET X=RND(0)*F/1000
58 LET Y=T*X:LET T=T+INT(Y):LET D=D-INT(Y)
60 LET X=INT(RND(0)*S)+1
62 LET Y=INT(RND(0)*10)+5
64 IF B>Y THEN GOSUB 220
66 LET X=X/1500
68 LET T=INT(T-(T*X))
70 LET X=(RND(0)*N)/5
72 LET S=S-INT(X) : LET F=F-INT(X)
74 G=G+(I*INT(RND(0)*100+1))
76 IF G<5 THEN GOTO 85
78 REM* LINES 55 TO 95 ARE CROP & INSECT ADJUST*
80 LET X=(INT(RND(0)*G+1))/20
82 LET D=D-INT(X)
84 LET X=(INT(RND(0)*G)+1)/20
86 LET T=T-INT(X):LET X=A-I
88 IF X>5 THEN GOSUB 260
90 LET X=C/20
92 IF X=INT(X) THEN GOTO 275
93 LET X=D+T
94 IF X<50 THEN GOTO 350
95 IF X>4000 THEN GOTO 370
96 REM* THE PLAY *
97 PRINT"1 TO RELEASE INSECTS. 2 TO SPRAY INSECTICIDE,"
98 LET G=INT(G+(G*.1)):LET F=INT(F+(F*.11))
99 LET S=INT(S+(S*.12)):LET N=INT(N+(N*.15))
100 PRINT"3 TO DIG IRRIGATION DITCH 4 TO CLOSE OLD"
102 PRINT"OLD DITCH OR 0 TO PASS"
104 INPUT L
106 IF L=1 THEN GOTO 300
108 IF L=2 THEN GOTO 125
110 IF L=3 THEN LET I=I+1
115 IF L=4 THEN LET I=I-1
117 IF I<1 THEN LET I=0
120 GOTO 20
125 REM* INSECTICIDE *

```




Fig. 1-4. Crops flowchart.

```

127 PRINT"1 FOR DDS, 2 FOR RQL, OR 3 FOR MPN"
130 INPUT L
132 IF L=1 THEN GOTO 160
135 IF L=2 THEN GOTO 180
137 REM* MPN EFFECTS *
140 LET D=INT(D-(D*.58))
142 LET T=INT(T-T*.58))
144 LET S=INT(S-S*.86))
146 LET N=INT(N-(N*.86))
148 LET F=INT(F-(F*.86))
150 LET G=INT(G-(G*.86))
155 GOTO 20
    
```

```

160 REM* DDS EFFECTS *
162 LET D=D-INT(D*.025)
164 LET T=T-INT(T*.025)
166 LET S=S-INT(S*.13)
168 LET N=N-INT(N*.13)
170 LET F=F-INT(F*.13)
172 LET G=G-INT(G*.13)
175 GOTO 20
180 REM* RQL EFFECTS *
182 LET D=D-INT(D*.27)
184 LET T=T-INT(T*.27)
186 LET S=S-INT(S*.38)
188 LET N=N-INT(N*.38)
190 LET F=F-INT(F*.38)
192 LET G=G-INT(G*.38)
195 GOTO 20
200 REM* RAIN SELECT *
202 IF W<7 THEN GOTO 50
205 PRINT "RAIN"
207 LET A=A-3
210 IF A<1 THEN LET A=0
215 GOTO 54
220 REM* WEATHER EFFECTS *
222 LET Z=INT (X/200)
224 IF D>Z THEN GOTO 230
226 LET D=INT(D-(RND(0)*D/6))
228 RETURN
230 LET D=INT(RND(0)*Z+2*(Z/3))
232 RETURN
235 LET B=B+1
240 IF E<20 THEN LET B=B+.5
245 IF E<10 THEN LET B=B+.75
250 IF E<0 THEN LET B=B+1
255 RETURN
260 REM * DROUGHT DAMAGE *
262 LET J=(INT(RND(0)*100)+1)/100
264 LET K=(INT(RND(0)*100)+1)/100
266 LET D=D-(J*D)
268 LET T=T-(T*K)
270 LET D=INT(D):LET T=INT(T)
272 RETURN
275 REM* CROP GROWTH *
280 LET Y=D*1.5
285 LET D=D+INT(RND(0)*Y+1)
290 LET Y=T*1.5
292 LET T=T+INT(RND(0)*Y-1)
295 RETURN

```

```

300 PRINT"1 FOR STRUNKFLIES, 2 FOR NELLUM SPIDERS, 3"
305 PRINT"FOR FRUBEES, OR 4 FOR GUNTHERBUGS"
307 INPUT X
310 PRINT"HOW MANY ";
312 INPUT Y
315 IF Y<0 THEN LET Y=0-Y
317 LET Y=INT(Y)
320 IF X=4 THEN GOTO 340
322 IF X=1 THEN LET S=S+Y
325 IF X=2 THEN LET N=N+Y
330 IF X=3 THEN LET F=F+Y
335 GOTO 20
340 PRINT"WHY WOULD YOU WANT ";Y;" MORE
GUNTHER BUGS!?"
342 LET G=G+Y
345 GOTO 20
350 REM* LOSE *
352 PRINT"YOU'VE SUCCEEDED IN KILLING OFF MOST OF
YOUR"
355 PRINT"CROPS. YOU'RE DOWN TO ";D;"DRECKMELON
PLANTS"
360 PRINT"AND ";T;" TREPHULE BUSHES"
362 PRINT
365 PRINT"WHAT A LOUSY FARMER!"
367 END
370 PRINT" WOW!"
375 PRINT"YOU HAVE ";D;" DRECKMELON PLANTS AND"
380 PRINT T;" TREPHULE BUSHES"
385 PRINT
390 REM * WIN*
395 PRINT"YOU DESERVE AN AWARD IN AGRICULTURE!"
400 END

```

TRS-80 BASIC

```

5 CLS:P." ","CROPS":P.:A=0
10 B=0:C=0:D=500:F=RND(500):G=0
15 I=0:N=RND(500):S=RND(500):T=500:E=RND(100)-10
20 C=C+1:W=RND(10)
25 P."YOU HAVE ";D;" DRECKMELON PLANTS. ";T;
30 P."TREPHULE BUSHES",I;" IRRIGATION DITCHES":P.
32 P.F.;"FRUBEES ",N;"NELLUM SPIDERS",S;""
STRUNKFLIES",
34 P.G.;"GUNTHERBUGS":P."DAY #";C,"TEMPERATURE"
;E;"DEGREES",
36 IF E<40 GOS.235
40 IF A>5 G.200
45 IF W>4 G.205

```

50 A=A+1:P.“DRY DAY”:P.
 55 E=E+RND(25)-12:X=RND(F)/1000:Y=T*X:T=T+INT(Y)
 :D=D-IN+(Y)
 60 X=RND(S) :Y=RND(10)+4:IF B>Y GOS.220
 65 X=X/1500:T=INT(T-(T*X))
 70 X=RND(N)/5:S=S-INT(X):G=G+(I*RND(100)):F=F-INT(X)
 75 IF G<5 G.85
 80 X=INT(RND(G)/20):D=D-X:X=INT(RND(G)/20):T=T-X
 85 X=A-I:IF X>5 GOS.260
 90 X=C/20:IF X=INT(X) G.275
 92 X=D+T:IF X<50 G.350
 93 IF X>4000 G.370
 95 P.“1 TO RELEASE INSECTS, 2 TO SPRAY INSECTICIDE, 3”
 97 G=INT(G+(G*.1)):F=INT(F+(F*.11)):S=INT(S+(S*.12))
 98 N=N+INT(X*.15)
 100 P.“TO DIG IRRIGATION DITCH, 4 TO CLOSE OLD DITCH,”
 102 IN.“OR 0 TO PASS ”;L
 105 IF L=1 G.300
 110 IF L=2 G.125
 115 IF L=3 THEN I=I+1
 117 IF L=4 THEN I=I-1
 118 IF I<1 THEN I=0
 120 G.20
 125 IN.“1 FOR DDS, 2 FOR RQL OR 3 FOR MPN ”;L
 130 IF L=1 G.160
 135 IF L=2 G.180
 140 D=INT(D-(D*.58)):T=T-INT(T*.58)
 145 S=S-(INT(S*.86)):N=N-INT(N*.86)
 150 F=F-INT(F*.86):G=INT(G-(G*.86))
 155 G.20
 160 D=D-INT(D*.025):T=T-INT(T*.025)
 165 S=S-INT(S*.13):N=N-INT(N*.13)
 170 F=F-INT(F*.13):G=G-INT(G*.13)
 175 G.20
 180 D=D-INT(D*.27):T=T-INT(T*.27)
 185 S=S-INT(S*.38):N=INT(N-(N*.38))
 190 F=F-INT(F*.38):G=G-INT(G*.38)
 195 G.20
 200 IF W<7 G.50
 205 P.“RAIN”:A=A-3
 210 IF A<1 THEN A=0
 215 G.55
 220 Z=INT(X/200):IF D<Z G.230
 225 D=INT(D-RND(D/6)):RET.
 230 D=INT(RND(Z/3)+2*(Z/3)):RET.
 235 B=B+1
 240 IF E<20 THEN B=B+.5

```

245 IF E<10 THEN B=B+.75
250 IF E<0 THEN B=B+1
255 RET.
260 J=RND(100)/100:K=RND(100)/100
265 D=INT(D-(J*D)):T=INT(T-(T*K))
270 RET.
275 D=D+INT(D*1.5):T=T+INT(T*1.5)
280 RET.
300 P.“1 FOR STRUNKFLIES, 2 FOR NELLUM SPIDERS, 3”
305 P.“FOR FRUBEES, OR 4 FOR GUNTHERBUGS”
307 IN.X
310 IN.“HOW MANY”;Y:Y=INT(Y)
315 Y=ABS(Y):IF X=4 G.340
320 IF X=1 THEN S=S+Y
325 IF X=2 THEN N=N+Y
330 IF X=3 THEN F=F+Y
335 G.20
340 P.“WHY WOULD YOU WANT”;Y;“MORE GUNTHERBUGS?!?”
345 G=G+Y:G.20
350 P.“YOU’VE SUCCEEDED IN KILLING OFF MOST OF YOUR”
355 P.“CROPS. YOU’RE DOWN TO”;D;“DRECKMELON PLANTS”
360 P.“AND”;T;“TREPHULE BUSHES”:P.
365 P.“WHAT A LOUSY FARMER!”:END
370 P.“WOW!”
375 P.“YOU HAVE”;D;“DRECKMELON PLANTS AND”;T
380 P.“TREPHULE BUSHES”
385 P.:P.“YOU DESERVE AN AWARD IN AGRICULTURE!”
390 END

```

Summary of Variables Used

- A Number of dry days
- B Number of cold days
- C Day number
- D Number of dreckmelon plants
- E Temperature
- F Number of frubees
- G Number of guntherbugs
- I Number of irrigation ditches
- J Various calculations
- K Various calculations
- N Number of nellum spiders
- S Number of strunkflies
- T Number of trephule bushes
- W Rain or dry
- X Various calculations
- Y Various calculations
- Z Various calculations

H, M, O, P, Q, and R are not used in this program.

Sample Run (Excerpt)

YOU HAVE 500 DRECKMELON PLANTS, 500
TREPHEULE BUSHES, 0 IRRIGATION DITCHES,
397 FRUBEES, 284 NELLUM SPIDERS, 152 STRUNKFLIES,
0 GUNTHERBUGS

DAY #1 TEMPERATURE 70 DEGREES
RAIN

1 TO RELEASE INSECTS, 2 TO SPRAY INSECTICIDE, 3
TO DIG IRRIGATION DITCH, 4 TO CLOSE OLD DITCH,
OR 0 TO PASS ?2

1 FOR DDS, 2 FOR RQL, OR 3 FOR MPN? 2

YOU HAVE 365 DRECKMELON PLANTS, 451
TREPHEULE BUSHES, 0 IRRIGATION DITCHES,
273 FRUBEES, 203 NELLUM SPIDERS, 94 STRUNKFLIES,
0 GUNTHERBUGS

DAY #2 TEMPERATURE 68 DEGREES

DRY DAY

1 TO RELEASE INSECTS, 2 TO SPRAY INSECTICIDE, 3
TO DIG IRRIGATION DITCH, 4 TO CLOSE OLD DITCH,
OR 0 TO PASS ?1

1 FOR STRUNKFLIES, 2 FOR NELLUM SPIDERS, 3
FOR FRUBEES, OR 4 FOR GUNTHERBUGS

?1

HOW MANY? 100

YOU HAVE 332 DRECKMELON PLANTS, 503
TREPHEULE BUSHES, 0 IRRIGATION DITCHES,
303 FRUBEES, 233 NELLUM SPIDERS, 187 STRUNKFLIES
0 GUNTHERBUGS

DAY #3 TEMPERATURE 77 DEGREES

DRY DAY

1 TO RELEASE INSECTS, 2 TO SPRAY INSECTICIDE, 3
TO DIG IRRIGATION DITCH, 4 TO CLOSE OLD DITCH,
OR 0 TO PASS ?3

YOU HAVE 299 DRECKMELON PLANTS, 567
TREPHEULE BUSHES, 1 IRRIGATION DITCHES,
296 FRUBEES, 267 NELLUM SPIDERS, 169 STRUNKFLIES
0 GUNTHERBUGS

DAY #4 TEMPERATURE 69 DEGREES

RAIN

1 TO RELEASE INSECTS, 2 TO SPRAY INSECTICIDE, 3
TO DIG IRRIGATION DITCH, 4 TO CLOSE OLD DITCH,
OR 0 TO PASS ?2

1 FOR DDS, 2 FOR RQL, OR 3 FOR MPN? 3

YOU HAVE 91 DRECKMELON PLANTS, 257
TREPHEULE BUSHES, 1 IRRIGATION DITCHES,
36 FRUBEES, 38 NELLUM SPIDERS, 19 STRUNKFLIES,
14 GUNTHERBUGS

DAY #5 TEMPERATURE 61 DEGREES

RAIN

There's Gold In Them There Skyscrapers

The directions for playing this game are included in the program listing, steps 450 to 520. You may omit these steps from the program along with steps 15, 20, and 525 to 575 if you want to save program space.

The building layout is included in the array. A(1) to A(100) are set to 0, 2, 3, or 4. Obviously each array position stands for a room. If a given array position is equal to 2, that's the room holding the gold. This can be any room from #2 to #100 (step 145). Room #1 is the room the game starts in, but it would be a very short game if the gold was found there.

If the array position is equal to 3 that room has an exit from the building. Three of the rooms from 1 to 10 have an exit. These are designated as the ground floor since an exit on the 8th floor wouldn't be too good an idea.

Any of the upper rooms (from 11 to 100) might have a trap door (designated by a 4 in the array position). If you enter one of these rooms, you will fall to the room immediately below it. For example, room 73 would drop you to room 63. Since each array position can hold only one number, the gold won't be in a room with a trap door.

A zero in the array position is just a space holder with no particular meaning.

If the room has the appropriate doors, you can move to the room on the immediate right, left, above or below the room you're in. The doors are set in array positions A(101) to A(500). A(101) to A(200) determine which rooms have doors to the right (room 38 would correspond to A(138). A(201) to A(300) are the doors to the left. A(301) to A(400) are the doors to the rooms above, and A(401) to A(500) are the doors to the rooms below. A 1 in any given array position indicates the presence of that particular door, while a 0 signifies its absence. (NOTE: you can't always get out of a room the way you came in).

In step 30 all the doors are set to 1, then the impossible doors are removed (steps 40 to 60). These are the doors along the perimeter of the building. There is no room to the left of room 41, for instance, so there shouldn't be a door there, either.

Finally up to 250 of the remaining 360 doors are randomly moved. It's theoretically possible for you to find yourself in a room with no doors at all. If this happens you have to hit the 'BREAK' key

to get out of the dead-ended program. I haven't included any protection against this kind of thing because it would take a great deal of program space, and the odds are the problem will never come up anyway. It's possible, but unlikely.

Your moves are limited to Right, Left, Up, Down, or Exit (you can use just the first letters or type out the entire word). Any other entry is an invalid move, but the number of moves will still be incremented by 1, worsening your score. If you try to use a door that isn't there, you'll receive the error message from step 355, and it will be treated as an invalid move.

The only way to end the game is to either BREAK the program, or leave the building via one of the exits either with the gold (you win) or empty-handed (you lose). After playing a few games you can determine for yourself what kind of score to consider "par." I've seen scores of 150 add up in this game, and once I lucked out and left with the gold in a mere 5 moves.

You can alter the odds by changing step 75. This step begins with the command Y=RND(250). Y is the number of doors to be randomly removed (the number might be reduced if the computer decides to remove certain doors more than once). If you want an easier game you can change this command to Y=RND(100). Or you might try Y=RND(150)+150. There could be up to 300 doors removed (out of 360!), and at least 150 will be removed (except where the computer gets redundant and gives you a break). If you alter this command, be sure to re-enter the rest of step 75 as it is written in the original program. See Fig. 1-5 for the flowchart.

Standard BASIC

```
5 CLS:PRINT:FOR X=1 TO 100: LET A(X)=0:NEXT X
10 PRINT" GOLD IN THEM THERE SKYSCRAPERS"
15 INPUT"ENTER 1 FOR INSTRUCTIONS OR 2 FOR GAME ";X
20 IF X=1 GOTO 450
25 REM ** PUT DOORS IN ALL ROOMS
30 FOR X=101 TO 500:LET Z(X)=1:NEXT X
35 REM ** REMOVE IMPOSSIBLE DOORS
40 FOR X=391 TO 410: LET A(X)=0:NEXT X
45 LET Y=201:FOR X= 1 TO 10
50 LET A(Y)=0:LET Y=Y+10:NEXT X
60 LET Y=110:FOR X=1TO 10
65 LET A(Y)=0:LET Y=Y+10:NEXT X
70 REM ** REMOVE RANDOM DOORS
75 LET Y=INT(RND(1)*250)+1
80 FOR X= 1 TO Y:LET Z=INT(RND(1)*400)+101
90 LET A(Z)=0:NEXT X
```

```

100 REM ** PLANT TRAP DOORS
105 LET Y=INT(RND(1)*20)+1
110 FOR X=1 TO Y:LET Z=INT(RND(1)*90)+11
120 LET A(Z)=4:NEXT X
125 REM ** PLANT EXITS & GOLD

```




Fig. 1-5. Flowchart for There's Gold in Them There Skyscrapers.

```
130 FOR X=1 TO 3:LET Y=INT(RND(1)*10)+1
135 LET A(Y)=3:NEXT X
140 LET X=INT(RND(1)*99)+2:LET A(X)=2
150 REM ** SET GAME VARIABLES
155 LET M=1: LET P=1:LET G=0
160 LET E=20:LET U=10:LET D=-10
170 LET R=1:LET L=-1
200 PRINT"MOVE #";M:LET M=M+1
205 PRINT"YOU ARE IN ROOM #";P
210 LET X=A(P)
215 IF X=2 GOSUB 400
220 IF X=3 PRINT"THESE IS AN EXIT."
225 IF X=4 GOTO 420
230 PRINT"YOU CAN MOVE IN THE FOLLOWING DIRECTIONS
---"
235 LET Y=P+100:IF A(Y)=1PRINT"RIGHT ";
240 LET Y=P+200:IF A(Y)=1 PRINT"LEFT ";
245 LET Y=P+300:IF A(Y)=1 PRINT"UP ";
250 LET Y=P+400:IF A(Y)=1 PRINT"DOWN ";
255 PRINT:PRINT
260 INPUT"YOUR MOVE";Q
265 IF Q=E GOTO 300
270 IF Q=R GOTO 340
275 IF Q=L GOTO 370
280 IF Q=U GOTO 380
285 IF Q=D GOTO 390
290 PRINT "INVALID MOVE!"
295 GOTO 200
300 IF A(P)=E GOTO 310
305 GOTO 290
310 PRINT "YOU HAVE JUST LEFT THE BUILDING WITH";
315 IF G=0 PRINT"OUT";
320 PRINT "THE GOLD."
325 LET M=M-1
330 PRINT "IT TOOK YOU ";M;" MOVES."
335 END
340 REM ** MOVE = RIGHT
345 LET X=P+100
350 IF A(X)=1 GOTO 365
355 PRINT"YOU JUST RAN INTO A WALL, CLOD!"
360 GOTO 200
365 LET P=P+Q:GOTO 200
370 REM ** MOVE = LEFT
375 LET X=P+200:GOTO 350
380 REM ** MOVE = UP
385 LET X=P+300:GOTO 350
390 REM ** MOVE = DOWN
```

```

395 LET X=P+400:GOTO 350
400 PRINT "YOU JUST FOUND THE GOLD!"
405 LET A(P)=0: LET G=1
410 RETURN
420 PRINT "TRAP DOOR!"
425 LET P=P-10
430 FOR X=1 TO 333: NEXT X
440 GOTO 200

```

Secret Passageway Option (Standard BASIC)

```

122 LET Y=INT(RND(1)*40)+1:FOR X=1 TO Y
123 LET Z=INT(RND(1)*100)+1
124 LET A(Z)=5:NEXT X:LET S=5
227 IF X=5 PRINT "THERE IS A SECRET PASSAGEWAY"
287 IF Q=S GOTO 600
600 IF A(P)=5 GOTO 620
610 GOTO 290
620CLS:LET P=INT(RND(1)*100)+1
630 FOR X=1 TO 456: NEXT X
640 GOTO 200

```

TRS-80 BASIC

```

10 CLS:P. :P." ", "GOLD IN THEM THERE SKYSCRAPERS"
15 F.X=101TO500:A(X)=1:N.X:F.X=1TO100:A(X)=0.N.X
20 F.X=391TO410:A(X)=0:N.X:Y=201:F.X=TO10
25 A(Y)=0:Y=Y+10:N.X:Y=110:F.X=1TO10
30 A(Y)=0:Y=Y+10:N.X
40 Y=RND(250):F.X=1TOY:Z=RND(400)+100
45 A(Z)=0:N.X:M=1:P=1:G=0:E=20
50 Y=RND(20):F.X=1TOY:Z=RND(90)+10:A(Z)=4:N.X
55 U=110:D=-10:R=1:L=-1:F.X=1TO3:Y=RND(10)
60 A(Y)=3:N.X:X=RND(99)+1:A(X)=2
100 P."MOVE #";M:M=M+1:P."YOU ARE IN ROOM #";P
105 X=A(P):IF X=2 GOS.250
110 IF X=3 P."THERE IS AN EXIT."
115 IF X=4 G.260
120 P."YOU CAN MOVE IN THE FOLLOWING DIRECTIONS --"
125 Y=P+100:IF A(Y)=1 P."RIGHT ";
1300 Y=P+200:IF A(Y)=1 P."LEFT ";
135 Y=P+300:IF A(Y)=1 P."UP ";
140 Y=P+400:IF A(Y)=1 P."DOWN ";
145 P.:P.:IN."YOUR MOVE";Q:IF Q=E G.180
150 IF Q=R G.210
155 IF Q=L G.230
160 IF Q=U G.235

```

```

165 IF Q=D G.240
170 P."INVALID MOVE!":G.100
180 IF A(P)=E G.190
185 G.170
190 P."YOU HAVE JUST LEFT THE BUILDING WITH";
195 IF G=0 P."OUT";
200 P."THE GOLD."M=M-1:P."IT TOOK YOU ";M;" MOVES."
205 END
210 X=P+100
215 IF A(X)=1 G.225
220 P."YOU JUST RAN INTO A WALL, CLOD!":G.100
225 P=P+Q:G.100
230 X=P+200:G.215
235 X=P+300:G.215
240 X=P+400:G.215
250 P."YOU FOUND THE GOLD!":A(P)=0:G=1:RET.
260 P=P-10:P."TRAP DOOR!":F.X=1TO 333:N.X:G.100

```

Secret Passageway Option (TRS-80)

```

35 Y=RND(40):F.X=1TOY:Z=RND(100):A(Z)=5:N.X:S=5
117 IF X=S GOS.300
167 IF Q=S G.310
300 P."THERE IS A SECRET PASSAGEWAY";RET.
310 IF A(P)=5G,620
315 G.170
320 CLS:P=RND(100):F.X=1TO456:N.X
330 G.100

```

Summary Of Variables Used

D Move down. D = -10
 E Exit. E=20
 G Gold located?
 L Move left. L = -1
 M Move #
 P Current room
 Q Current move
 R Move right.R = 1
 U Move up. U = 10
 X Set-up & timing variable
 Y "
 Z "

Sample Run (Excerpt)

GOLD IN THEM THERE SKYSCRAPERS

MOVE #1
 YOU ARE IN ROOM #1

YOU CAN MOVE IN THE FOLLOWING DIRECTIONS ---
RIGHT UP

YOUR MOVE? RIGHT

MOVE #2

YOU ARE IN ROOM #2

THERE IS AN EXIT

YOU CAN MOVE IN THE FOLLOWING DIRECTIONS ---

LEFT UP

YOUR MOVE? UP

MOVE #3

YOU ARE IN ROOM #12

YOU CAN MOVE IN THE FOLLOWING DIRECTIONS ---

RIGHT LEFT UP DOWN

YOUR MOVE? RIGHT

MOVE #4

YOU ARE IN ROOM #13

YOU CAN MOVE IN THE FOLLOWING DIRECTIONS ---

RIGHT LEFT DOWN

YOUR MOVE? RIGHT

MOVE #5

YOU ARE IN ROOM #14

TRAP DOOR!

MOVE #6

YOU ARE IN ROOM #4

THERE IS AN EXIT

YOU CAN MOVE IN THE FOLLOWING DIRECTIONS ---

RIGHT LEFT UP

YOUR MOVE? EXIT

YOU HAVE JUST LEFT THE BUILDING WITHOUT THE GOLD
IT TOOK YOU 6 MOVES

Lost In The Jungle

This is an adventure game of the type increasingly popular among computer hobbyists. The player has an objective (getting out of the jungle alive) and the computer offers various obstacles to the goal (wolves, lions, black widow spiders, waterholes, quicksand, poisonous berries). The player is then offered a series of options to select from. The computer analyzes the results of each choice, usually with an element of weighted randomness. On each move the player moves one mile in any of the four basic directions (north, south, east, or west).

An added trick to this game is the main scoring variables (distance to the outside of the jungle, and the player's remaining strength). If you're not careful you may suddenly find yourself dropping from exhaustion. You can build up your level of strength by eating, but watch out! Some of the food you'll come across might turn out to be tainted.

If you get close to the edge of the jungle you'll be told you can see light through the trees ahead.

Unlike many adventure games, the obstacles are randomly placed and the solutions are randomly weighted to keep the game interesting even after you've played it a number of times. See Fig. 1-6 for the flowchart.

Standard Basic

```
2 LET A=11:LET B=12
4 LET C=13:LET D=14
6 LET E=25:LET M=20
8 LET H=12:LET K=100
10 LET F=INT(RND(0)*50)+21
18 PRINT:PRINT
20 PRINT",","LOST IN THE JUNGLE":PRINT
22 OGSUB 500
25 PRINT"YOU ARE LOST IN A JUNGLE. YOU HAVE A GUN
WITH 6"
30 PRINT"BULLETS, A KNIFE, & A SLINGSHOT. THE TRICK IS
TO"
35 PRINT"FIND YOUR WAY BACK TO CIVILIZATION WITHOUT
GETTING"
40 PRINT"YOURSELF KILLED. EACH MOVE IS ONE MILE. THE
JUNGLE"
```

```
45 PRINT"IS 100 MILES SQUARE. PRESS 'ENTER' TO START"
50 INPUT A$
55 PRINT"DO YOU WANT AN EASY, MEDIUM OR HARD
      GAME";
60 INPUT J
62 LET H=INT(RND(0)*50)+21
64 LET G=100-F:LET I=100-H:LET M=6
65 REM* BEGIN GAME*
70 GOSUB 1000
72 LET Z=X:LET S=J-4
74 REM* EVENT SELECTION*
76 LET Y=INT(RND(0)*S)+1
78 GOSUB 500
80 IF Y=1 THEN GOTO 150
82 IF Y<4 THEN GOTO 225
85 IF Y=4 THEN GOTO 520
90 IF Y=6 THEN GOTO 520
95 IF Y=7 THEN GOTO 105
100 GOTO 70
105 PRINT"YOU COME TO A BUSH OF BERRIES"
107 LET Y=17:LET N=18
110 PRINT"DO YOU EAT THEM?"; 
112 INPUT Q
115 IF Q=17 THEN GOTO 122
117 REM* NOT EAT BERRIES*
120 GOTO 70
122 REM* EAT BERRIES*
124 LET K=K+(J/10)
126 LET L=INT(RND(0)*J)+1
128 IF L<(J*.75) THEN GOTO 140
130 PRINT"THEY MAKE YOU QUITE ILL"
132 LET K=K-INT(RND(0)*K)
135 GOTO 70
140 REM* BERRIES OK*
142 LET K=K+(J/10)
145 GOTO 70
150 PRINT"YOU STEP INTO QUICKSAND!"
152 GOSUB 500
154 PRINT"WHAT DO YOU DO?"
156 PRINT A — TRY TO CLAW YOUR WAY OUT"
158 PRINT"B — GIVE UP"
160 PRINT"C — GRAB ONTO A TREE BRANCH"
162 PRINT"D — START PRAYING"
165 INPUT X
170 IF X=C THEN GOTO 200
180 PRINT"SORRY      ";
```





Fig. 1-6. Flowchart for Lost in the Jungle.







Fig. 1-6. Flowchart for Lost in the Jungle.



```
185 GOSUB 500
190 PRINT"YOU'RE DECEASED."
195 END
200 REM* TREE BRANCH ATTEMPT*
202 LET L=J*10
204 LET Q=INT(RND(0)*L)+1
206 IF Q>100 THEN GOTO 215
208 PRINT"YOU DIDN'T MAKE IT"
210 GOSUB 500
212 GOTO 190
215 PRINT"YOU MADE IT!"
217 LET K=K-INT(RND(0)*K*.67)
220 GOTO 70
225 REM* ANIMAL ATTACK*
227 IF Y=2 THEN PRINT"WOLF";
230 IF Y=3 THEN PRINT" LION";
232 PRINT"AHEAD. WHAT DO YOU DO?"
234 PRINT"A — FLEE"
236 PRINT"B — SHOOT IT"
238 PRINT"C — USE YOUR SLINGSHOT"
240 PRINT"D — USE YOUR KNIFE"
242 PRINT"E — CLIMB A TREE"
245 INPUT X
250 GOSUB 500
252 IF X=A THEN GOTO 285
255 IF X=B THEN GOTO 330
260 IF X=C THEN GOTO 350
265 IF X=D THEN GOTO 355
270 IF X=E THEN GOTO 390
272 REM*NO VALID CHOICE*
275 LET K=K-1
277 PRINT"IT'S STILL THERE. NOW WHAT?"
280 GOTO 234
285 GOSUB 1000
287 REM* FLEE*
290 IF Z=X THEN GOTO 760
292 LET Q=INT(RND(0)*J)+1
294 LET K=K-INT(RND(0)*K/2)
296 IF Q<9 THEN GOTO 315
298 PRINT"WHEW!"
300 GOSUB 500
305 PRINT"YOU MADE IT TO SAFETY!"
310 GOTO 70
315 PRINT"IT CATCHES YOU!"
317 LET K=K-INT(RND(0)*K)+1
318 IF K<10 THEN GOTO 185
```

```
320 LET J=J-1
322 PRINT“NOW WHAT DO YOU DO?”:GOTO 234
325 IF M<1 THEN GOTO 615
327 GOSUB 500
330 PRINT:PRINT“ ”,“BANG!!!”:PRINT
332 LET L=J*2.5:LET M=M-1
334 GOSUB 500
336 LET Q=INT(RND(0)*L)+1
338 IF Q>17 THEN GOTO 440
340 PRINT“YOU MISSED!”
345 GOTO 315
350 LET L=J*1.5
352 GOTO 334
355 LET L=J*1.7
357 IF Y=3 THEN LET L=L*2
360 LET Q=INT(RND(0)*L)+1
365 IF Q>18 THEN GOTO 440
370 PRINT“IT OVERPOWERS YOU!”
372 LET K=K-(2*Q)
375 IF K<15 THEN GOTO 185
380 GOTO 320
390 IF Y=3 THEN GOTO 430
392 LET Q=INT(RND(0)*40)+1
395 GOSUB 500
397 LET K=K-Q
400 PRINT“THE WOLF DOES NOT LEAVE FOR”;Q;“ HOURS”
405 IF K<7 THEN GOTO 420
410 PRINT“THEN YOU CAN CLIMB DOWN”
415 GOTO 70
420 PRINT“YOU DIE OF THIRST AND STARVATION”
425 END
430 PRINT“LIONS CAN CLIMB TREES BETTER THAN PEOPLE.”
435 GOTO 185
440 PRINT“GOT ‘EM!”
442 GOSUB 500
445 PRINT“DO YOU EAT IT?”
447 PRINT“A — YES”
450 PRINT“B — NO”
452 INPUT X
455 IF X=A THEN LET K=K+J
460 GOTO 70
470 PRINT“YOU DROP FROM EXHAUSTION”
475 GOTO 185
500 LET W=INT(RND(0)*888)+1
505 FOR V=1 TO W
507 NEXT V
```

```
510 RETURN
520 PRINT“A BLACK WIDOW SPIDER LANDS ON YOUR NECK!”
522 PRINT“WHAT DO YOU DO?”
525 PRINT“A — HOLD AS STILL AS POSSIBLE”
527 PRINT“B — TRY TO BRUSH IT OFF”
530 PRINT“C — SHOOT IT”
532 PRINT“D — CRY”
535 INPUT X
537 GOSUB 500
540 IF X=A THEN GOTO 580
542 IF X=B THEN GOTO 580
545 IF X=C THEN GOTO 570
550 GOTO 180
570 PRINT:PRINT“, “,“BANG!”:PRINT
573 GOSUB 500
575 GOTO 180
580 LET L=J*4
582 LET Q=RND(0)*L)+1
584 IF Q<J*2 THEN GOTO 600
586 PRINT“IT BITES!”
588 GOSUB 500
590 GOTO 185
600 PRINT“IT CRAWLS OFF WITHOUT BITING”
605 GOTO 70
615 PRINT“YOU ARE OUT OF BULLETS”
620 GOTO 370
630 PRINT“YOU COME TO A FRESH WATER POND”
632 PRINT“WHAT DO YOU DO?”
634 PRINT“A — DRINK”
636 PRINT“B — SWIM”
638 PRINT“C — LEAVE”
640 INPUT X
642 GOSUB 500
645 IF X=A THEN GOTO 670
650 IF X=B THEN GOTO 690
655 IF X=C THEN GOTO 735
660 GOTO 640
670 LET Q=INT(RND(0)*Q)+1
672 IF Q<6 THEN GOTO 680
675 LET K=K+(RND(0)*K)-9
677 GOTO 70
680 PRINT“IT MAKES YOU QUITE ILL”
682 LET K=K-INT(RND(0)*K)
685 GOTO 70
690 LET Q=INT(RND(0)*4)+1
692 GOSUB 500
```

695 LET K=K-2
697 IF Q=1 THEN GOTO 710
700 IF Q=2 THEN GOTO 715
702 GOSUB 500
705 GOTO 70
710 PRINT“YOU DROWN!”
712 GOTO 185
715 PRINT“A PIRANHA STRIKES!”
717 GOSUB 500
720 LET Q=INT(RND(0)*J)+1
722 IF Q<20 THEN GOTO 185
725 PRINT“YOU MANAGE TO ESCAPE WITHOUT SERIOUS IN-
JURY”
730 LET K=K-INT(RND(0)*K)
732 GOTO 70
735 GOSUB 1000
737 IF Z=X THEN GOTO 750
740 GOTO 70
750 PRINT:PRINT“ ”,“SPLASH!”:PRINT
755 GOTO 690
760 PRINT“YOU JUST COLLIDED WITH THE”;
765 IF Y=2 THEN PRINT“WOLF!”
770 IF Y=3 THEN PRINT“LION!”
775 GOSUB 500
780 GOTO 185
1000 LET K=K-.5
1002 PRINT“WHICH WAY DO YOU GO?”
1004 PRINT“A — EAST”
1006 PRINT“B — WEST”
1008 PRINT“C — NORTH”
1010 PRINT“D — SOUTH”
1012 INPUT X
1015 IF X=A THEN GOTO 1050
1017 IF X=B THEN GOTO 1055
1020 IF X=C THEN GOTO 1060
1025 IF X=D THEN GOTO 1065
1030 GOTO 1000
1050 LET F=F-1:LET G=G+1
1052 GOTO 1070
1055 LET F=F+1:LET G=G-1
1057 GOTO 1070
1060 LET H=H-1:LET I=I+1
1062 GOTO 1070
1065 LET H=H+1:LET I=I-1
1070 IF F=0 THEN GOTO 1120
1075 IF G=0 THEN GOTO 1120

```
1080 IF H=0 THEN GOTO 1120
1085 IF I=0 THEN GOTO 1120
1090 IF K<5 THEN GOTO 470
1095 IF F<10 THEN GOTO 1130
1100 IF G<10 THEN GOTO 1130
1105 IF H<10 THEN GOTO 1130
1110 IF I<10 THEN GOTO 1130
1115 RETURN
1120 PRINT"YOU MADE IT SAFELY BACK TO CIVILIZATION!"
1125 END
1130 PRINT"YOU CAN SEE SUNLIGHT THROUGH THE
TREETOPS!"
1135 RETURN
```

TRS-80 BASIC

```
5 A=11:B=12:C=13:D=14:E=25:P.:P.
10 P." ", "LOST IN THE JUNGLE":P.:GOS.500
15 P."YOU ARE LOST IN A JUNGLE. YOU HAVE A GUN WITH 6"
20 P."BULLETS, A KNIFE, & A SLINGSHOT THE TRICK IS TO"
25 P."FIND YOUR WAY BACK TO CIVILIZATION WITHOUT
GETTING"
30 P."YOURSELF KILLED. EACH MOVE IS ONE MILE. THE
JUNGLE"
35 P."IS 100 MILES SQUARE. PRESS 'ENTER' TO START"
40 IN.A$:M=20:H=12
45 IN."DO YOU WANT AN EASY, MEDIUM OR HARD GAME";J
50 K=100:F=RND(50)+20:H=RND(50)+20:I=100-H:
M=6:G=100-F
70 GOS.1000:Z=X:S=J-4
75 Y=RND(S):GOS.500:IF Y=1 G.150
80 IF Y<4 G.225
85 IF Y=4 G.520
90 IF Y=6 THEN GOTO 630
95 IF Y=7 G.105
100 G.70
105 P."YOU COME TO A BUSH OF BERRIES":Y=17:N=18
110 IN."DO YOU EAT THEM";Q:IF Q=17 G.120
115 B.70
120 K=K+(J/10):L=RND(J):IF L<(J*.75)G.140
130 P."THEY MAKE YOU QUITE ILL":K=K-RND(K)
135 G.70
140 K=K+(J/10):G.70
150 P."YOU STEP INTO QUICKSAND!":GOS.500
155 P."WHAT DO YOU DO?":P."A— TRY TO CLAW YOUR WAY
OUT"
160 P."B— GIVE UP":P."C— GRAB ONTO A TREE BRANCH"
165 P."D— START PRAYING":IN.X
```

170 IF X=C G.200
180 P."SORRY ";:GOS.500
190 P."YOU'RE DECEASED."
195 END
200 L=J*10:Q=RND(L):IF Q>100G.215
205 P."YOU DIDN'T MAKE IT."
210 GOS.500:G.190
215 P."YOU MADE IT!":K=K-INT(RND(K)*.67)
220 G.70
225 IF Y=2P."WOLF";
230 IF Y=3P."LION";
232 P."AHEAD!":P."WHAT DO YOU DO?"
235 P."A— FLEE":P."B— SHOOT IT":P."C— USE YOUR
SLINGSHOT"
240 P."D— USE YOUR KNIFE":P."E— CLIMB A TREE"
245 IN.X
250 GOS.500:IF X=A G.285
255 IF X=B G.330
260 IF X=C G.350
265 IF X=D G.355
270 IF X=E G.390
275 K=K-1:P."IT'S STILL THERE. NOW WHAT?"
280 G.235
285 GOS.1000:IF Z=X G.760
290 Q=RND(J(:K=K-INT(RND(K)/2)
295 IF Q<9 G.315
300 P."WHEW!":GOS.500:P."YOU MADE IT TO SAFETY!"
305 B.70
315 P."IT CATCHES YOU!":K=K-RND(K)
317 IF K<10 G.185
320 J=J-1:P."NOW WHAT DO YOU DO?":G.235
325 IF M< G.615
330 GOS.500:P.:P." ","BANG!!":P.:L=J*2.5:M=M-1
335 GOS.500:Q=RND(L):IF Q<17 G.440
340 P."YOU MISSED!"
345 G.315
350 L=J*1.5:G.335
355 L=J*1.7:IF Y=3 THEN L=L*2
360 Q=RND(L):IF Q>18 G.440
370 P."IT OVERPOWERS YOU!!":K=K-(2*Q):IF K<15 G.185
380 G.320
390 IF Y=3 G.430
395 Q=RND(4):GOS.500:K=K-Q
400 P."THE WOLF DOES NOT LEAVE FOR ";Q:" HOURS."
405 IF K<7 G.420
410 P."THEN YOU CAN CLIMB DOWN":G.70

420 P."YOU DIE OF THIRST AND STARVATION."
425 END
430 P."LIONS CAN CLIMB TREES BETTER THAN PEOPLE."
435 G.185
440 P."GOT'EM!":GOS.500:P."DO YOU EAT IT?"
450 P."A— YES":P."B— NO":IN.X
455 IF X=A THEN K=K+J
460 G.70
470 P."YOU DROP FROM EXHAUSTION":G.185
500 W=RND(888)
505 F.V=1TOW:N.V
510 RET.
520 P."A BLACK WIDOW SPIDER LANDS ON YOUR NECK!"
525 P."WHAT DO YOU DO?":P."A— HOLD AS STILL AS
POSSIBLE"
530 P."B— TRY TO BRUSH IT OFF":P."C— SHOOT IT"
535 P."D— CRY":IN.X:GOS.500:IF X=A G.580
540 IF X=B G.580
545 IF X=C G.570
550 G.180
570 P.:P."","BANG!":P.:GOS.500:G.180
580 L=J*4:Q=RND(L):IF Q<J G.600
585 P."IT BITES!":GOS.500
590 G.180
600 P."IT CRAWLS OFF WITHOUT BITING":G.70
615 P."YOU ARE OUT OF BULLETS!":G.370
630 P."YOU COME TO A FRESH WATER POND":P."WHAT DO
YOU DO?"
635 P."A— DRINK":P."B— SWIM":P."C— LEAVE"
640 IN.X
645 GOS.500:IF X=A G.670
650 IF X=B G.690
660 IF X=C G.735
665 G.640
670 Q=RND(J):IF Q<6 G.680
675 K=K+(RND(K)-10):G.70
680 P."IT MAKES YOU QUITE ILL"
685 K=K-RND(K):G.70
690 Q=RND(4):GOS.500:K=K-2:IF Q=1 G.710
700 IF Q=2 G.715
705 GOS.500:G.70
710 P."YOU DROWN!":GOS.500:G.190
715 P."A PIRANHA STRIKES!":GOS.500:Q=RND(J)
720 IF Q<20 G.185
725 P."YOU MANAGE TO ESCAPE WITHOUT SERIOUS INJURY"
730 K=K-RND(K):G.70

735 GOS.1000:IF Z=X G.750
740 G.70
750 P.:P. " ", "SPLASH!":P.G.690
760 P."YOU JUST COLLIDED WITH THE ";
765 IF Y=2 P."WOLF!"
770 IF Y=3 P."LION!"
775 GOS.500:G.185
1000 K=K-.5:P."WHICH WAY WILL YOU GO?"
1005 P."A— EAST":P."B— WEST":P."C— NORTH"
1010 P."D— SOUTH":IN.X:IF X=A G.1050
1015 IF X=B G.1055
1020 IF X=C G.1060
1025 IF X=D G.1065
1030 G.1000
1050 F=F-1:G=G+1:G.1070
1055 F=F+1:G=G-1:G.1070
1060 H=H-1:I=I+1:G.1070
1065 H=H+1:I=I-1
1070 IF F=0 G.1120
1075 IF G=0 G.1120
1080 IF H=0 G.1120
1085 IF I=0 G.1120
1090 IF K<5 G.470
1095 IF F<10 G.1130
1100 IF G<10 G.1130
1105 IF H<10 G.1130
1110 IF I<10 G.1130
1115 RET.
1120 P."YOU MADE IT SAFELY BACK TO CIVILIZATION!"
1125 END
1130 P."YOU CAN SEE SUNLIGHT THROUGH THE TREETOPS!"
1140 RET.

Summary Of Variables Used

- A OPTION SELECTOR
- B OPTION SELECTOR
- C OPTION SELECTOR
- D OPTION SELECTOR
- E OPTION SELECTOR/EASY GAME
- F DISTANCE EAST
- G DISTANCE WEST
- H DISTANCE NORTH/HARD GAME
- I DISTANCE SOUTH
- J GAME DIFFICULTY
- K STRENGTH

L CHANCE SELECTOR
M MEDIUM GAME/# OF BULLETS LEFT.
Q CHANCE SELECTOR
S CHANCE SELECTOR
W TIMING
X SELECTED OPTION
Y CHANCE SELECTOR

Sample Run (Excerpt)

DO YOU WANT AN EASY, MEDIUM OR HARD GAME? EASY
WHICH WAY WILL YOU GO?

A — EAST
B — WEST
C — NORTH
D — SOUTH
?C

A BLACK WIDOW SPIDER LANDS ON YOUR NECK!
WHAT DO YOU DO?

A — HOLD AS STILL AS POSSIBLE
B — TRY TO BRUSH IT OFF
C — SHOOT IT
D — CRY
?A

IT CRAWLS OFF WITHOUT BITING
WHICH WAY WILL YOU GO?

A — EAST
B — WEST
C — NORTH
D — SOUTH
?A

LION AHEAD!
WHAT DO YOU DO?
A — FLEE
B — SHOOT IT
C — USE YOUR SLINGSHOT
D — USE YOUR KNIFE
E — CLIMB A TREE
?B

BANG!

YOU MISSED
IT CATCHES YOU
NOW WHAT DO YOU DO?
A — FLEE
B — SHOOT IT
C — USE YOUR SLINGSHOT

D — USE YOUR KNIFE

E — CLIMB A TREE

?A

WHICH WAY WILL YOU GO?

A — EAST

B — WEST

C — NORTH

D — SOUTH

?D

WHEW!

YOU MADE IT TO SAFETY!

WHICH WAY WILL YOU GO?

A — EAST

B — WEST

C — NORTH

D — SOUTH

?A


Galactic Search

Galactic Search is another adventure game. While random chance plays a part, there's more opportunity for planning strategy in this game. The instructions are given within the program. See Fig. 1-7 for the flowchart.

Standard BASIC

```
5 FOR X=1TO200:PRINT
10 LET Y=INT(RND(0)*26)+1
12 LET A(X)=Y:NEXT X
15 PRINT"YOU ARE ";
17 LET A=INT(RND(0)*5)+1
18 FOR X=501TO600:LET Y=INT(RND(0)*1000)+1
20 IF A=1 THEN LET A$="AKRANAID"
22 LET A(X)=Y: NEXT X
25 IF A=2 THEN LET A$="OLK"
30 IF A=3 THEN LET A$="ZOLAR ZINNK"
35 IF A=4 THEN LET A$="GUAMBNA"
40 IF A=5 THEN LET A$="JOHN DOE"
45 PRINT A$;"—CAPTAIN OF THE SPACESHIP
50 PRINT"‘EVENING STAR’ FROM THE PLANET RAMSNEID"
55 LET J=0:PRINT"RAMSNEID CIRCLES THE STAR KNOWN AS
FR-972"
60 PRINT"AND IS IN DIRE NEED OF THE RARE
SUBSTANCE,BRETCHENKOEL"
65 PRINT"THEIR IS NO BRETCHENKOEL LEFT IN THE SOLAR
SYSTEM"
70 PRINT"AND WITHOUT THIS SUBSTANCE TO FUEL THE
PLANET'S"
75 PRINT"POWER STATIONS, YOUR PEOPLE WILL DIE OUT."
80 PRINT"THIS IS WHY THE RAMSNEIDIANS HAVE TAKEN UP
SPACE"
85 PRINT"TRAVEL, DESPITE ITS STRANGE & UNPREDICTABLE
DANGERS"
87 FOR X=201 TO 300
90 LET Y=INT(RND(0)*999)+1
92 LET A(X)=Y:NEXT X
94 PRINT"YOUR MISSION, CAPTAIN "A$:", IS TO SEARCH
THE"
96 PRINT"GALAXY FOR MORE BRETCHENKOEL. YOU WILL
BE"
100 PRINT"FLYING BY GUESS WORK MOST OF THE TIME"
102 FOR X=301 TO 400
```

Fig. 1-7. Galactic Search flowchart.



```
104 LET Y=INT(RND(0)*17)+1
106 IF Y=17 THEN Y=0
108 IF Y=1 THEN Y=0
110 LET A(X)=Y:NEXT X
115 GOSUB 1230
117 PRINT“BUT WATCH OUT FOR METEORS, AND THE
     GREATLY FEARED”
120 PRINT“WAR-LOVING THYLTHRENS!”
122 FOR X=401TO500
124 LET Y=INT(RND(0)*10)+1
126 LET A(X)=Y:NEXT X
128 LET N=15
130 PRINT“ARE YOU PREPARED TO UNDERTAKE THIS
     MISSION”;
132 INPUT X
135 IF X=Y GOTO 180
140 IF X=N GOTO 160
145 PRINT“YOU MUST ANSWER YES OR NO,”;A$
150 GOTO 130
160 PRINT“YOU ARE A GREAT SHAME TO YOUR RACE,”;A$;“!”
162 PRINT:GOSUB 2000
164 PRINT“, “GOOD-BYE!”
166 END
180 PRINT:PRINT
182 LET Z=1:LET T=1
185 LFT G=30000:LET J=0
195 REM* THE PLAY *
200 PRINT“WARP DRIVE”;
202 INPUT X
204 LET T=T+1
206 IF X> 50 THEN GOTO 1000
208 LET F=100:LET A=200
210 PRINT“FORE OR AFT”;
212 INPUT Y
215 IF Y=F THEN GOTO 240
220 IF Y=A THEN GOTO 240
225 LET T=T+3
230 GOSUB 2100
235 GOTO 200
240 LET B=X*10:LET C=B/2
242 GOSUB 1130
244 GOSUB 2100
246 FOR L=1 TO C:PRINT“ * ”;
248 FOR M=1TO234:NEXT M
250 NEXT L
252 LET G=G-B
```

```
254 IF G< 1 THEN GOTO 1150
256 IF Y=A THEN GOTO 1200
260 LET Z=Z+X
265 IF Z>100 THEN GOTO 1220
270 IF Z<1 THEN GOTO 1220
275 IF Z=1 THEN GOTO 1300
280 LET S=Z-1
285 PRINT"YOU ARE ";S;" SECTORS FROM HOME"
290 GOSUB 1500
295 LET A=Z+100:LET B=Z+200
300 LET C=Z+300:LET D=Z+400
302 LET E=Z+500
305 PRINT"THE NEAREST STAR IS ";
307 LET K=A(Z)
310 GOSUB 1240
312 LET K=A(A)
314 GOSUB 1240
316 PRINT" *";
318 LET K=A(B):PRINT K
320 PRINT"DISTANCE IS ";
322 LET K=A(E)
324 PRINT K;" LIGHT YEARS"
326 IF A(Z)=6 THEN GOTO 1320
330 LET P=A(C)
332 PRINT" THERE ARE ";P"PLANETS"
335 LET H=A(D)
340 PRINT H;" THYLTHRENS IN SOLAR SYSTEM"
345 LET N=601
350 FOR M=1 TO P
355 LET A(N)=INT(RND(0)*40)+1
357 LET N=N+1:NEXT M
360 IF H>0 THEN GOTO 830
365 PRINT"1 FOR SCAN-2 FOR SOLAR APPROACH-3 FOR
WARP";
367 INPUT S
370 IF S=3 THEN GOTO 200
375 IF S=2 THEN GOTO 800
380 IF S=1 THEN GOTO 430
385 PRINT"BAD INPUT!!!"
387 GOSUB 2000
390 PRINT"COMPUTER JAMMED!"
392 GOSUB 2000
394 PRINT:LET S=ABS(S)
396 LET T=T+S+1
398 GOSUB 2000
400 GOSUB 2100
```

```
402 LET S=INT(RND(0)*5)+1
405 IF S=4 THEN GOTO 365
410 IF S=5 THEN GOTO 365
415 GOTO 370
430 REM* SCAN MODE*
432 IF K>5000THEN GOTO 635
435 PRINT" .,SCANNING -----":PRINT
437 LET N=601:FOR Q=1 TO P
440 LET O=A(N)
442 GOSUB 2000
444 PRINT"PLANET #";Q,
446 GOSUB 2000
448 LET N=N+1
450 PRINT O
455 NEXT Q
457 GOSUB 2000
460 LET T=T+.67
462 GOSUB 2100
465 PRINT"3 FOR WARP,4 FOR PLANETARY APPROACH";
467 INPUT S
470 IF S=4 THEN GOTO 500
475 IF S=3 THEN GOTO 200
480 GOTO 385
500 PRINT:PRINT" .,PLANETARY APPROACH":PRINT
502 LET T=T+1.3
504 GOSUB 2100
506 PRINT"TARGET—PLANET#"
508 INPUT S
510 IF S> P THEN GOTO 1370
515 IF S<1 THEN GOTO 385
520 LET W=(K/P)*S
525 LET X=INT(RND(0)*100)+1
527 LET Y=INT(RND(0)*4)+1
530 IF Y> 2 THEN LET X=0-X
535 LET W=W+X
540 PRINT"DISTANCE TO PLANET #";S;"---";W
542 IF W<100 THEN GOTO 650
545 PRINT"APPROACH DRIVE POWER";
547 INPUT V
550 LET T=T+V:LET G=G-V
552 GOSUB 1500
555 IF V> 12 THEN GOTO 1020
557 IF V<2.2 THEN GOTO 385
560 LET U=INT(RND(0)*13)+1
562 LET V=V*U*3
565 GOSUB 2100
```

```
567 LET W=W-V
570 IF W<7 THEN GOTO 580
575 GOTO 540
580 IF W<-10 THEN GOTO 600
582 IF W<-6 THEN GOTO 650
585 PRINT"YOU COLLIDED WITH THE PLANET!"
587 GOSUB 2000
590 PRINT" YOU CLUMSY CLOD!"
592 GOSUB 2000
595 GOTO 1050
600 PRINT"OVER-SHOOT!"
602 GOSUB 2000
605 PRINT"EXCESS VELOCITY PROPELS YOU AWAY FROM
TARGET"
607 LET T=T+10
610 LET K=K+INT(RND(0)*8000)+(2*V)-(W+INT(RND(0)*1000)
615 PRINT"SOLAR DISTANCE IS ";K;" LIGHT YEARS"
620 GOSUB 2000
625 GOTO 365
635 PRINT"DISTANCE TOO GREAT FOR SCAN OR PLANETARY
APPROACH"
640 LET T=T+6.73
645 GOTO 365
650 PRINT"TARGET ACHIEVED!"
655 GOSUB 2000
660 PRINT"CLOSE RANGE SCAN---";
662 GOSUB 2000
664 LET T=T-2.3
666 LET F=S+600
670 LET P=A(F)
675 LET P=(P/4)*(INT(RND(0)*25)+2)
680 IF P>100 THEN P=0
685 PRINT"ATMOSPHERIC% OF BRETCHENKOEL--PLANET
#";S
690 GOSUB 2000
695 PRINT" ", " ", P; "%":PRINT
700 IF P=0 THEN GOTO 790
705 IF P<25 THEN GOTO 780
710 IF P>87 THEN GOTO 750
712 PRINT"LOADING---"
714 GOSUB 2000
716 GOSUB 2000
718 LET B=INT(RND(0)*17)+4
720 LET P=P*B
722 PRINT:PRINT
724 PRINT P;"UNITS OF BRETCHENKOEL COLLECTED"
```

```
726 LET G=G+P-20
728 LET J=J+P
730 GOSUB 2000
732 PRINT"TOTAL BRETCHENKOEL NOW ON BOARD---";
734 GOSUB 2000
736 PRINT J:PRINT:PRINT
738 LET T=T-4
740 IF J<5020 THEN GOTO 365
750 PRINT"MISSION ACCOMPLISHED!"
752 GOSUB 2000
755 PRINT"GOOD WORK, ";A$
757 GOSUB 2000
760 PRINT"YOUR PEOPLE ARE SAVED!"
765 END
780 LET T=T+2
782 PRINT"THAT'S SCARCELY WORTH BOTHERING WITH!"
784 GOSUB 2000
786 GOSUB 2100
788 LET J=J+P:GOTO 730
790 PRINT"THEY'RE NO BRETCHENKOEL HERE, STUPID!"
792 LET T=T+13
794 GOSUB 2100
796 GOTO 365
800 LET K=K-INT(RND(0)*K)
802 LET T=T+1.5
805 IF K<200 THEN GOTO 1600
810 GOTO 320
830 PRINT"THYLATHERNS ATTACK!!"
832 LET Q=0:GOSUB 2000
835 LET N=INT(RND(0)*10)+601
837 FOR L=1 TO H
840 LET U=A(N)
842 PRINT U;"MEGABLORTS OF ENERGY FROM THYLATHERN
#2":L
845 LET N=N+1:LET Q=Q+U-INT(RND(0)*10)
847 NEXT L:LET T=T+H
850 GOSUB 2100
852 IF Q>400 THEN GOTO 1400
855 PRINT"3 FOR WARP, 5 TO RETURN FIRE ";
857 INPUT S
860 IF S=3 THEN GOTO 200
865 IF S=5 THEN GOTO 880
870 PRINT"BAD INPUT"
872 GOTO 830
880 PRINT"MEGABLORTS";
882 INPUT M
```

```
884 LET T=T+M
886 LET F=INT(RND(0)*201)+1
888 IF M> 1000 THEN GOTO 1040
890 IF M<F THEN GOTO 910
895 LET M=M-35-INT(RND(0)*100)
897 LET H=H-1:LET T=T+1.2
900 PRINT“1THYLTHERN DESTROYED!”
902 IF H=0 THEN GOTO 930
910 PRINT H;“THYLTHRENS LEFT”
915 LET Q=0
920 GOTO 835
930 PRINT“GOT 'EM ALL!”
932 GOSUB 2000
934 LET T=T+INT(RND(0)*10)+1
936 GOSUB 2100
940 GOTO 365
1000 FOR L=1 TO 500:NEXT L
1005 LET Y=INT(RND(0)*30)+1
1010 FOR M=1 TO Y
1015 PRINT“ *”;
1020 NEXT M
1025 PRINT“OVER-DRIVE!!!”
1030 FOR M=1TO Y
1035 PRINT“ * ”;
1040 NEXT M
1045 PRINT:FOR L=1 TO 470:NEXT L
1050 PRINT“YOU BLEW IT, ”;A$
1055 GOSUB 2000
1060 PRINT“YOU ARE DEAD ”;
1065 GOSUB 2000
1070 PRINT“AND SO ARE ALL THE PEOPLE BACK ON
RAMSNEID!”
1075 FOR L=1 TO 908:NEXT L
1080 GOTO 160
1100 PRINT“IT'S TOO LATE!”
1102 GOSUB 2000
1105 PRINT“EVERYONE BACK HOME IS DEAD!”
1110 GOSUB 2000
1115 GOTO 160
1130 IF C<80 THEN GOTO 1140
1135 LET C=C/2
1137 GOTO 1130
1140 RETURN
1150 PRINT“ ”,“OOPS!”
1152 GOSUB 2000
```

```
1155 PRINT"YOU ARE OUT OF FUEL!"
1160 GOSUB 2000
1165 PRINT"WHAT SHODDY PLANNING!"
1170 GOTO 1055
1200 LET Z=Z-X
1205 IF Z<1 THEN GOTO 1220
1210 GOTO 285
1220 PRINT"YOU JUST WARPED 'EVENING STAR' RIGHT OUT"
1222 PRINT"OF THE GALAXY!"
1225 GOTO 1160
1230 LET A(1)=6
1232 LET A(101)=18
1234 LET A(201)=972
1235 GOTO 1270
1240 IF K=1 THEN PRINT"A";
1241 IF K=2 THEN PRINT"B";
1242 IF K=3 THEN PRINT"C";
1243 IF K=4 THEN PRINT"D";
1244 IF K=5 THEN PRINT"E";
1245 IF K=6 THEN PRINT"F";
1246 IF K=7 THEN PRINT"G";
1247 IF K=8 THEN PRINT"H";
1248 IF K=9 THEN PRINT"I";
1249 IF K=10 THEN PRINT"J";
1250 IF K=11 THEN PRINT"K";
1251 IF K=12 THEN PRINT"L";
1252 IF K=13 THEN PRINT"M";
1253 IF K=14 THEN PRINT"N";
1254 IF K=15 THEN PRINT"O";
1255 IF K=16 THEN PRINT"P";
1256 IF K=17 THEN PRINT"Q";
1257 IF K=18 THEN PRINT"R";
1258 IF K=19 THEN PRINT"S";
1259 IF K=20 THEN PRINT"T";
1260 IF K=21 THEN PRINT"U";
1261 IF K=22 THEN PRINT"V";
1262 IF K=23 THEN PRINT"W";
1263 IF K=24 THEN PRINT"X";
1264 IF K=25 THEN PRINT"Y";
1265 IF K=26 THEN PRINT"Z";
1267 RETURN
1270 FOR X=401 TO 500
1272 LET Y=INT(RND(0)*30)+1
1275 IF Y> 13 THEN LET Y=0
1277 LET A(X)=Y
1280 NEXT X
1285 FOR X=501 TO 600
```

```
1290 LET Y=INT(RND(0)*8010)+1
1292 LET A(X)=Y:NEXT X
1295 RETURN
1300 PRINT"YOU CAME HOME WITHOUT ENOUGH
      BRETCHENKOEL?!""
1305 GOSUB 2000
1310 PRINT"YOU ARE PROMPTLY LYNCHED!"
1315 GOTO 160
1320 IF A(A)=18 THEN GOTO 1330
1325 GOTO 340
1330 IF A(B)=972 THEN GOTO 1340
1335 GOTO 340
1340 PRINT"HEY"
1342 GOSUB 2000
1344 PRINT"YOU MUST'VE GONE THROUGH A BLACK HOLE OR"
1346 PRINT" SOMETHING, BECAUSE SUDDENLY YOU'RE
      HOME!""
1348 GOSUB 2000
1350 PRINT"AND WHAT A GREETING YOU RECEIVE!"
1352 GOSUB 2000
1355 PRINT"THE PEOPLE ALL GATHER AROUND AND SAY—"
1357 GOSUB 2000
1360 GOTO 1300
1370 PRINT"NO SUCH PLANET"
1372 PRINT:PRINT
1375 LET F=INT(RND(0)*30)+1
1377 LET T=T+F
1380 IF F> 23 THEN GOTO 1390
1385 GOTO 390
1390 PRINT" YOU ARE HEREBY STRIPPED OF YOUR
      COMMAND!"
1392 GOSUB 2000
1395 GOTO 160
1400 PRINT"EVENING STAR IS DESTROYED!"
1405 GOSUB 2000
1410 GOTO 1045
1500 LET M=INT(RND(0)*101)+1
1505 IF M>97 THEN GOTO 1520
1510 RETURN
1520 PRINT"METEOR STRIKE!!!!"
1522 LET M=INT(RND(0)*34)+1
1525 LET T=T+M
1530 IF T>100 THEN GOTO 1100
1535 RETURN
2000 LET Y=INT(RND(0)*1000)+100
2005 FOR X=1 TO Y
```

```
2010 NEXT X
2015 RETURN
2100 LET U=RND(0)
2105 LET T=T+U
2110 IF T>100 THEN GOTO 1100
2115 RETURN

1600 PRINT"YOU'VE DIVED INTO THE SUN!"
1605 GOSUB 2000
1610 GOTO 1400
```

TRS-80 BASIC

```
10 CLS:P.;P." YOU ARE ";:A=RND(5)
20 IF A=1 A$="AKRANAID"
25 IF A=2 A$="OLK"
30 IF A=3 A$="ZOLAR ZINNK"
35 IF A=4 A$="GUAMBNA"
40 IF A=5 A$="JOHN DOE"
45 P.A$;" — CAPTAIN OF THE SPACESHIP"
50 P."EVENING STAR' FROM THE PLANET RAMSNEID"
55 J=0:F.X=1TO200:Y=RND(26):A(X)=Y:N.X
60 P."RAMSNEID CIRCLES THE STAR KNOWN AS FR-972 AND"
65 P."IS IN DIRE NEED OF THE RARE SUBSTANCE,
BRETCHENKOEL"
70 P."THERE IS NO BRETCHENKOEL LEFT IN YOUR SOLAR
SYSTEM"
72 P."AND WITHOUT THIS SUBSTANCE TO FUEL THE
PLANET'S"
75 P."POWER STATIONS, YOUR PEOPLE WILL DIE OUT."
77 P."THIS IS WHY THE RAMSNEIDIANS HAVE TAKEN UP
SPACE"
80 P."TRAVEL, DESPITE ITS STRANGE & UNPREDICTABLE
DANGERS"
85 F.X=201TO300:Y=RND(999):A(X)=Y:N.X
90 PRINT"YOUR MISSION, CAPTAIN";A$;“, IS TO SEARCH
THE"
95 PRINT"GALAXY FOR MORE BRETCHENKOEL. YOU WILL
BE FLYING"
100 PRINT"BY GUESS WORK MOST OF THE TIME"
105 F.X=301TO400:Y=RND(17)
110 IF Y=17 THEN Y=0
112 IF Y=1 THEN Y=0
115 A(X)=Y:N.X:GOS.1230
117 PRINT“BUT WATCH OUT FOR METEORS, AND THE
GREATLY FEARED,”
120 P.“WAR-LOVING THYLTHRENS!”
125 F.X=401TO500:Y=RND(10):A(X)=Y:N.X:N=15
```

```

127 F.X=501TO600:Y=RND(1050):A(X)=Y:N.X
130 IN."ARE YOU PREPARED TO UNDERTAKE THIS
    MISSION":X
135 IF X=Y G.180
140 IF X=N G.160
145 P."YOU MUST ANSWER YES OR NO,";A$:G.130
160 P."YOU ARE A GREAT SHAME TO YOUR RACE,";A$;"!"
165 P.:GOS.2000:P." ","GOOD-BYE!"
170 END
180 P.:P.:Z=1:T=1
185 G=30000:J=0
200 IN."WARP DRIVE";X:T=T+1
205 IF X> 50 G.1000
210 F=100:A=200:IN."FORE OR AFT";Y
215 IF Y=F G.240
220 IF Y=A G.240
225 T=T+3:GOS.2100:G.200
240 B=X*10:C=B/2:GOS.1130:GOS.2100
245 F.L=1TOC:P."**";:F.M=1TO134:N.M:N.L
250 G=G-B:IF G> G.1150
260 IF Y=A G.1200
270 Z=Z+X
280 IF Z>100 G.1220
285 IF Z> 1 G.1220
290 IF Z=1 G.1300
295 S=Z-1:P."YOU ARE";S;"SECTORS FROM HOME"
300 GOS.1500:A=Z+100:B=Z+200:C=Z+300:D=Z+400:E+500
305 P."THE NEAREST STAR IS":K=A(Z):GOS.1240
310 K=A(A):GOS.1240:P."**";:K=A(B):P.K
315 P."DISTANCE IS ";
320 K=A(E):P.K;" LIGHT YEARS"
325 IF A(Z)=6 G.1320
330 P=A(C):H=A(D):P."THERE ARE ";P;" PLANETS"
335 P."THERE ARE ";H;" THYLTHRENS IN SOLAR SYSTEM"
340 N=601:F.M=1TOP:A(N)=RND(40):N=N+1:N.M
350 IF H>0 G.830
365 IN."1 FOR SCAN — 2 FOR SOLAR APPROACH— 3 FOR
    WARP";S
370 IF S=3 G.200
375 IF S=2 G.800
380 IF S=1 G.430
385 P." BAD INPUT!!!!":GOS.2000
390 P."COMPUTER JAMMED!!":GOS.2000:P.
395 S=ABS(S):T=T+S+1:GOS.2000:GOS.2100
400 S=RND(5):IF S=4 G.365
405 IF S=5 G.365

```

410 G.370
430 IF K>5000 G.635
435 P. " ", "SCANNING ----":P:N=601
440 F.Q=1TOP:O=A(N):GOS.2000:P."PLANET #";Q,
445 GOS.2000:N=N+1:P.O:N.Q
460 GOS.2000:T=T+.67:GOS.2100
465 IN."3 FOR WARP,4 FOR PLANETARY APPROACH";S
470 IF S=4 G.500
475 IF S=3 G.200
480 G.385
500 P.:P." ", "PLANETARY APPROACH":P.
502 T=T+1.3:GOS.2100
505 IN."TARGET --- PLANET #";S
510 IF S>P G.1370
515 IF S<1 G.385
520 W=(K/P)*S
525 X=RND(100):Y=RND(4)
530 IF Y> 2 THEN X=0-X
535 W=W+X
540 P."DISTANCE TO PLANET #";S;--- ";W
542 IF W<100 G.650
545 IN."APPROACH DRIVE POWER ";V
547 T=T+V:G=G-V:GOS.1500
550 IF V>12 G.1020
555 IF V< 2.2 G.385
560 U=RND(13):V=U*V*3:GOS.2100
565 W=w-V
570 IF W< 7 G.580
575 G.540
580 IF W<-10 G.600
582 IF W<-6 G.650
585 P."YOU COLLIDED WITH THE PLANET!":GOS.2000
590 P." YOU CLUMSY CLOD!"
595 GOS.2000:G.1050
600 PRINT"OVER-SHOOT!":GOS.2000
605 PRINT"EXCESS VELOCITY PROPELS YOU AWAY FROM
TARGET"
610 T=T+10:K=K+RND(8000)+2*V-(W+RND(1000))
615 PRINT "SOLAR DISTANCE IS ";K;"LIGHT YEARS"
620 GOS.2000:G.365
635 PRINT"DISTANCE TOO GREAT FOR SCAN OR PLANETARY
APPROACH"
640 T=T+6.73:G.365
650 PRINT"TARGET ACHIEVED!":GOS.2000
660 PRINT"CLOSE RANGE SCAN --- ";
665 GOS.2000:T=T+2.3:F=S+600

670 P=A(F)
675 P=(P/4)*(RND(25)+2)
680 IF P> 100 THEN P=0
685 PRINT"ATMOSPHERIC% OF BRETCHENKOEL — PLANET
#";S
690 GOS.2000
695 P." ", " ,P;"%":P.
700 IF P=0 G.790
705 IF P<25 G.780
710 IF P> 87 G.750
715 P.“LOADING --”:GOS.2000:GOS.2000
720 B=RND(17)+3:P=P*B
722 P.P;“ UNITS OF BRETCHENKOEL COLLECTED”
725 G=G+P-20:J=J+P
730 GOS.2000:P.“TOTAL BRETCHENKOEL NOW ON BOARD --”;
732 GOS.2000:P.J:P.:P.:T=T-4
735 IF J<5020G.365
750 P.“MISSION ACCOMPLISHED!”:GOS.2000
755 P.“GOOD WORK, ”;A\$:GOS.2000
760 P.“YOUR PEOPLE ARE SAVED!”
765 END
780 T=T+2:P.“THAT'S SCARECELY WORTH BOTHERING
WITH!”
785 GOS.2000:GOS.2100:J=J+P:G.730
790 P.“THERE'S NO BRETCHENKOEL HERE, STUPID!”
792 T=T+13:GOS.2100
795 G.365
800 K=K-RND(K):T=T+1.5
805 IF K<200 G.1600
810 G.315
830 P.“THYLATHERNS ATTACK!!”Q=0:GOS.2000
835 N=RND(10)+600:F.L=1TOH
840 U=A(N)
842 PRINTU;“ MEGABLORTS OF ENERGY FROM THYLATHERN
#”;L
845 N=N+1:Q=Q+U-RND(10):N.L:T=T+H:GOS.2100
850 IF Q>400 G.1400
855 IN.“3 FOR WARP, 5 TO RETURN TO FIRE”;S
860 IF S=3 G.200
865 IF S=5 G.880
870 P.“ BAD INPUT!”:G.830
880 IN.“MEGABLORTS”;M:T=T+M:F=RND(201)
885 IF M>1000G. 1040
890 IF M<F G.910
895 M=M-35-RND(100):H=H-:T=T+1.2
900 P.“1 THYLATHERN DESTROYED!”

902 IF H=0 G.930
910 P.H;“ THYLTHRENS LEFT”:Q=0
915 G.835
930 P.“ GOT 'EM ALL! ”:GOS.2000
935 T=T+RND(10):GOS. 2100
940 G.365
1000 F.L=1TO500:N.L:Y=RND(30)
1005 F.M=1TOY:P.“ * ”;:N.M
1010 P.“ OVER-DRIVE!!! ”
1015 F.M=1TOY:P.“ * ”;:N.M
1045 P.:F.L=1TO470:N.L
1050 P.“ YOU BLEW IT, ”;A\$:GOS.2000
1060 P.“ YOU ARE DEAD ”;:GOS.2000
1070 P.“ AND SO ARE ALL THE PEOPLE BACK ON REMSNEID! ”
1080 F.L=1TO908:N.L:G.160
1100 P.“ IT'S TOO LATE! ”:GOS.2000
1105 P.“ EVERYONE BACK HOME IS DEAD! ”:GOS.2000
1110 G.160
1130 IF C<20 THEN G.1140
1135 C=C/2:G.1130
1140 RET.
1150 P.“ , “OOPS! ”:GOS.2000
1160 P.“ YOU ARE OUT OF FUEL! ”:GOS.2000
1165 P.“ WHAT SHODDY PLANNING! ”:G.1060
1200 Z=Z-X:IF Z<1G1220
1210 G.285
1220 P.“ YOU JUST WARPED 'EVENING STAR' RIGHT OUT OF ”
1225 P.“ THE GALAXY! ”:G.1165
1230 A(1)=6:A(101)=18:A(201)=972
1235 G.1270
1240 IF K=1P.“A”;
1241 IF K=2P.“B”;
1242 IF K=3P.“C”;
1243 IF K=4P.“D”;
1244 IF K=5P.“E”;
1245 IF K=6P.“F”;
1246 IF K=7P.“G”;
1247 IF K=8P.“H”;
1248 IF K=9P.“T”;
1249 IF K=10P.“J”;
1250 IF K=11P.“K”;
1251 IF K=12P.“L”;
1252 IF K=13P.“M”;
1253 IF K=14P.“N”;
1254 IF K=15P.“O”;
1255 IF K=16P.“P”;

1256 IF K=17P."Q";
1257 IF K=18P."R";
1258 IF K=19P."S";
1259 IF K=20P."T";
1260 IF K=21P."U";
1261 IF K=22P."V";
1262 IF K=23P."W";
1263 IF K=24P."X";
1264 IF K=25P."Y";
1265 IF K=26P."Z";
1268 RET.
1270 F.X=401TO500:Y=RND(30)
1275 IF Y>13 THEN Y=0
1280 A(X)=Y:N.X
1285 F.X=501TO600:Y=RND(8010)
1290 A(X)=Y:N.X:RET.
1300 P."YOU CAME HOME WITHOUT ENOUGH
BRETCHENKOEL?!?"
1305 GOS.2000
1310 P."YOU ARE PROMPTLY LYNCHED!":G.160
1320 IF A(A)=18 G.1330
1325 G.340
1330 IF A(B)=972 G.1340
1335 G.340
1340 P."HEY! "1:GOS.2000
1345 P."YOU MUST'VE GONE THROUGH A BLACK HOLE OR"
1347 P."SOMETHING, BECAUSE SUDDENLY YOU'RE HOME!"
1350 GOS.2000:P."AND WHAT A GREETING YOU RECEIVE!"
1352 GOS.2000
1355 P."THE PEOPLE ALL GATHER AROUND AND SAY — "
1360 GOS.2000:G.1300
1370 P."NO SUCH PLANET!":P.:P.:F=RND(30):T=T+F
1375 IF F>23 THEN G.1390
1380 G.390
1390 P."YOU ARE HEREBY STRIPPED OF YOUR COMMAND!"
1395 GOS.2000:G.160
1400 P." 'EVENING STAR' IS DESTROYED!":GOS.2000
1410 G.1050
1500 M=RND(101)
1505 IF M>97 THEN G.1520
1510 RET
1520 P."METEOR STRIKE!!!":M=RND(34):T=T+M
1525 IF T>100 G.1100
1530 RET.
1600 P."YOU'VE DIVED INTO THE SUN!":GOS.2000
1610 G.1400

```
2000 Y=RND(1000)+100
2005 F.X= 1 TOY:N.X
2010 RET.
2100 U=RND(0):T=T+U
2110 IF T>100 G.1100
2120 RET.
```

Summary of Variables Used

A Name selection/AFT/various
B various
C various
D various
E various
F FORE/various
G various
H # of Thytherns
I Not used
J Bretchenkoel on board
K Star name/distance
L Timing
M Timing
N various
O Planetary scan
P # of planets/% of Bretchenkoel
Q Planet #
R Not used
S various
T energy used
U various
V Approach drive power
W Planetary approach
X Timing/various
Y various
Z Location/various

Sample Run (excerpt)

```
WARP DRIVE? 10
FORE OR AFT? FORE
*****
```

```
YOU ARE 18 SECTORS FROM HOME
THE NEAREST STAR IS QB*19
DISTANCE IS 170 LIGHT YEARS
THERE ARE 5 PLANETS
THERE ARE 3 THYLATHERNS IN SOLAR SYSTEM
1 FOR SCAN — 2 FOR SOLAR APPROACH — 3 FOR WARP ? 1
SCANNING—
PLANET #1
```

PLANET #2 13
PLANET #3 37
PLANET #4 8
PLANET #5 25

3 FOR WARP, 4 FOR PLANETARY APPROACH? 3
 PLANETARY APPROACH

TARGET PLANET #? 3

DISTANCE TO PLANET #3 — 64

APPROACH DRIVE POWER? 2

DISTANCE TO PLANET #3 — 34

APPROACH DRIVE POWER? 2

TARGET ACHIEVED

CLOSE RANGE SCAN

ATMOSPHERE % OF BRETCHENKOEL — PLANET #3

85%

LOADING---

1275 UNITS OF BRETCHENKOEL COLLECTED

TOTAL BRETENKOEL COLLECTED -- 2010

WARP DRIVE? 8

High Bid

This is a simple game. The instructions are included in the program and are self-explanatory.


At first glance it would seem that whoever went first would inevitably lose that round, but as you get in some practice you will learn how to use strategy to break up the tie games.

The sample run included is an early game without too much strategy being used. See Fig. 1-8 for the flowchart.

Standard BASIC

```
5 PRINT " ", "HIGH BID":PRINT
10 PRINT"WE EACH GET THE NUMBERS FROM 1 TO 10."
15 PRINT"EACH # MAY BE PLAYED ONLY ONCE. ON EACH"
20 PRINT"ROUND WE BOTH BID A # AND THE HIGH BID"
25 PRINT"WINS THE ROUND. WHOEVER WON THE LAST"
30 PRINT"ROUND GOES FIRST. THE FIRST ROUND WILL"
32 PRINT"BE SELECTED RANDOMLY."
35 REM* SET CHOICE ARRAYS & SCORE
37 FOR X=1 TO 20
40 LET A(X)=1
42 NEXT X
44 LET A=0:LET B=0
46 REM* PLAYER SELECT
48 FOR X=1 TO 470
50 LET P=INT(RND(0)*2)+1
52 NEXT X
54 REM* SET COUNTER
56 FOR T=1 TO 10
58 IF P=1 THEN GOTO 180
60 PRINT"I GO FIRST"
62 GOSUB 120
64 GOSUB 150
66 GOSUB 170
70 LET A(X)=0
72 LET A(Y+10)=0
75 IF X=Y THEN GOTO 205
80 IF X>Y THEN GOTO 220
85 PRINT"I WIN THIS ROUND"
90 LET B=B+1
92 LET P=2
95 NEXT T
100 PRINT:PRINT"THE GAME IS OVER --- FINAL SCORE"
105 PRINT:PRINT"YOU", "ME"
```

Fig. 1-8. High Bid flowchart.



```
110 PRINT A,B
115 END
120 PRINT "I PLAY --- ";
125 LET Y=INT(RND(0)*10)+1
130 IF A(Y+10)=0 THEN GOTO 125
135 FOR Z=1 TO 333: NEXT Z
137 PRINT Y
140 RETURN
150 PRINT"YOUR PLAY";
155 INPUT X
160 IF A(X)=0 THEN GOTO 155
165 RETURN
170 FOR Z=1 TO 400:NEXT Z
175 RETURN
180 PRINT"YOU GO FIRST"
185 GOSUB 150
190 GOSUB 120
195 GOSUB 170
200 GOTO 70
205 PRINT"tie"
210 LET P=INT(RND(0)*2)+1
215 GOTO 95
220 PRINT"YOU WIN THIS ROUND"
225 LET A=A+1
230 LET P=1
235 GOTO 95
```

TRS-80 BASIC

```
5 CLS:P.“ ”“HIGH BID”:P.
10 P.“WE EACH GET THE NUMBERS FROM 1 TO 10. EACH”
15 P.“# MAY BE PLAYED ONLY ONCE. ON EACH ROUND”
20 P.“WE BOTH BID A # AND THE HIGH BID WINS THE”
25 P.“ROUND. WHOEVER WON THE LAST ROUND GOES”
30 P.“FIRST. THE FIRST ROUND WILL BE SELECTED”
32 P.“RANDOMLY”
35 F.X= 1TO20:A(X)=1:N.X:A=0:B=0
40 F.X=1TO470:P=RND(2):N.X
45 F.T=1TO10
50 IF P=1 G.180
55 P.“I GO FIRST”:GOS.120
60 GOS.150
65 GOS.170
70 A(X)=0:A(Y+10)=0
75 IF X=Y G.205
80 IF X>Y G.220
```

```
85 P."I WIN THIS ROUND"
90 B=B+1:P=2
95 N.T
100 P.:P."GAME IS OVER --- FINAL SCORE":P.
105 P."YOU","ME"
110 P.A,B
115 END
120 P."I PLAY --- ";
125 Y=RND(10)
130 IF A(Y+10)=0 G.125
135 F.Z=1TO333:N.Z
140 P.Y:RET.
150 IN."YOUR PLAY";X
155 IF A(X)=0 G.150
160 RET.
170 F.Z=1TO400:N.Z:RET.
180 P."YOU GO FIRST"
185 GOS.150
190 GOS.120
195 GOS.170
200 G.70
205 P."TIE":P=RND(2)
210 G.95
220 P."YOU WIN THIS ROUND"
225 A=A+1:P=1
230 G.95
```

Sample Run

HIGH BID

WE EACH GET THE NUMBERS FROM 1 TO 10.
EACH # MAY BE PLAYED ONLY ONCE. ON EACH
ROUND WE BOTH BID A # AND THE HIGH BID
WINS THE ROUND. WHOEVER WON THE LAST
ROUND GOES FIRST. THE FIRST ROUND WILL
BE SELECTED RANDOMLY

I GO FIRST

I PLAY --- 7

YOUR PLAY?9

YOU WIN THIS ROUND

YOU GO FIRST

YOUR PLAY?3

I PLAY --- 4

I WIN THIS ROUND

I GO FIRST

I PLAY --- 2

YOUR PLAY?3
YOUR PLAY?2
TIE
I GO FIRST
I PLAY --- 5
YOUR PLAY?6
YOU WIN THIS ROUND
YOU GO FIRST
YOUR PLAY?1
I PLAY --- 3
I WIN THIS ROUND
I GO FIRST
I PLAY --- 10
YOUR PLAY?2
YOUR PLAY?4
I WIN THIS ROUND
I GO FIRST
I PLAY --- 8
YOUR PLAY?10
YOU WIN THIS ROUND
YOU GO FIRST
YOUR PLAY?8
I PLAY --- 9
I WIN THIS ROUND
I GO FIRST
I PLAY --- 6
YOUR PLAY?7
YOU WIN THIS ROUND
YOU GO FIRST
YOUR PLAY?5
I PLAY --- 1
YOU WIN THIS ROUND
GAME IS OVER --- FINAL SCORE
YOU ME
 5 4

Summary of Variables Used

- A PLAYER'S SCORE
- B COMPUTER'S SCORE
- P PLAYER SELECT
- T TURN COUNTER
- X TIMING/PLAYER'S BID
- Y COMPUTER'S BID

High Bid II

This game is essentially similar to the original *High Bid* game, but there is a greater element of chance to sustain continued interest.

Both the player and the computer still get ten numbers to play, but instead of a fixed 1 to 10, each of the numbers is selected randomly. Each number may be played only once each time it appears in your hand. To make things a little trickier, you are not told what numbers the computer is holding (it only looks at your hand when making a validity check, so the game is fair).

Since this game is played the same way as the other version, no sample run is included.

Standard BASIC

```
5 PRINT:PRINT:PRINT
10 PRINT" ","HIGH BID II"
12 PRINT
14 FOR X=1 TO 20
16 LET A(X)=INT(RND(0)*10)+1
18 NEXT X
20 LET A=0:LET B=0:LET G=1
22 LET P=INT(RND(0)*2)+1
25 IF G>10 THEN GOTO 130
30 PRINT"YOUR HAND -- ";
32 FOR X=1TO 10
34 LET Y=A(X)
36 IF Y=0 THEN GOTO 40
38 PRINT Y;" ";
40 NEXT X:PRINT
45 IF P=1 THEN GOTO 200
50 PRINT"I GO FIRST"
55 LET X=11
60 IF A(X)>0 THEN GOTO 70
62 LET X=X+1
65 GOTO 60
70 LET Y=A(X)
75 LET A(X)=0
80 PRINT "I'LL PLAY -- ";
85 GOSUB 330
90 PRINT Y
95 GOSUB 275
100 IF X=Y THEN GOTO 150
105 IF X>Y THEN GOTO 170
```

```
110 PRINT "I WIN THIS ROUND"
112 LET B=B+1
114 LET P=2
116 LET G=G+1
118 GOSUB 330
120 GOTO 25
130 PRINT:PRINT"GAME IS OVER — FINAL SCORE"
135 PRINT"YOU","ME"
140 PRINT A,B
145 END
150 PRINT"TIE"
155 LET P=INT(RND(0)*2)+1
160 GOTO 116
170 PRINT"YOU WIN THIS ROUND"
175 LET A=A+1
180 LET P=1
185 GOTO 116
200 PRINT"YOU GO FIRST"
202 GOSUB 275
204 LET Y=X+1
206 LET C=0
208 PRINT"ILL PLAY --- ";
210 IF Y=11 THEN Y=1
215 FOR Z=11 TO 20
220 IF A(Z)=Y THEN C=Z
225 NEXT Z
230 IF C>0 THEN GOTO 250
235 LET Y=Y+1
240 GOTO 210
250 LET A(C)=0
255 PRINT Y
260 GOSUB 330
265 GOTO 100
275 PRINT"YOUR PLAY";
280 INPUT X
285 LET C=0
290 FOR Z=1TO10
295 IF A(Z)=X THEN LET C=Z
300 NEXT Z
305 IF C>0 THEN GOTO 320
310 PRINT"INVALID PLAY"
315 GOTO 275
320 LET A(C)=0
330 FOR Z=1 TO 333
335 NEXT Z
340 RETURN
```

TRS-80 BASIC

```
5  CLS:P."","HIGH BID IT":P.  
10 F.X=1TO20:A(X)=RND(10):N.X.  
15 P=RND(2):A=0:B=0:G=1  
20 IF G<10 G. 130  
25 ."YOUR HAND --- ";  
30 F.X=1TO 10:Y=A(X):IF Y=0 G. 40  
35 P.Y;" ";  
40 N.X.:P.  
50 IF P=1 G. 200  
55 P." I GO FIRST":X=11  
60 IF A(X)>0 G. 70  
65 X=X+1:G.60  
70 Y=A(X):A(X)=0  
75 P."TLL PLAY --- ";  
80 GOS. 320  
85 P.Y  
90 GOS. 275  
100 IF X=Y G. 150  
105 IF X>Y G. 170  
110 P."I WIN THIS ROUND"  
115 B=B+1:P=2  
120 G=G+1:GOS.320  
125 G.20  
130 P.:P."GAME IS OVER --- FINAL SCORE"  
135 P."YOU","ME"  
140 P.A,B  
145 END  
150 P."TIE"  
155 P=RND(2)  
160 G.120  
170 P."YOU WIN THIS ROUND"  
175 A=A+1:P=1:G.120  
200 P."YOU GO FIRST"  
205 GOSUB 275  
210 Y=X+1:C=0:P."TLL PLAY --- ";  
215 IF Y=11 THEN Y=1  
220 F.Z=11 TO 20  
225 IF A(Z)=Y THEN C=Z  
230 N.Z  
235 IF C>0 G. 250  
240 Y=Y+1:G. 215  
250 A(C)=0  
255 P.Y  
260 GOS. 320  
265 G. 100
```

275 IN.“YOUR PLAY”;X
280 C=0:F.Z=1TO 10
285 IF A(Z)=X THEN C=Z
290 N.Z
295 IF C>0 G. 310
300 P. “INVALID PLAY”
305 G.275
310 A(C)=0
320 F.Z=1TO333:N.Z
325 RET.

Balancing The Scales

This game is simple in concept, but playing it can be quite tricky. The computer gives you 25 weights of up to 10 grams each, and you have to place them all on a four-way scale arranged like this:



If two opposite plates (A and B or C and D) have more than a 5 gram difference, the scale assembly will topple over. Also, if the two crossbeams (A+B and C+D) have more than a 7.5 gram difference, the scale will topple.

If you manage to get all 25 weights on the scale without toppling it, you win. If the scale topples, you lose.

An additional note: once a weight has been placed on the scale, it can't be moved. See Fig. 1-9 for the flowchart.

Standard BASIC

```
5 PRINT:PRINT
7 PRINT "", "BALANCING THE SCALES":PRINT
10 REM * SET WEIGHTS & CLEAR SCALE *
12 FOR X=1TO 25
15 LET Y=INT (RND(0)*100)+1
17 LET Y=Y/10:LET A(X)=Y
20 NEXT X
22 LET A=50:LET B=60
25 LET C=70:LET D=80
30 LET E=0:LET F=0
32 LET G=0:LET H=0
35 REM * THE PLAY *
37 FOR X=1TO 470:PRINT:NEXT X
40 PRINT "THE AVAILABLE WEIGHTS ARE",
42 FOR X=1TO 25
45 PRINT X;" ";A(X);";";
47 NEXT X
50 PRINT:PRINT:PRINT "SCALE RANKING"
55 LET X=10
60 IF E=X THEN PRINT "A--";X,
65 IF F=X THEN PRINT "B--";X,
70 IF G=X THEN PRINT "C--";X,
75 IF H=X THEN PRINT "D--";X,
80 IF X=0 THEN GOTO 90
82 LET X=X-.1
```

```
85 GOTO 60
90 PRINT“WHICH WEIGHT?”
92 INPUT W
95 IF W>25 THEN GOTO 135
100 IF A(W)=0 THEN GOTO 135
105 PRINT“WHICH SCALE?”
107 INPUT S
110 FOR X=1TO390:NEXT X
115 IF S=A THEN GOTO 140
120 IF S=B THEN GOTO 145
125 IF S=C THEN GOTO 150
130 IF S=D THEN GOTO 155
135 REM* INVALID ENTRY*
137 PRINT“DON’T GET CUTE, PAL!”
138 GOTO 90
140 LET E=E+A(W)
142 GOTO 160
145 LET F=F+A(W)
147 GOTO 160
150 LET G=G+A(W)
152 GOTO 160
155 LET H=H+A(W)
160 LET A(W)=0
162 REM* TOPPLE CHECK*
165 IF E>(F+5) THEN GOTO 250
170 IF F>(E+5) THEN GOTO 250
175 IF G>(H+5) THEN GOTO 250
180 IF H>(G+5) THEN GOTO 250
185 LET I=E+F:LET J=G+H
190 IF I>(J+7.5) THEN GOTO 250
195 IF J>(I+7.5) THEN GOTO 250
200 REM* REMAINING WEIGHT CHECK*
202 LET Y=0
205 FOR X=1 TO 25
207 LET Y=Y+A(X)
210 NEXT X
212 IF Y=0 THEN GOTO 225
215 GOTO 35
225 PRINT“THAT WAS THE LAST WEIGHT”:PRINT
230 FOR X=1TO333:NEXT X
235 PRINT“YOU WIN!”
240 END
250 PRINT“SCALE STRUCTURE TOPPLES!”:PRINT
255 FOR X=1TO333:NEXT X
260 PRINT“YOU LOSE!”
265 END
```




Fig. 1-9. Flowchart for Balancing the Scales.

TRS-80 BASIC

```

10  CLS:P.P.:P"","BALANCING THE SCALES":P.
15  F.X=1TO25
20  Y=RND(100)/10:A(X)=Y
25  N.X:A=50:B=60:C=70
30  D=80:E=0:F=0:G=0:H=0
35  F.X=1TO470:N.X
40  CLS:P."THE AVAILABLE WEIGHTS ARE",
45  F.X=1TO 25:P.X;"*";A(X);";";
50  N.X:P.:P."SCALE RANKING
55  X=10
    
```

```

60 IF E=X P.“A-”;X,
65 IF F=X P.“B-”;X,
70 IF G=X P.“C-”;X,
75 IF H=X P.“D-”;X,
80 IF X=0 G.90
85 X=X-.1:G.60
90 IN.“WHICH WEIGHT”;W
95 IF W>25 G. 135
100 IF A(W)=0 G.135
105 IN.“WHICH SCALE”;S
110 F.X=1TO390:N.X
115 IF S=A G.140
120 IF S=B G.145
125 IF S=C G.150
130 IF S=D G.155
135 P.“DON’T GET CUTE, PAL!”:G.90
140 E=E+A(W)::G.160
145 F=F+A(W):G.160
150 G=G+A(W):G.160
155 H=H+A(W)
160 A(W)=0
165 IF E>(F+5) G.250
170 IFF>(E+5) G.250
175 IF G>(H+5) G.250
180 IF H>(G+5) G.250
185 I=E+F:J=G+H
190 IF I>(J+7.5) G.250
195 IF J>(I+7.5) G.250
200 Y=0:F.X=1TO 25
205 Y=Y+A(X):N.X
210 IF Y=0 G.225
215 G.35
225 P.“THAT WAS THE LAST WEIGHT”:P.
230 F.X=1TO333:N.X
235 P.“YOU WIN!”
240 END
250 PRINT“SCALE STRUCTURE TOPPLES!”
255 F.X=1TO333:N.X
260 P.“YOU LOSE!”
265 END

```

Summary Of Variables Used

A,B,C,D	Scale positions
E	Weight of scale A
F	Weight of scale B
G	Weight of scale C

- H Weight of scale D
- I Weight of crossbar AB
- J Weight of crossbar CD
- S Scale choice
- W Weight choice
- X Timing
- Y Various

Sample Run

BALANCING THE SCALES

THE AVAILABLE WEIGHTS ARE 1*3.32*73*2.9

4*5.55*86*7.17*7.78*9.69*6.210*1.7
 11*8.112*1.913*414*3.815*6.516*8.8
 17*9.418*2.619*720*4.221*7.922*5.3
 23*124*0.725*6.2

SCALE RANKING

A-0 B-0 C-0 D-0

WHICH WEIGHT? 1

WHICH SCALE? A—

THE AVAILABLE WEIGHTS ARE 1*0*7.3*2.9

4*5.55*86*7.17*7.78*9.69*6.210*1.7
 11*8.112*1.913*414*3.815*6.516*8.8
 17*9.418*2.619*720*4.221*7.922*5.3
 23*124*0.725*6.2

SCALE RANKING

A-3.3 B-0 C-0 D-0

WHICH WEIGHT? 1

DON'T GET CUTE, PAL!

WHICH WEIGHT? 3

WHICH SCALE? C—

THE AVAILABLE WEIGHTS ARE...

(from here on I won't bother repeating the weights. When a weight is used it becomes 0.)

SCALE RANKING

A-3.3 C-2.9 B-0 D-0

WHICH WEIGHT? 4

WHICH SCALE? B—

THE AVAILABLE WEIGHTS ARE ...

SCALE RANKING

B--5.5 A--3.3 C--2.9 D--0

WHICH WEIGHT? 2

WHICH SCALE? D

THE AVAILABLE WEIGHTS ARE

...

SCALE RANKING

D--7 B--5.5 A--3.3 C--2.9

WHICH WEIGHT? 5

WHICH SCALE? A

SCALE STRUCTURE TOPPLES!

YOU LOSE!

Memory Test

You have to keep alert to win this two-player game. There are 10 rounds. On each round three numbers are separately flashed on the screen for a brief period of time (which gets progressively shorter on each round). If you want to alter the flash time, change the value of C in line 35.

After displaying the three numbers, the computer asks one of the players (randomly selected) to repeat one of the numbers. Then the other player is asked for one of the remaining numbers. A correct entry by either player scores a point for that player.

After 10 rounds the total score for both players is displayed and the game ends.

In the easy level the numbers can be up to three digits. The medium level numbers go up to four digits, and the hard numbers have up to five digits.

Since this is a memory game, there would be no point in including a sample run. See Fig. 1-10 for the flowchart.

Standard BASIC

```
5 PRINT:PRINT:PRINT  
10 PRINT "", "MEMORY TEST":PRINT  
12 PRINT "PLAYER #1";  
15 INPUT A$  
17 PRINT "PLAYER #2";  
20 INPUT B$  
25 LET A=0:LET B=0  
30 LET E=10:LET M=100:LET H=1000  
35 LET C=150  
40 PRINT "EASY, MEDIUM OR HARD GAME ";  
45 INPUT G  
50 REM* BEGIN THE PLAY*  
55 FOR K=1 TO 10  
57 LET X=INT(RND(0)*100)*INT(RND(0)*G)  
60 LET Y=INT(RND(0)*100)*INT(RND(0)*G)  
62 LET Z=INT(RND(0)*100)*INT(RND(0)*G)  
65 PRINT "ENTER 1 WHEN READY"  
70 INPUT J  
72 REM* FLASH NUMBERS*  
75 LET M=INT(RND(0)*888)+1  
77 FOR N=1 TO M:NEXT N  
80 FOR N=1 TO 50:PRINT:NEXT N
```

```
82 PRINT X
85 FOR D=1 TO C:NEXT D
87 FOR N=1 TO 50:PRINT:NEXT N
90 PRINT Y
92 FOR D=1 TO C:NEXT D
95 FOR N=1 TO 50:PRINT:NEXT N
97 PRINT Z:M=M+100
100 FOR D=1 TO C:NEXT D
105 FOR N=1 TO M:PRINT:NEXT N
110 LET T=INT(RND(0)*2)
115 IF T=1 THEN GOTO 155
120 PRINT A$;" WHAT WAS THE ";
122 LET F=0
125 LET H=INT(RND(0)*3)+1
130 IF H=1 THEN GOSUB 200
135 IF H=2 THEN GOSUB 220
140 IF H=3 THEN GOSUB 240
145 LET A=A+F
150 IF T=1 THEN GOTO 195
155 PRINT B$;" WHAT WAS THE ";
157 LET F=0
160 LET I=INT(RND(0)*3)+1
165 IF I=H THEN GOTO 160
170 IF I=1 THEN GOSUB 200
175 IF I=2 THEN GOSUB 220
180 IF I=3 THEN GOSUB 240
185 LET B=B+F
190 IF T=1 THEN GOTO 120
195 NEXT K
197 GOTO 275
200 PRINT "FIRST NUMBER";
202 INPUT W
205 IF W=X THEN GOTO 260
210 PRINT "NO, IT WAS ";X
215 RETURN
220 PRINT "SECOND NUMBER";
222 INPUT W
225 IF W=Y THEN GOTO 260
230 PRINT "NO, IT WAS ";Y
235 RETURN
240 PRINT "THIRD NUMBER";
242 INPUT W
245 IF W=Z THEN GOTO 260
250 PRINT "NO, IT WAS ";Z
255 RETURN
260 PRINT "CORRECT"
```

```
265 LET F=1
270 RETURN
275 PRINT“GAME IS OVER”
280 PRINT A$;“ GOT”;A;“RIGHT”
285 PRINT“AND ”;B$;“ GOT”;B
290 END
```

TRS-80 BASIC

```
10 CLS:P.:P.“”,“MEMORY TEST”:P.
15 IN.“PLAYER #1”;A$
20 IN.“PLAYER #2”;B$
25 E=10:M=100:H=1000
30 IN.“EASY, MEDIUM OR HARD GAME”;G
35 A=0:B=0:C=150
50 F.K=1TO10
55 X=RND(100)*RND(G)
60 Y=RND(100)*RND(G)
65 Z=RND(100)*RND(G)
70 IN.“ENTER 1WHEN READY”;J
75 M=RND(888):F.N=1TOM:N.N
80 CLS:P.X
85 F.D=1TO C:N.D
90 CLS:P.Y
95 F.D=1TO C:N.D
100 CLS:P.Z
105 F.D=1TO C:N.D:C=C-10
110 CLS:T=RND(2):F.N=1TOM:N.N
115 IF T=1 G.155
120 P.A$;“WHAT WAS THE”;
125 H=RND(3):F=0
130 IF H=1 GOS.200
135 IF H=2 GOS. 220
140 IF H=3 GOS. 240
145 A=A+F
150 IFT=1 G.195
155 P.B$;“WHAT WAS THE”;
160 I=RND(3):F=0
165 IF I=H G. 160
170 IF I=1 GOS.200
175 IF I=2 GOS.220
180 IF I=3 GOS. 240
185 B=B+F
190 IFT=1 G.120
195 N.K:G.275
200 IN.“FIRST NUMBER”;W
205 IF W=X G.260
```

```
210 P.“NO, IT WAS”;X
215 RET.
220 IN.“SECOND NUMBER”;W
225 IF W=Y G. 260
230 P.“NO, IT WAS”;Y
235 RET.
240 IN.“THIRD NUMBER”;W
245 IF W=Z G.260
250 P.“NO, IT WAS”;Z
255 RET.
260 P.“CORRECT”:F=1
265 RET.
275 P.“GAME IS OVER”
280 P.A$;“ GOT”;A;“ RIGHT AND ”;B$;“ GOT”;B
285 END
```

Summary of Variables Used

A\$	Player #1
B\$	Player #2
A	A\$ score
B	B\$ score
C	Flash timing
D	Flash timing
E	Easy game
F	Correct guess?
G	Game level
H	Hard game/A\$ number selection
I	B\$ number selection
J	Round start
K	Round counter
M	Medium game/Timing
N	Timing
T	Player select
W	Guess
X	First number
Y	Second number
Z	Third number

What Comes Next

If you enjoy solving mathematical puzzles, this game is for you. The computer generates a sequence of numbers in the pattern of $A = ((A+B)*C) - D$. It gives you the first three numbers in the sequence and you have to figure out what the next number will be. If you guess wrong the computer will give you the correct number, and you must determine the next step in the sequence. Once you correctly identify a number, the game moves on to the next sequence. If you have no idea what the next number might be you can enter "E" to pass. An E counts as a half try.

The computer keeps track of how many tries it takes you to get through ten sequences. Obviously, the lower your score, the better. See Fig. 1-11 for the flowchart.

Standard BASIC

```
5 LET S=0
10 FOR X=1 to 10
12 LET T=1
14 LET A=INT(RND(0)*10)+1
16 LET B=INT(RND(0)*100)+1
18 LET Z=INT(RND(0)*10)+1
20 IF Z>6 THEN LET B=0
22 LET C=INT(RND(0)*50)+1
24 LET Z=INT(RND(0)*10)+1
26 IF Z>4 THEN LET C=1
28 LET D=INT(RND(0)*25)+1
30 LET Z=INT(RND(0)*10)+1
35 IF Z>3 THEN LET D=1
40 REM* DISPLAY INITIAL SERIES*
45 FOR Y=1 TO 3
50 PRINT A,
55 LET A=((A+B)*C)-D
60 NEXT Y
65 PRINT
70 REM* PLAYER'S GUESS *
72 LET E=0.5
75 PRINT "WHAT COMES NEXT? (ENTER E TO PASS)";
77 INPUT F
80 IF F=E THEN GOTO 95
85 IF F=A THEN GOTO 110
90 PRINT "INCORRECT",
92 LET T=T+0.5
```

```

95 LET T=T+0.5
97 LET A=((A+B)*C)-D
100 PRINT A
105 GOTO 70
110 PRINT“CORRECT!”,T;“TRIES”
115 LET S=S+T
120 PRINT“YOUR SCORE SO FAR IS”;S
125 NEXT X
130 PRINT“GAME OVER”
135 IF S<13 THEN GOTO 160
140 IF S<20 THEN GOTO 170
145 IF S>30 THEN GOTO 180
150 END
160 PRINT“FANTASTIC!”
165 END
170 PRINT“VERY GOOD”
175 END
180 PRINT“FRANKLY, YOU DID LOUSY”
185 END

```

TRS-80 BASIC

```

10 S=0
15 F.X=1TO10:T=1
20 A=RND(10):B=RND(100):Z=RND(10)
25 IF Z>6 THEN B=0
30 C=RND(50):Z=RND(10)
35 IF Z>4 THEN C=1
40 D=RND(25):Z=RND(10)
45 IF Z>3 THEN D=1
50 F.Y=1TO3
55 P.A,
60 A=((A+B)*C)-D
65 N.Y
70 P.:E=0.5
75 IN.“WHAT COMES NEXT? (ENTER E TO PASS)”;F
80 IF F=E G.95
85 IF F=A G.110
90 P.“INCORRECT”,:T=T+0.5
95 T=T+0.5:A=((A+B)*C)-D
100 P.A:G.70
110 P.“CORRECT”,T;“TRIES”
115 S=S+T:P.“YOUR SCORE SO FAR IS”;S
120 N.X
125 IF S<13 G.150
130 IF S<20 G.160
135 IF S>30 G.170

```

```

145 END
150 P."FANTASTIC!"
155 END
160 P."VERY GOOD"
165 END
170 P."FRANKLY, YOU DID LOUSY."
175 END

```




Fig. 1-11. Flowchart for What Comes Next.

Sample Run (excerpt)

```
7 99      111
WHAT COMES NEXT? (ENTER E TO PASS)? 222222
INCORRECT    11143
WHAT COMES NEXT? (ENTER E TO PASS)? 111595
CORRECT!    2 TRIES
YOUR SCORE SO FAR IS 2
3 2 1
WHAT COMES NEXT? (ENTER E TO PASS)? 0
CORRECT!    1 TRIES
YOUR SCORE SO FAR IS 3
9 17 33
WHAT COMES NEXT? (ENTER E TO PASS)? 66
INCORRECT    65
WHAT COMES NEXT? (ENTER E TO PASS)? E
129
WHAT COMES NEXT? (ENTER E TO PASS)? 357
CORRECT!    2.5 TRIES
YOUR SCORE SO FAR IS 5.5
10 968 10538
WHAT COMES NEXT? (ENTER E TO PASS)? 11428
INCORRECT    106238
WHAT COMES NEXT? (ENTER E TO PASS)? E
1063238
WHAT COMES NEXT? (ENTER E TO PASS)? 1063328
CORRECT!    2.5 TRIES
YOUR SCORE SO FAR IS 8
```

Summary of Variables Used

A-D	SEQUENCE VARIABLES
E	PASS
F	PLAYER'S GUESS
S	TOTAL SCORE
T	ROUND SCORE
X	ROUND COUNT
Y	TIMING
Z	VARIOUS

Go Fish

Here's a computerized version of a popular card game. The "deck" consists of up to four each of 16 different card types, identified by the letters A through P. The object is to make as many matches as possible.

On each play you are shown the cards in your hand and the number of cards in the computer's hand. You first make any matches you can (and the computer does likewise); then you request a card. If the computer has it, the card is exchanged. If not, you take the top card from the main deck. Then the computer takes a similar turn (it does not examine your hand until after it has made its request, so it doesn't cheat).

The game is over when either of the players has no cards in hand, or when the main deck runs out of cards. The winner is the player with the highest number of matches. All card values count the same. See Fig. 1-12 for the flowchart.

Standard BASIC

```
3 FOR X=1 TO 302
5 PRINT
7 LET A(X)=0
10 NEXT X
12 PRINT "", "GO FISH":PRINT
15 FOR X=1 TO 4
17 FOR Y=1 TO 16
20 LET Z=INT(RND(0)*100)+201
22 IF A(Z)>0 THEN GOTO 20
25 LET A(Z)=Y
27 NEXT Y
30 NEXT X
31 LET Q=200
32 LET A=1:LET B=2
33 LET C=3:LET D=4
34 LET E=5:LET F=6
35 LET G=7:LET H=8
36 LET I=9:LET J=10
37 LET K=11:LET L=12
38 LET M=13:LET N=14
39 LET O=15:LET P=16
40 REM*OPENING DEAL*
42 FOR X=1 TO 7
45 LET Q=Q+1
46 IF A(Q)=0 THEN GOTO 45
47 LET A(X)=A(Q):LET A(Q)=0
```

```
48 NEXT X
50 FOR X=101 TO 107
51 LET Q=Q+1
52 IF A(Q)=0 THEN GOTO 51
53 LET A(X)=A(Q):LET A(Q)=0
54 NEXT X
55 REM*THE PLAY*
57 PRINT:PRINT"YOUR HAND---";
60 FOR X=1 TO 100:LET Y=A(X)
62 IF Y>0 THEN GOSUB 240
64 NEXT X
66 PRINT:PRINT:PRINT"MY HAND---";
68 FOR X=101 TO 200
70 LET Y=A(X)
72 IF Y=0 THEN GOTO 80
74 LET Z=X-100
76 PRINT Z;".)X";
78 NEXT X
80 PRINT:PRINT
82 GOTO 335
85 PRINT"WHAT DO YOU NEED";
87 INPUT S
90 FOR X=101 TO 200
92 LET Y=A(X)
94 IF Y=S THEN LET T=X
96 NEXT X
98 IF T>0 THEN GOTO 470
100 PRINT"GO FISH!"
102 FOR Z=1TO 470:NEXT Z
104 PRINT:PRINT
106 LET Y=A(Q)
108 IF Y=0 THEN GOSUB 530
110 LET A(Q)=0:LET X=1:LET Q=Q+1
112 IF A(X)=0 THEN GOTO 120
115 LET X=X+1
117 GOTO 112
120 LET A(X)=Y
122 PRINT"YOU DREW A";
124 GOSUB 245
126 PRINT:PRINT
128 GOSUB 330
130 PRINT"I NEED A ";
132 LET X=200
135 IF A(X)>0 THEN GOTO 145
140 LET X=X-1
142 GOTO 135
```




Fig. 1-12. Flowchart for Go Fish.

```
145 LET Y=A(X)
146 GOSUB 245
148 PRINT
150 LET T=0
152 FOR X=1 TO 100
154 LET S=A(X)
156 IF S=Y THEN LET T=X
158 NEXT X
160 GOSUB 330
165 IF T>0 THEN GOTO 490
167 PRINT "I MUST GO FISH"
170 LET Y=A(Q):LET X=101
172 IF Y=0 THEN GOSUB 530
174 LET A(Q)=0:LET Q=Q+1
176 IF A(X)=0 THEN GOTO 185
178 LET X=X+1
180 GOTO 176
185 LET A(X)=Y
190 GOSUB 330
195 GOSUB 330
200 GOTO 55
240 PRINT X;" ";
245 IF Y=1 THEN PRINT "A ";
250 IF Y=2 THEN PRINT "B ";
255 IF Y=3 THEN PRINT "C ";
260 IF Y=4 THEN PRINT "D ";
265 IF Y=5 THEN PRINT "E ";
270 IF Y=6 THEN PRINT "F ";
275 IF Y=7 THEN PRINT "G ";
280 IF Y=8 THEN PRINT "H ";
285 IF Y=9 THEN PRINT "I ";
290 IF Y=10 THEN PRINT "J ";
295 IF Y=11 THEN PRINT "K ";
300 IF Y=12 THEN PRINT "L ";
305 IF Y=13 THEN PRINT "M ";
310 IF Y=14 THEN PRINT "N ";
315 IF Y=15 THEN PRINT "O ";
320 IF Y=16 THEN PRINT "P ";
325 RETURN
330 FOR Z=1 TO 555:NEXT Z
333 RETURN
335 LET Y=50:LET N=14
337 PRINT "CAN YOU MATCH ANY PAIRS";
340 INPUT U
342 IF U=N THEN GOTO 380
344 PRINT "CARD#";
```

```
346 INPUT S
348 PRINT"AND CARD#";
350 INPUT T
352 IF A(S)= THEN GOTO 360
354 IF S=T THEN GOTO 360
356 IF A(S)=A(T) THEN GOTO 365
360 PRINT"INVALID MATCH"
362 GOTO 335
365 LET Y=A(S):LET A(S)=0
367 LET A(T)=0:LET T=A(301)
370 LET T=T+1:LET A(301)=T
372 PRINT"YOU'VE MATCHED A PAIR OF";
374 GOSUB 245
376 GOTO 335
380 FOR X=101 TO 200:LET R=A(X)
382 IF R=0 THEN GOTO 400
385 FOR Z=101 TO 200
387 IF Z=X THEN GOTO 395
390 LET S=A(Z)
392 IF R=S THEN GOSUB 430
395 NEXT Z
400 NEXT X
402 IF Q>275 THEN GOTO 465
405 LET W=0
407 FOR X=1 TO 100:LET W=W+A(X)
410 NEXT X
412 IF W=0 THEN GOTO 440
415 LET W=0
417 FOR X=101 TO 200:LET W=W+A(X)
420 NEXT X
422 IF W=0 THEN GOTO 460
425 LET T=0
427 GOTO 85
430 LET Y=R
432 PRINT"I'VE MATCHED A PAIR OF";
434 GOSUB 245
436 LET R=-5:LET A(X)=0:LET A(Z)=0
438 LET T=A(302):LET T=T+1:LET A(302)=T
439 RETURN
440 PRINT"YOU ARE OUT OF CARDS"
442 LET X=A(301):LET Y=A(302)
444 PRINT"YOU MATCHED";X;"PAIRS"
446 PRINT"AND I MATCHED";Y;"PAIRS"
450 END
460 PRINT"I AM OUT OF CARDS"
462 GOTO 442
```

```
465 PRINT "THE DECK IS OUT OF CARDS"
467 GOTO 442
470 LET A(T)=0
472 FOR X=1 TO 100
475 IF A(X)=0 THEN GOTO 485
480 NEXT X:PRINT:PRINT
482 GOSUB 330
484 GOTO 130
485 LET A(X)=S:LET S=0
487 GOTO 480
490 LET A(T)=0
495 FOR X=1 TO 100
500 IF A(X)=0 THEN GOTO 515
505 NEXT X
507 GOSUB 330
510 GOTO 70
515 LET A(X)=S
520 LET S=0
525 GOTO 505
530 LET Q=Q+1
535 LET Y=A(Q)
540 IF Y=0 THEN GOTO 530
545 RETURN
```

TRS-80 BASIC

```
10 CLS:P.:P."","GO FISH":F.X=1TO302
15 A(X)=0:N.X
20 F.X=1TO4:F.Y=1TO16
25 Z=RND(100)+200
30 IF A(Z)>0 G.25
35 A(Z)=Y:N.Y:N.X:Q=200
40 F.X=1TO7
42 Q=Q+1
44 IF A(Q)=0 G.42
46 A(X)=A(Q):A(Q)=0:N.X
48 F.X=101 TO 107
50 Q=Q+1
52 IF A(Q)=0 G.50
54 A(X)=A(Q):A(Q)=0:N.X
55 CLS:P."YOUR HAND---";:F.X=1 TO 100:Y=A(X)
60 IF Y>0 GOS.240
65 N.X:P.:P."MY HAND---";
67 F.X=101TO200:Y=A(X)
70 IF Y=0 G.80
75 Z=X-100:P.Z;"")X";
80 N.X:P.:P.:G.335
```

85 IN.“WHAT DO YOU NEED”;S
90 F.X=101 TO 200:Y=A(X)
92 IF Y=S THEN T=X
95 N.X:IF T>0 G.470
100 P.“GO FISH!”:F.Z=1TO470:N.Z:P.:P.
102 Y=A(Q)
105 IF Y=0 GOS.530
107 A(Q)=0:X=1:Q=Q+1
110 IF A(X)=0 G.120
115 X=X+1:G.110
120 A(X)=Y:P.“YOU DREW A ”:GOS.245
125 P.:P.:GOS.330
130 P.“I NEED A ”:X=200
135 IF A(X)>0 G.145
140 X=X-1:G.135
145 Y=A(X):GOS.245:P.:P.
150 T=0:F.X=1TO100:S=A(X)
155 IF S=Y THEN T=X
160 N.X:GOS.330
165 IF T>0 G.490
170 P.“I MUST GO FISH”:Y=A(Q)
172 IF Y=0 GOS.530
176 A(Q)=0:X=101:Q=Q+1
178 IF A(X)=0 G.185
180 X=X+1:G.178
185 A(X)=Y:GOS.330
190 G.55
240 P.X;“.”;
245 IF Y=1 P.“A ”;
250 IF Y=2 P.“B ”;
255 IF Y=3 P.“C ”;
260 IF Y=4 P.“D ”;
265 IF Y=5 P.“E ”;
270 IF Y=6 P.“F ”;
275 IF Y=7 P.“G ”;
280 IF Y=8 P.“H ”;
285 IF Y=9 P.“I ”;
290 IF Y=10 P.“J ”;
295 IF Y=11 P.“K ”;
300 IF Y=12 P.“L ”;
305 IF Y=13 P.“M ”;
310 IF Y=14 P.“N ”;
315 IF Y=15 P.“O ”;
320 IF Y=16 P.“P ”;
325 RET.
330 F.Z=1TO555:N.Z:RET.

```

335 Y=50:N=14:IN.“CAN YOU MAKE ANY PAIRS”;U
340 IF U=N G.380
345 IN.“CARD#”;S:IN.“AND CARD#”;T
350 IF A(S)=0 G.360
352 IF S=T G.360
355 IF A(S)=A(T) G.365
360 P.“INVALID MATCH”:G.335
365 Y=A(S):A(S)=0:A(T)=0:T=A(301):T+1:A(301)=T
370 P.“YOU’VE MATCHED A PAIR OF ”;:GOS.245:G.335
380 F.X=101TO200:R=A(X):IF R=0 G.400
385 F.Z=101TO200:IF X=Z G.395
390 S=A(Z):IF R=S GOS.430
395 N.Z
400 N.X:IF Q > 275 G.465
405 W=0:F.X=101TO200:W=W+A(X):N.X:T=0
410 IF W=0 G.460
415 W=0:F.X=1TO100:W=W+A(X):N.X
420 IF W=0 G.440
425 G.85
430 Y=R:P.“I’VE MATCHED A PAIR OF ”;:GOS.245
435 R=-5:A(X)=0:A(Z)=0:T=A(302):T=T+1
437 A(302)=T:P.:RET.
440 P.“YOU ARE OUT OF CARDS”
445 X=A(301):Y=A(302):P.“YOU MATCHED ”;X;“PAIRS”
450 P.“AND I MATCHED ”;Y;“PAIRS”
455 END
460 P.“I AM OUT OF CARDS”:G.445
465 P.“THE DECK IS OUT OF CARDS”:G.445
470 A(T)=0:F.X=1TO100:IF A(X)=0 G.485
480 N.X:P.:P.:GOS.330:G.165
485 A(X)=S:S=0:G.480
490 A(T)=0:F.X=1TO100:IF A(X)=0 G.510
500 N.X:GOS.330:G.55
510 A(X)=S:S=0:G.500
530 Q=Q+1:Y=A(Q)
535 IF Y=0 G.530
540 RET.

```

Sample Run

GO FISH

YOUR HAND --- 1.)0 2.)M 3.)J 4.)C 5.)K
 6.)G 7.)A

MYHAND--- 1.)X 2.)X 3.)X 4.)X 5.)X
 6.)X 7.)X

CAN YOU MAKE ANY PAIRS? NO

I’VE MATCHED A PAIR OF C

I'VE MATCHED A PAIR OF H
WHAT DO YOU NEED? O
GO FISH!
YOU JUST DREW A D
I NEED A K

YOUR HAND --- 1.)O 2.)M 3.)J 4.)C 6.)G 7.)A 8.)D
MY HAND --- 1.)X 2.)X 3.)X 4.)X 6.)X
CAN YOU MAKE ANY PAIRS? NO
I'VE MATCHED A PAIR OF K
WHAT DO YOU NEED? M
I NEED A B
I MUST GO FISH

YOUR HAND --- 1.)O 2.)M 3.)J 4.)C 5.)M
6.)G 7.)A 8.)D
MY HAND --- 1.)X 4.)X

CAN YOU MAKE ANY PAIRS? YES
CARD#? 2
ANY CARD #? 5
YOU'VE MATCHED A PAIR OF M
WHAT DO YOU NEED? J
I NEED A B
I MUST GO FISH

YOUR HAND --- 1.)O 2.)J 3.)J 4.)C 6.)G
7.)A 8.)D
MY HAND --- 1.)X 4.)X

Summary of Variables Used

A-P	CARD VALUES
Q	LOCATION IN DECK
R	PAIR MATCHING
S	PLAYER'S REQUEST/PAIR MATCHING
T	OPPONENT'S HAND SEARCH/PAIR MATCHING
U	MATCH ANY PAIRS?
V	NOT USED
W	CARD COUNT
X-Z	VARIOUS

Quiz Whiz

Quiz Whiz is a simple game in concept. The computer gives you a number (X) and four larger numbers. You have to determine which one is the square of X (X times X). There are ten rounds to a game.

In the easy game X can be any whole number from 1 to 50. The medium game allows half numbers (such as 37.5), and the hard game allows any digit behind the decimal point (39.7, 42.3, 16.8, etc. . .). Of course you can change the over-all complexity of the game by altering the value of X in line 30. As written X = 50 times L. If you substituted "X=40*L", the maximum number is 40 (the L factor is divided back out in a later step; it is used to provide the fractional quantities in the medium and hard games. In the easy game, L = 1).

If you can get a score of 10 without using a calculator or slide rule, you're really a whiz kid. See Fig. 1-13 for the flowchart.

Standard BASIC

```
10 PRINT "", "WHIZ QUIZ"
12 PRINT
14 LET E=1
16 LET M=2
18 LET H=10
20 LET S=0
22 PRINT"EASY, MEDIUM, OR HARD GAME";
24 INPUT L
26 REM* THE GAME *
28 FOR T=1 TO 10
30 LET X=50*L
32 LET Y=INT(RND(0)*X)+1
34 LET X=Y/L
36 LET Y=X*X
38 REM* X IS THE ROOT AND Y IS THE CORRECT ANSWER*
40 FOR M=1 TO 333
42 NEXT M
44 PRINT"What is ";X;" SQUARED?"
46 LET A=0
48 LET B=0
50 LET C=0
52 LET D=0
54 REM*PLANT CORRECT ANSWER *
57 LET E=INT(RND(0)*4)+1
60 IF E=1 THEN LET A=Y
```




Fig. 1-13. Flowchart for Whiz Quiz.

```

65 IF E=2 THEN LET B=Y
70 IF E=3 THEN LET C=Y
75 IF E=4 THEN LET D=Y
80 REM * WRONG ANSWERS *
82 IF A>0 THEN GOTO 95
85 GOSUB 220
90 LET A=H
95 IF B>0 THEN GOTO 110
100 GOSUB 220
105 LET B=H
110 IF C>0 THEN GOTO 125
115 GOSUB 220
120 LET C=H
125 IF D>0 THEN GOTO 140
130 GOSUB 220
135 LET D=H
137 REM*PLAYER'S CHOICE *
140 PRINT"A --- ";A
145 PRINT"B --- ";B
150 PRINT"C --- ";C
155 PRINT"D --- ";D
160 INPUT K
162 FOR M=1 TO 333
165 NEXT M
170 IF K=Y THEN GOTO 200
175 PRINT"WRONG. THE CORRECT ANSWER IS ";Y
180 NEXT T
185 PRINT"THE WHIZ QUIZ IS OVER"
190 PRINT"YOU GOT ";S;" RIGHT"
195 END
200 REM* CORRECT ANSWER*
205 PRINT"RIGHT!"
210 LET S=S+1
215 GOTO 180
220 REM* WRONG ANSWER SELECTION *
225 LET F=INT(RND(0)*0.4*X)+1
230 LET G=INT(RND(0)*F)+1 + X + (F/2)
235 IF G=X THEN GOTO 230
240 LET H =G*G
245 IF L=1 THEN GOTO 260
250 LET G=INT(RND(0)*10)+1
255 LET H=H+(G/L)
260 RETURN

```

TRS-80 BASIC

```

10 P. "", "WHIZQUIZ":P.
15 E=1:M=2:H=10:S=0

```

20 IN."EASY, MEDIUM, OR HARD GAME";L
25 F.T=1TO10
30 X=50*L;Y=RND(X)
35 X=Y/L;Y=X*X
40 F.M=1TO333:N.M
45 P."WHAT IS ";X;"SQUARED?"
50 A=0:B=0:C=0:D=0
55 E=RND(4)
60 IF E=1 THEN A=Y
65 IF E=2 THEN B=Y
70 IF E=3 THEN C=Y
75 IF E=4 THEN D=Y
80 IF A>0 G.95
85 GOS.220
90 A=H
95 IF B>0 G. 110
100 GOS.220
105 B=H
110 IF C>0 G. 125
115 GOS.220
120 C=H
125 IF D>0 G. 140
130 GOS.220
135 D=H
140 P."A---";A
145 P."B---";B
150 P."C---";C
155 P."D---";D
160 IN.K
165 F.M=1TO33:N.M
170 IF K=Y G.200
175 P."WRONG. THE CORRECT ANSWER IS ";Y
180 N.T
185 PL"THE WHIZ QUIZ IS OVER."
190 P."YOU GOT";S;"RIGHT"
195 END
200 P."RIGHT!"
205 S=S+1
210 G.180
220 F=INT(.4*X)
225 G=RND(F)+(X-F/2)
230 IF G=X G. 225
235 H=G*G
240 IF L=1 G. 250
245 H=H+(RND(10)/L)
250 RET.

Variables Used

- A POSSIBLE ANSWER
- B POSSIBLE ANSWER
- C POSSIBLE ANSWER
- D POSSIBLE ANSWER
- E EASY GAME/CORRECT ANSWER SELECTION
- F FALSE ANSWER CALCULATIONS
- G FALSE ANSWER CALCULATIONS
- H FALSE ANSWER CALCULATIONS/HARD GAME
- K PLAYER'S ANSWER
- L GAME LEVEL
- M MEDIUM GAME/TIMING
- S SCORE
- T TURN COUNTER
- X ROOT SELECTION
- Y CORRECT ANSWER

Sample Run (Excerpt)

QUIZ WHIZ

WHAT IS 17 SQUARED?

- A---256
- B---289
- C---324
- D---196

?A

WRONG. THE CORRECT ANSWER IS 289

WHAT IS 32 SQUARED?

- A---941
- B---900
- C---1225
- D---1024

?D

RIGHT!

WHAT IS 50 SQUARED?

- A---3249
- B---2304
- C---2500
- D---2025

?C

RIGHT!

Tic Tac Toe

Here is a computerized version of that old favorite. The object, of course, is to get three in a row: across, down or diagonally.

It's not easy to beat the computer, but anything's possible. See Fig. 1-14 for the flowchart.

Standard BASIC

```
10 PRINT" LET'S PLAY TIC-TAC-TOE"
12 FOR X=1 TO 9
14 LET A(X)=0
16 NEXT X
18 REM ** THE BOARD IS NOW CLEARED **
20 GOSUB 280
25 PRINT"YOU TAKE THE 'O' AND I'LL TAKE THE 'X'"
27 PRINT:PRINT
30 LET N=INT (RND(0)*2)+1
32 LET M=3
35 GOTO 300
40 IF N=2 THEN GOTO 290
45 PRINT"I'LL GO FIRST THIS TIME"
50 GOSUB 280
55 PRINT'I'LL TAKE SPACE # ";
60 GOSUB 280
65 LET S=INT (RND(0)*9)+1
67 LET A(S)==-1: LET M=1
70 PRINT S
75 GOSUB 280
80 GOTO 300
85 PRINT "WHICH SPACE WILL YOU PLAY";
87 INPUT S
90 LET S=INT(S)
92 LET S=ABS(S): LET M=2
95 IF S>9 THEN GOTO 85
100 100 IF S<1 THEN GOTO 85
105 IF A(S)<0 THEN GOTO 85
110 LET A(S)==-2
115 GOTO 300
120 PRINT "I'LL PAY SPACE # ";
125 LET S=0 : LET M=1
127 REM ** BOARD CHECK FOR COMPUTER'S PLAY **
130 IF A(1)=0 THEN GOTO 147
132 IF A(1)=A(2) THEN LET S=3
```

```
135 IF A(1)=A(3) THEN LET S=2
137 IF A(1)=A(5) THEN LET S=9
140 IF A(1)=A(9) THEN LET S=5
142 IF (1)=A(4) THEN LET S=7
145 IF A(1)=A(7) THEN LET S=4
147 IF A(2)=0 THEN GOTO 157
150 IF A(2)=A(3) THEN LET S=1
152 IF A(2)=A(5) THEN LET S=8
155 IF A(2)=A(8) THEN LET S=5
157 IF A(3)=0 THEN GOTO 170
160 IF A(3)=A(5) THEN LET S=7
162 IF A(3)=A(7) THEN LET S=5
165 IF A(3)=A(6) THEN LET S=9
167 IF A(3)=A(9) THEN LET S=6
170 IF A(4)=0 THEN GOTO 180
172 IF A(4)=A(7) THEN LET S=1
175 IF A(4)=A(5) THEN LET S=6
177 IF A(4)=A(6) THEN LET S=5
180 IF A(5)=0 THEN GOTO 192
182 IF A(5)=A(6) THEN LET S=4
185 LET A(5)=A(9) THEN LET S=1
187 IF A(5)=A(8) THEN LET S=2
190 IF A(5)=A(7) THEN LET S=3
192 IF A(6)=0 THEN GOTO 197
195 IF A(6)=A(9) THEN LET S=3
197 IF A(7)=0 THEN GOTO 210
200 IF A(7)=A(8) THEN LET S=9
205 IF A(7)=A(9) THEN LET S=8
210 IF A(8)=0 THEN GOTO 220
215 IF A(8)=A(9) THEN LET S=7
220 IF S>0 THEN GOTO 230
225 LET S=INT (RND(0)*9)+1
230 IF A(S)<0 THEN GOTO 225
235 GOSUB 280
240 PRINT S
245 LET A(S)=-1
250 GOTO 75
280 REM ** PAUSE **
282 FOR X=1 TO 333
285 NEXT X
287 RETURN
290 PRINT "YOU GO FIRST"
295 GOTO 85
300 REM ** DRAW GRID **
302 PRINT:PRINT:PRINT
305 LET Z=1
```

```

307 FOX X=1 TO 3
310 PRINT " ",
312 FOR Y = 1 TO 3
315 LET V=A(Z)
320 IF V<0 THEN GOTO 475
325 PRINT Z;" ";
330 LET Z=Z+1

```




Fig. 1-14. Flowchart for Tic Tac Toe.

332 NEXT Y
335 PRINT
337 NEXT X
340 PRINT:PRINT
345 REM* WIN CHECK **
347 LET Y=0
350 FOR Z=1 TO 9
355 LET X=Z
360 IF A(X)<0 THEN X=0
365 LET Y=Y+X
370 NEXT Z
375 IF Y=0 THEN GOTO 490
380 IF A(1)=0 THEN GOTO 405
385 LET X=A(1)
390 IF X=A(2) THEN GOSUB 500
395 IF X=A(4) THEN GOSUB 510
400 IF X=A(5) THEN GOSUB 520
405 IF A(2)=0 THEN GOTO 420
410 LET X=A(2)
415 IF X=A(5) THEN GOSUB 530
420 IF A(3)=0 THEN GOTO 440
425 LET X=A(3)
430 IF X=A(6) THEN GOSUB 520
435 IF X=A(5) THEN GOSUB 510
440 IF A(4)=0 THEN GOTO 455
445 LET X=A(4)
450 IF X=A(5) THEN GOSUB 540
455 IF A(7)=0 THEN GOTO 470
460 LET X=A(7)
465 IF X=A(8) THEN GOSUB 520
470 IF M=1 THEN GOTO 85
471 IF M=2 THEN GOTO 120
472 IF M=3 THEN GOTO 40
475 IF V=-1 THEN PRINT "X";
480 IF V=-2 THEN PRINT"0 ";
485 GOTO 330
490 PRINT" TIE GAME — LET'S TRY AGAIN"
495 GOTO 12
500 IF X=A(3) THEN GOTO 550
505 RETURN
510 IF X=A(7) THEN GOTO 550
515 RETURN
520 IF X=A(9) THEN GOTO 550
525 RETURN
530 IF X=A(8) THEN GOTO 550
535 RETURN

```
540 IF X=A(6) THEN GOTO 550
545 RETURN
550 IF X=-1 THEN PRINT "I ";
555 IF X=-2 THEN PRINT "YOU";
560 PRINT "WIN!"
565 END
```

TRS-80 BASIC

```
10 P."LET'S PLAY TIC-TAC-TOE"
15 F.X=1 TO 9:A(X)=0:N.X.
20 GOS.280
25 P."YOU TAKE THE 'O' AND I'LL TAKE THE 'X' ":P.:P.
30 N=RND(2):M=3
35 G.300
40 IF N=2 G. 290
45 P."I'LL GO FIRST THIS TIME"
50 GOS.280
55 P."I'LL TAKE SPACE #";
60 GOS.280
65 S=RND(9):A(S)=-1
70 P.S.:M=1
75 GOS. 280
80 G.300
85 IN."WHICH SPACE WILL YOU PLAY":S
90 S=INT(S):S=ABS(S):M=2
95 IF S>9 G.85
100 IF S<1 G. 85
105 IF A(S)<0 G.85
110 A(S)=-2
115 G .300
120 P."I'LL PLAY SPACE #";
125 S=0:M=1
130 IF A(1)=0 G. 147
132 IF A(1)=A(2) THEN S=3
135 IF A(1)=A(3) THEN S=2
137 IF A(1)=A(5) THEN S=9
140 IF A(1)=A(9) THEN S=5
142 IF A(1)=A(4) THEN S=7
145 IF A(1)=A(7) THEN S=4
147 IF A(2)=0 G. 157
150 IF A(2)=A(3) THEN S=1
152 IF A(2)=A(5) THEN S=8
155 IF A(2)=A(8) THEN S=5
157 IF A(3)=0 G. 170
160 IF A(3)=A(5) THEN S=7
```

162 IF A(3)=A(7) THEN S=5
165 IF A(3)=A(6) THEN S=9
167 IF A(3)=A(9) THEN S=6
170 IF A(4)=0 G. 180
172 IF A(4)=A(7) THEN S=1
175 IF A(4)=A(5) THEN S=6
177 IF A(4)=A(6) THEN S=5
180 IF A(5)=0 G. 192
182 IF A(5)=A(6) THEN S=4
185 IF A(5)=A(9) THEN S=1
187 IF A(5)=A(8) THEN S=2
190 IF A(5)=A(7) THEN S=3
192 IF A(6)=0 G. 197
195 IF A(6)=A(9) THEN S=3
197 IF A(7)=0 G. 210
200 IF A(7)=A(8) THEN S=9
205 IF A(7)=A(9) THEN S=8
210 IF A(8)=0 G. 220
215 IF A(8)=A(9) THEN S=7
220 IF S>0 G. 230
225 S=RND (9)
230 IF A(S)<0 G. 225
235 GOS. 280
240 P.S.
245 A(S)==-1
250 G.75
280 F.X.=1TO333:N.X.
285 RET.
290 P.“YOU GO FIRST”
295 G. 85
300 P.:P.:P.:Z=1
305 F.X=1TO3:P.“ ”,
310 F.Y=1TO3
315 V=A(Z)
320 IF V<0 G. 475
325 P.Z;“ ”;
330 Z=Z+1:N.Y
335 P.:N.X
340 P.:P.
345 Y=0
350 F.Z=1TO9
355 X=Z
360 IF A(X)<0 THEN X=0
365 Y=Y+X
370 N.Z
375 IF Y=0 G. 490

380 IF A(1)=0 G. 405
385 X=A(1)
390 IF X=A(2) GOS. 500
395 IF X=A(4) GOS. 510
400 IF X=A(5) GOS. 520
405 IF A(2) =0 G. 420
410 X=A(2)
415 IF X=A(5) GOS. 530
420 IF A(3)=0 G. 440
425 X=A(3)
430 IF X=A(6) GOS. 520
435 IF X=A(5) GOS. 510
440 IF A(4)=0 G. 455
445 X=A(4)
450 IF X=A(5) GOS. 540
455 IF A(7)=0 G. 470
460 X=A(7)
465 IF X=A(8) GOS. 520
470 IF M=1 G. 85
471 IF M=2 G. 120
472 IF M=3 G. 40
475 IF V=-1 P."X";
480 IF V=-2 P."O";
485 G. 330
490 P."TIE GAME --- LET'S TRY AGAIN"
495 G. 15
500 IF X=A(3) G. 550
505 RET.
510 IF X=A(7) G. 550
515 RET.
520 IF X=A(9) G. 550
525 RET.
530 IF X=A(8) G. 550
535 RET.
540 IF X=A(6) G. 550
545 RET.
550 IF X=-1 P."I";
555 IF X=-2 P."YOU";
560 P."WIN!";
565 END

Summary Of Variables Used

N TURN SELECT
S SPACE PLAYED
V SPACE VALUES
X TIMING/VARIOUS

Y TIE CHECK / BOARD DRAWING
Z SPACE NUMBER

Sample Run

LET'S PLAY TIC-TAC-TOE
YOU TAKE THE 'O' AND I'LL TAKE THE 'X'

1	2	3
4	5	6
7	8	9

YOU GO FIRST

WHICH SPACE WILL YOU PLAY? 5

1	2	3
4	0	6
7	8	9

I'LL PLAY SPACE #3

1	2	X
4	0	6
7	8	9

WHICH SPACE WILL YOU PLAY? 7

1	2	X
4	0	6
0	8	9

I'LL PLAY SPACE #6

1	2	X
4	0	X
0	8	9

WHAT SPACE WILL YOU PLAY? 9

1	2	X
4	0	X
0	8	0

I'LL PLAY SPACE #8

1	2	X
4	0	X
0	X	0


WHAT SPACE WILL YOU PLAY? 1

0	2	X
4	0	X
0	X	0

YOU WIN!!

Chapter 2

Two-Player Games



With the three games in this chapter you can enjoy your computer with a friend. You play against each other and the computer keeps tabs on everything.

To And Fro and Passing Points are designed on a modified board game concept and Money Mad is a light-hearted approach to the stock market.

To And Fro

This is a fairly simple two-player game. The players move along an imaginary board of 100 spaces (the computer does not draw the board, but it continually prints out what spaces the players are on). On each move a player can move from 1 to 6 spaces: the object is to reach the 100th space. But here's the catch; each time you move to a space, the computer tells you its value from -6 to +6. If the value is zero, nothing happens. If the value is positive you are moved ahead that many spaces, and if it is negative you're moved back. The value of the second space you land on is ignored.

The space values are identical for both players, and remain constant throughout the game. See Fig. 2-1 for the flowchart.

Standard BASIC

```
5 FOR X=1TO25:PRINT:NEXT X
10 PRINT",","TO AND FRO":PRINT
12 PRINT"PLAYER #1";
15 INPUT A$
17 PRINT"PLAYER #2";
20 INPUT B$
22 PRINT
25 PRINT"THE OBJECT IS TO MOVE FROM POSITION 1 TO"
30 PRINT"POSITION 100. IF YOU GO PAST 100 YOU WILL"
35 PRINT"LOOP BACK TO THE START. EACH MOVE MAY BE"
40 PRINT"FROM 1 TO 6 SPACES. SOME SPACES WILL SEND"
45 PRINT"YOU FORWARD, OTHERS BACKWARD. THE SPACE"
47 PRINT"VALUE IS THE SAME FOR BOTH PLAYERS. THE"
50 PRINT"FIRST TO LAND ON POSITION 100 WINS."
52 FOR X=1 TO 100
54 LET Y=INT(RND(0)*13)-6
56 LET A(X)=Y
58 NEXT X
60 FOR X=1 TO 20
62 LET Y=INT(RND(0)*100)
64 LET A(Y)=0:NEXT X
66 LET A(1)=0
68 LET A(100)=0
70 LET A=1:LET B=1
75 LET T=INT(RND(0)*2)+1
77 IF T=2 THEN GOTO 120
80 GOSUB 150
85 GOSUB 170
90 LET A=A+M
```

```

95 LET C=A
100 GOSUB 150
105 GOSUB 200
110 LET A=A+M
115 GOSUB 150
120 GOSUB 160
125 GOSUB 170
127 LET B=B+M
130 LET C=B
135 GOSUB 160
140 GOSUB 200

```




Fig. 2-1. Flowchart for To and Fro.

```
142 LET B=B+M
145 GOSUB 160
147 GOTO 85
150 PRINT A$;" IS AT SPACE #";A
155 GOTO 255
160 PRINT B$;" IS AT SPACE #";B
165 GOTO 255
170 PRINT"YOUR MOVE";
172 INPUT M
175 LET M=INT(M)
177 IF M<0 THEN GOTO 170
180 IF M>6 THEN GOTO 170
185 RETURN
200 FOR X=1 TO 470:NEXT X
205 LET M=A(C)
210 IF M=0 THEN GOTO 235
215 IF M<0 THEN GOTO 240
220 PRINT"MOVE AHEAD";M;" SPACES"
225 FOR X=1TO470:NEXT X
227 PRINT
230 RETURN
235 PRINT"BLANK SPACE"
237 GOTO 225
240 LET N=-M
245 PRINT"MOVE BACK ";N;" SPACES"
250 GOTO 225
255 IF A>100 THEN LET A=A- 100
260 IF A=100 THEN GOTO 285
265 IF A<1THEN LET A=1
267 IF B>100 THEN LET B=B- 100
270 IF B=100 THEN GOTO 290
275 IF B<1 THEN LET B=1
280 RETURN
285 PRINT A$;
287 GOTO 295
290 PRINT B$;
295 PRINT" WINS!"
300 END
```

TRS-80 BASIC

```
10 CLS:P.:P." ", "TO AND FRO":P.
15 IN."PLAYER #1";A$
20 IN."PLAYER #2";B$
25 P.:P."THE OBJECT IS TO MOVE FROM POSITION 1 TO
POSITION 100"
```

30 P.“IF YOU GO PAST 100 YOU WILL LOOP BACK TO THE
START.”
35 P.“EACH MOVE MAY BE FROM 1 TO 6 SPACES”
40 P.“SOME SPACES WILL SEND YOU FORWARD, OTHERS”
45 P.“BACKWARD. THE SPACE VALUE IS THE SAME FOR
BOTH”
50 P.“PLAYERS. THE FIRST TO LAND ON POSITION 100 WINS.”
55 F.X=1TO100:Y=RND(13)-7:A(X)=Y:N.X:A(1)=0
60 F.X=1TO20:Y=RND(100):A(Y)=0:N.X:A(100)=0
65 A=1:B=1
70 T=RND(2)
75 IF T=2 G.120
80 GOS.150
85 GOS.170
90 A=A+M
95 C=A
100 GOS.150
105 GOS.200
110 A=A+M
115 GOS.150
120 GOS.160
125 GOS.170
130 B=B+M:C=B:GOS.160
135 GOS.200
140 B=B+M:GOS.160:G.85
150 P.A\$;“ IS AT SPACE #”;A
155 G.255
160 P.B\$;“ IS AT SPACE #”;B
165 G.255
170 IN.“YOUR MOVE”;M
175 M=INT(M):IF M<1 G.170
180 IF M>6 G.170
185 RET.
200 F.X=1TO470:N.X
205 M=A(C)
210 IF M=0 G.235
215 IF M<0 G.240
220 P.“MOVE AHEAD ”;M;“ SPACES”
225 F.X=1TO470:N.X:P.
230 RET.
235 P.“BLANK SPACE”:G.225
240 N=-M
245 P.“MOVE BACK ”;N;“ SPACES”
250 G.225
255 IF A>100 THEN A=A-100
260 IF A=100 G.280

```
262 IF A<1 THEN A=1
265 IF B>100 THEN B=B- 100
267 IF B=100 G-285
270 IF B<1 THEN B=1
275 RET.
280 P.A$;:G.290
285 P.B$;
290 P.“WINS!”
295 END
```

Summary Of Variables Used

A\$	PLAYER #1
B\$	PLAYER #2
A	PLAYER #1'S SPACE
B	PLAYER #2'S SPACE
C	INTERMEDIATE SPACE
M	MOVE
N	BACKWARD MOVE
T	TURN COUNTER
X	TIMING
Y	VALUE SETTING

Sample Run (Excerpt)

```
JOE IS AT SPACE #6
YOUR MOVE? 5
JOE IS AT SPACE #11
BLANK SPACE
JOE IS AT SPACE #11
BILL IS AT SPACE #8
YOUR MOVE? 4
BILL IS AT SPACE #12
MOVE AHEAD 6 SPACES
BILL IS AT SPACE #18
JOE IS AT SPACE #11
YOUR MOVE? 6
JOE IS AT SPACE #17
MOVE BACK 2 SPACES
JOE IS AT SPACE #15
BILL IS AT SPACE #18
YOUR MOVE? 4
BILL IS AT SPACE #22
MOVE AHEAD 2 SPACES
BILL IS AT SPACE #24
JOE IS AT SPACE #15.
```

Passing Points

In this game the players are on a circular "board" of 25 spaces. Each move can be from 1 to 5 spaces. After passing space #25, the players loop back around to space #1 again. Each space has a point value from -10 to +10. Before the computer reveals the point value of the space, you have the option of either keeping the points yourself or passing them over to your opponent. Obviously, the way to win this game is to remember as many space values from previous laps that you can (taking notes is cheating). Whoever reaches 100 points first, wins. Conversely, if a player reaches -100 points, he or she loses.


The strategy in this game is in deciding whether to go for negative points to hurt your opponent, or positive points to help yourself. See Fig. 2-2 for the flowchart.

Standard BASIC

```
5 PRINT:PRINT
7 DIM A(25)
10 PRINT",","PASSING POINTS"
15 PRINT"ENTER 1 FOR INSTRUCTIONS OR 0 TO PLAY"
17 INPUT X
20 IF X>0 THEN GOTO 350
25 REM*SET SPACE VALUES*
30 FOR X=1TO25
35 LET Y=INT(RND(0)*21)-10
40 LET A(X)=Y
45 NEXT X
50 LET A=1:LET B=1
55 LET C=0:LET D=0
60 PRINT"PLAYER #1";
62 INPUT A$
65 PRINT"PLAYER #2";
67 INPUT B$
70 REM* PLAYER #1'S TURN*
75 GOSUB 200
80 PRINT A$;
85 GOSUB 250
90 LET A=A+M
95 IF A>25 THEN LET A=A-25
100 PRINT"SPACE #";A
105 LET J=A(A)
110 GOSUB 300
115 IF G=1 THEN LET D=D+J
```

```
120 IF G=2 THEN LET C=C+J
125 REM* PLAYER #2'S TURN *
130 GOSUB 200
135 PRINT B$;
140 GOSUB 250
145 LET B=B+M
150 IF B>25 THEN LET B=B-25
155 PRINT"SPACE #";B
160 LET J=A(B)
165 GOSUB 300
170 IF G=1 THEN LET C=C+J
175 IF G=2 THEN LET D=D+J
180 GOTO 70
200 FOR X=1 TO 30
202 PRINT:NEXT X
205 PRINT"PLAYER",A$,B$
210 PRINT"SPACE",A,B
215 PRINT"SCORE",C,D
217 PRINT
220 IF C<-99 THEN GOTO 405
225 IF D<-99 THEN GOTO 415
230 IF C>99 THEN GOTO 415
235 IF D>99 THEN GOTO 405
240 RETURN
250 PRINT"YOUR MOVE";
255 INPUT M
260 REM* VALIDITY CHECK *
265 LET M=INT(M)
270 IF M<1 THEN GOTO 250
275 IF M>5 THEN GOTO 250
280 FOR X=1 TO 333
285 NEXT X
290 RETURN
300 REM* THE CHOICE *
305 PRINT "ENTER 1 TO PASS THE POINTS."
310 PRINT"ENTER 2 TO KEEP THE POINTS."
315 INPUT G
320 IF G=1 THEN 335
325 IF G=2 THEN GOTO 335
330 GOTO 310
335 FOR X=1 TO 400:NEXT X
340 PRINT"VALUE =";J
345 RETURN
350 PRINT"","INSTRUCTIONS"
355 PRINT"YOU ARE ON A CIRCULAR PATH OF 25 SPACES"
360 PRINT"EACH SPACE HAS A VALUE FROM -10 TO +10."
```

Fig. 2-2. Flowchart for Passing Points.



```

365 PRINT"YOU CAN CHOOSE TO KEEP THE POINTS OR PASS"
370 PRINT"THEM TO YOUR OPPONENT. THE GAME IS OVER"
375 PRINT"WHEN EITHER PLAYER REACHES 100 OR -100."
380 PRINT"THE OBJECT IS TO REMEMBER AS MANY SPACE"
385 PRINT"VALUES AS POSSIBLE SO YOU CAN KNOW WHAT"
390 PRINT"TO CHOOSE. IT'S CHEATING TO KEEP NOTES"
395 PRINT"PRESS 'ENTER' TO PLAY."
400 INPUT A$
402 GOTO 25
405 PRINT B$;
410 GOTO 420
415 PRINT A$;
420 PRINT" WINS!"
425 END

```

TRS-80 BASIC

```

10 P.:P.:P."","PASSING POINTS":P.
15 IN."ENTER 1 FOR INSTRUCTIONS OR 0 TO PLAY";X
20 IF X>0 G. 350
25 F.X=1TO25:Y=RND(21)-11
30 A(X)=Y:N.X
35 A=1:B=1:C=0:D=0
40 IN."PLAYER #1";A$
45 IN."PLAYER #2";B$
70 GOS.200
75 P.A$;
80 GOS.250
85 A=A+M
90 IF A>25 THEN A=A-25
95 P."SPACE #";A
100 J=A(A)
105 GOS.300
110 IF G=1 THEN D=D+J
115 IF G=2 THEN C=C+J
120 GOS.200
125 P.B$;
130 GOS. 250
135 B=B+M
140 IF B>25 THEN B=B-25
145 P."SPACE #";B
150 J=A(B)
155 GOS.300
160 IF G=1 THEN C=C+J
165 IF G=2 THEN D=D+J
170 G.70
200 CLS:P."PLAYER",A$,B$

```

```

205 P.“SPACE”,A,B
210 P.“SCORE”,C,D:P.
215 IF C<-99 G.405
220 IF D<-99 G.415
225 IF C>99 G.415
230 IF D>99 G. 405
235 RET.
250 IN.“YOUR MOVE”;M
255 M=INT(M)
260 IF M<1 G.250
265 IF M>5 G. 250
270 F.X=1TO333:N.X.
275 RET.
300 P.“ENTER 1 TO PASS THE POINTS.”
305 IN. “ENTER 2 TO KEEP THE POINTS.”; G
310 IF G=1 G.325
315 IF G=2 G.325
320 G.305
325 F.X=1TO400:N.X
330 P.“VALUE =”;J
335 RET.
350 P.“”,“INSTRUCTIONS”
355 P.“YOU ARE ON A CIRCULAR PATH OF 25 SPACES.”
360 P.“EACH SPACE HAS A VALUE FROM -10 TO +10.”
365 P.“YOU CAN CHOOSE TO KEEP THE POINTS OR PASS”
370 P.“THEM TO YOUR OPPONENT. THE GAME IS OVER”
375 P.“WHEN EITHER PLAYER REACHES 100 OR -100”
380 P.“POINTS. THE OBJECT IS TO REMEMBER AS MANY”
385 P.“SPACE VALUES AS POSSIBLE SO YOU CAN KNOW”
390 P.“WHAT TO CHOOSE. IT’S CHEATING TO KEEP”
395 P.“NOTES. PRESS ‘ENTER’ TO PLAY.””
400 IN.A$:G.25
405 P.B$;
410 G.420
415 P.A$
420 P.“WINS!”
425 END

```

Summary of Variables Used

A\$	PLAYER #1
B\$	PLAYER #2
A	PLAYER #1'SPACE
B	PLAYER #2'SPACE
C	PLAYER #1'SCORE
D	PLAYER #2'SCORE
G	PASSOPTION

J SPACE VALUE
K KEEP(=2)
M MOVE
P PASS(=1)
X TIMING
Y VALUE SETTING

Sample Run (Excerpt)

PASSING POINTS

ENTER 1 FOR INSTRUCTIONS OR 0 TO PLAY?0

PLAYER #1? JOE

PLAYER #2? BILL

PLAYER JOE BILL

SPACE 1 1

SCORE 0 0

JOE YOUR MOVE? 5

SPACE #6

PASS IT OR KEEP IT? KEEP

VALUE =6

PLAYER JOE BILL

SPACE 6 1

SCORE 6 0

BILL YOUR MOVE? 4

SPACE #5

PASS IT OR KEEP IT? KEEP

VALUE == -10

PLAYER JOE BILL

SPACE 6 5

SCORE 6 -10

JOE YOUR MOVE? 5

SPACE #11

PASS IT OR KEEP IT? PASS

VALUE =4

PLAYER JOE BILL

SPACE 11 5

SCORE 6 -6

Money Mad

The object in *Money Mad* is to make as much money off your "investments" as possible. Each player starts out with \$10,000. The first player to double that amount wins. Conversely, the first player to squander down to a net value of \$1000, or cash holdings of less than \$500, loses.

On each turn you can buy any of six offered stocks. By watching the way their values fluctuate on each report, you can try to determine which stocks are the most valuable. Remember though, the stocks which gain the greatest amounts can also lose the greatest amounts. Sometimes a stock's value may go negative. A player could cheat by buying up a few hundred shares of a negative stock to increase cash holdings. If you want to prevent this you can add the following step:

242 IF Q<0 THEN GOTO 255
The TRS-80 version would read—
242 IF Q<0 G.250

Rather than buying one of the established stocks, the players can choose to invest in an "independent venture". This is a one-shot deal that affects only the player's cash, not his stockholdings. The player can invest as much as he wants (provided, of course, he has the cash to pay for it). He can gain up to 100% of his investment, or he can lose up to 100%. For example, let's assume the player has \$10,000, and invests \$1000. If his investment gains 100% he'll end up with \$11,000—if it loses 100% he'll only have \$8000. Usually, of course, it will be somewhere between the extremes, but the loss or gain is entirely random.

All in all, this game is designed to be fairly unpredictable, but not pure chance. Strategy can be used to your advantage. At any rate, the result is a game that's almost as screwy as the "real" world of big finance.

Standard BASIC

```
5 PRINT:PRINT:PRINT
7 PRINT"","MONEY MAD"
10 PRINT
12 LET A(70)=0
14 PRINT"TYCOON #1";
16 INPUT A$
18 PRINT"TYCOON #2";
```

```
20 INPUT B$  
22 LET X=10000:LET U=0  
24 LET Y=10000:LET V=0  
26 FOR Z=1TO20  
28 LET A(Z)=0:NEXT Z  
30 LET A=INT(RND(0)*500)+1  
32 LET B=INT(RND(0)*500)+1  
34 LET C=INT(RND(0)*500)+1  
36 LET D=INT(RND(0)*500)+1  
38 LET E=INT(RND(0)*500)+1  
40 LET F=INT(RND(0)*500)+1  
42 PRINT“YOU EACH START OUT WITH $10,000”  
44 PRINT“THE FOLLOWING STOCKS ARE AVAILABLE”  
46 GOSUB 370  
48 LET G=INT(RND(0)*100)/100  
50 LET H=INT(RND(0)*100)/100  
52 LET I=INT(RND(0)*100)/100  
54 LET J=INT(RND(0)*100)/100  
56 LET K=INT(RND(0)*100)/100  
58 LET L=INT(RND(0)*100)/100  
60 FOR Z=1 to 666:NEXT Z  
65 PRINT“NEXT REPORT”  
70 GOSUB 300  
75 PRINT“TYCOON”,“CASH, “STOCKHOLDINGS”  
80 PRINTA$,X,U  
82 PRINT B$,Y,V  
85 GOSUB 260  
90 PRINT A$:“WHAT WILL YOU INVEST IN?”  
92 LET A(23)=1  
95 GOSUB 195  
100 IF T>X GOTO 410  
105 LET R=A(M)  
107 LET X=X-T  
110 LET R=R+S  
112 LET A(M)=R  
115 PRINT B$:“ WHAT WILL YOU INVEST IN?”  
117 LET A(23)=2  
120 GOSUB 195  
122 IF T>Y THEN GOTO 410  
125 LET M=M+10  
127 LET R=A(M):LET Y=Y-T  
130 LET R=R+S  
132 LET A(M)=R  
135 GOTO60  
140 PRINT“AMOUNT INVESTED”  
142 INPUT T
```

```
144 LET S=0
146 LET M=0
148 LET N=INT(RND(0)*200)-99
150 FOR Z=1TO555:NEXT Z
152 IF A(23)=2 THEN GOTO 175
155 IF T>X THEN GOTO 410
160 PRINT“PAYOFF IS”;N;“%”
162 LET X=X-T
164 LET N=N/100:LET T=T+(T*N)
166 LET X=X+T
168 LET T=0
170 RETURN
175 IF T>Y THEN GOTO 410
180 PRINT“PAYOFF IS”;N;“%”
182 LET Y=Y-T
184 LET N=N/100
186 LET T=T+(T*N)
188 LET Y=Y+T
190 LET T=0
192 RETURN
195 PRINT“(ENTER 1-7) ”;
197 INPUT M
200 LET M=INT(M)
202 IF M<1 THEN GOTO 195
205 IF M>7 THEN GOTO 195
210 IF M=1 THEN LET Q=A
215 IF M=2 THEN LET Q=B
220 IF M=3 THEN LET Q=C
225 IF M=4 THEN LET Q=D
230 IF M=5 THEN LET Q=E
235 IF M=6 THEN LET Q=F
237 REM* M=7 IS FOR INDEPENDENT VENTURES*
240 IF M=7 THEN GOTO 140
245 PRINT“ # OF SHARES”;
247 INPUT S
250 LET T=S*Q
252 PRINT“TOTAL COST = $”;T
255 RETURN
260 LET M=X+U:LET N=Y+V
262 IF X<500 THEN GOTO 505
263 IF Y<500 THEN GOTO 505
265 IF M<1000 THEN GOTO 290
270 IF N<1000 THEN GOTO 420
275 IF M>20000 THEN GOTO 445
280 IF N>20000 THEN GOTO 430
285 RETURN
```

290 PRINT A\$; " IS BANKRUPT"
295 END
300 LET S=1:LET R=INT(RND(0)*5)+1
302 IF R>3 THEN LET S== -1
305 LET A=A+(A*S*(G+R/170))
310 LET S=1:LET R=INT(RND(0)*5)+1
312 IF R>3 THEN LET S== -1
315 LET B=B+(B*S*(H+R/174))
320 LET S=1:LET R=INT(RND(0)*5)+1
322 IF R>3 THEN LET S== -1
325 LET C=C+(C*S*(I+R/170))
330 LET S=1:LET R=INT(RND(0)*5)+1
332 IF R>3 THEN LET S== -1
335 LET D=D+(D*S*(J+R/171))
340 LET S=1:LET R=INT(RND(0)*5)+1
342 IF R>3 THEN LET S== -1
345 LET E=E+(E*S*(K+R/171))
350 LET S=1:LET R=INT(RND(0)*5)+1
352 IF R>3 THEN LET S== -1
355 LET F=F+(F*S*(L+R/171))
360 LET U=(A(1)*A)+(A(2)*B)+(A(3)*C)+(A(4)*D)
362 LET U=U+(A(5)*E)+(A(6)*F)
365 LET V=(A(11)*A)+(A(12)*B)+(A(13)*C)+(A(14)*D)
367 LET V=V+(A(15)*E)+(A(16)*F)
370 PRINT "#1 ALLWEED ACRES", "\$";A
375 PRINT "#2 BURNT BAKERIES", "\$";B
380 PRINT "#3 CRUMBLED COMMUNICATIONS", "\$";C
385 PRINT "#4 DENTED DESKS, INC.", "\$";D
390 PRINT "#5 ELONGATED ENTERPRISES", "\$";E
395 PRINT "#6 FIZZLED FINANCE", "\$";F
400 PRINT "#7 INDEPENDENT VENTURES", "?"
405 RETURN
410 PRINT "YOU HAVE OVER-INVESTED! YOU CAN'T AFFORD"
415 PRINT "IT! YOU LOSE"
417 END
420 PRINTB\$;" IS BANKRUPT"
425 END
430 PRINT B\$;" HAS DOUBLED HIS FORTUNE!"
435 PRINT "A FINANCIAL WIZARD!"
440 END
445 IF N>20000 THEN GOTO 460
450 PRINT A\$;" HAS DOUBLED HIS FORTUNE!"
455 GOTO 435
460 PRINT "YOU HAVE BOTH DOUBLED YOUR FORTUNES!"
465 IF N>M THEN GOTO 485
470 IF M>N THEN GOTO 495

```

475 PRINT "IN AN EXACT TIE!?!"
480 END
485 PRINT "BUT ";B$;" IS RICHER THAN ";A$
490 END
495 PRINT "BUT ";A$;" IS RICHER THAN ";B$
500 END
505 PRINT "TYCOON", "CASH", "HOLDINGS", "TOTAL ASSETS"
510 PRINT A$,X,U,M
515 PRINT B$,Y,V,N
520 IF M=N THEN PRINT "TIE—NO ONE";
525 IF M>N THEN PRINT A$;
530 IF N>M THEN PRINT B$;
535 PRINT " WINS!"
540 END

```

TRS-80 BASIC

```

10 CLS:P.:P."","MONEY MAD":P.:X=10000:Y=10000
15 IN."TYCOON #1";A$:IN."TYCOON #2"B$:A(70)=0
20 F.Z=1TO20:A(Z)=0:N.Z:A=RND(500):B=RND(500):
C=RND(500)
25 D=RND(500):E=RND(500):F=RND(500):G=RND(100)/100
30 P."YOU EACH START OUT WITH $10,000":H=RND(100)/100
35 P."THE FOLLOWING STOCKS ARE AVAILABLE":GOS.370
40 I=RND(100)/100:J=RND(100/100:K=RND(100/100:
L=RND(100)/100
60 F.Z=1TO666:N.Z:P."NEXT REPORT":GOS.300
65 P."TYCOON", "CASH", "STOCKHOLDINGS"
70 P.A$,X,U:P.B$,Y,V:GOS. 260:A(23)=1
75 P.A$;"WHAT WILL YOU INVEST IN?":GOS.195
80 IFT>X G.410
85 R=A(M):X=X-T:R=R+S:A(M)=R:A(23)=2
90 P.B$;"WHAT WILL YOU INVEST IN?":GOS.195
95 IFT>Y G.410
100 M=M+10:R=A(M):Y=Y-T:R=R+S:A(M)=R:G.60
140 IN."AMOUNT INVESTED";T:M>2:N=RND(200)-100:S=Q
145 F.Z=1TO555:N.Z:IFA(23)=2 G.175
150 IFT>X G.410
155 P."PAYOFF IS";N;"%":X=X-T:N=N/100:T=T+(T*N)
160 X=X+T:T=0:RET.
175 IFT>Y G.410
180 P."PAYOFF IS";N;"%":Y=Y-T:N=N/100:T=T+(T*N)
185 Y=Y+T:T=0:RET.
195 IN."(ENTER 1-7) ";M:M=INT(M)
200 IF M<1 GOTO 195
205 IF M>7 GOTO 195
210 IF M=1 THEN Q=A
215 IF M=2 THEN Q=B
220 IF M=3 THEN Q=C

```

225 IFM=4THENQ=D
230 IFM=5THENQ=E
235 IFM=6THENQ=F
240 IFM=7G.140
245 IN.“# OF SHARES”;S:T=S*Q:P.“TOTAL COST = \$”;T
250 RET.
260 M=X+U:N=Y+V:IF X<500 G.505
262 IF Y>500 G.505
265 IF M<1000 G.290
270 IF N<1000 G.290
275 IF M>2000 G.445
280 IF N>2000 G.430
285 RET.
290 P.A\$;“ IS BANKRUPT”:END
300 S=1:R=RND(5):IF R>3 THEN S=-1
305 A=A+(A*S*(G+R/170)):S=1:R=RND(5):IF R>3 THEN S=-1
310 B=B+(B*S*(H+R/170)):S=1:R=RND(5):IF R>3 THEN S=-1
315 C=C+(C*S*(I+R/170)):S=1:R=RND(5):IF R>3 THEN S=-1
320 D=D+(D*S*(J+R/170)):S=1:R=RND(5):IF R>3 THEN S=-1
325 E=E+(E*S*(K+R/170)):S=1:R=RND(5):IF R>3 THEN S=-1
330 F=F+(F*S*(L+R/1.7))
360 U=(A(1)*A)+(A(2)*B)+(A(3)*C)+(A(4)*D)+(A(5)*E)+(A(6)*F)
365 V=(A(11)*A)+(A(12)*B)+(A(13)*C)+(A(14)*D)+(A(15)*E)+(A(16)*F)
370 P.“#1 ALLWEED ACRES”,“\$”;A:P.“#2 BURNT BAKERIES”,“\$”;B
375 P.“#3 CRUMBLED COMMUNICATIONS”,“\$”;C:P.“#4 DENTED DESKS,INC.”,
380 P.“\$”;D:P.“#5 ELONGATED ENTERPRISES”,“\$”;E
385 P.“# 6 FIZZLED FINANCE”,“\$”;F:P.“#7 INDEPENDENT VENTURES”,“?”
390 RET.
410 P.“YOU HAVE OVER-INVESTED! YOU CAN’T AFFORD IT”
415 P.“YOU LOSE!”:END
420 P.B\$;“ IS BANKRUPT”:END
430 P.B\$;“HAS DOUBLED HIS FORTUNE!”
435 P.“A FINANCIAL WIZARD!”:END
445 IF N>2000 G.460
450 P.A\$;“ HAS DOUBLED HIS FORTUNE!”:G.435
460 P.“YOU HAVE BOTH DOUBLED YOUR FORTUNES!”:IF N>M
G.485
465 IF M>N G.495
470 P.“IN AN EXACT TIE!”:END
485 P.“BUT”;B\$;“ IS RICHER THAN ”;A\$:END
495 P.“BUT”;A\$;“ IS RICHER THAN ”;B\$:END
505 P.“TYCOON”,“CASH”,“STOCKHOLDINGS”,“TOTAL ASSETS”

510 P.A\$,X,U,M:P.B\$,Y,V,N:IF M=N P."TIE — NO ONE";
515 IF M>N P.A\$;
520 IF M<N P.B\$;
525 P."WINS!":END

Money Mad

A\$ TYCOON #1
B\$ TYCOON #2
A ALLWEED ACRES * STOCK VALUE
B BURNT BAKERIES * STOCK VALUE
C CRUMBLED COMMUNICATIONS * STOCK VALUE
D DENTED DESKS, INC * STOCK VALUE
E ELONGATED ENTERPRISES * STOCK VALUE
F FIZZLED FINANCE * STOCK VALUE
G ALLWEED ACRES * GROWTH PERCENTAGE
H BURNT BAKERIES * GROWTH PERCENTAGE
I CRUMBLED COMMUNICATIONS * GROWTH PERCENTAGE
J DENTED DESKS, INC * GROWTH PERCENTAGE
K ELONGATED ENTERPRISES * GROWTH PERCENTAGE
L FIZZLED FINANCE * GROWTH PERCENTAGE
M INVESTMENT CHOICE/TYCOON #1's TOTAL WORTH
N TYCOON #2's TOTAL WORTH
Q VARIOUS CALCULATIONS
R VARIOUS CALCULATIONS
S VARIOUS CALCULATIONS
T AMOUNT INVESTED
U TYCOON #1'S STOCKHOLDINGS
V TYCOON #2'S STOCKHOLDINGS
X TYCOON #1'S MONEY
Y TYCOON #'S MONEY

Sample Run (Excerpt)

MONEY MAD
TYCOON #1 ?JOE
TYCOON #2 ?BILL
YOU EACH START OUT WITH \$10000
THE FOLLOWING STOCKS ARE AVAILABLE
#1 ALLWEED ACRES \$376
#2 BURNT BAKERIES \$452
#3 CRUMBLED COMMUNICATIONS \$165
#4 DENTED DESKS, INC. \$200
#5 ELONGATED ENTERPRISES \$419
#6 FIZZLED FINANCE \$68
#7 INDEPENDENT VENTURES ?
NEXT REPORT
#1 ALLWEED ACRES \$454.5712

#2 BURNT BAKERIES		\$880.3152
#3 CRUMBLED COMMUNICATIONS		\$221.9725
#4 DENTED DESKS, INC.		\$98.94
#5 ELONGATED ENTERPRISES		\$145.4349
#6 FIZZLED FINANCE		\$110.6768
#7 INDEPENDENT VENTURES		?
TYCOON	CASH	STOCKHOLDINGS
JOE	10000	0
BILL	10000	0

JOE WHAT WILL YOU INVEST IN?

(ENTER 1-7) ?1

OF SHARES? 10

TOTAL COST = \$4544.712

BILL WHAT WILL YOU INVEST IN?

(ENTER 1-7) ?2

OF SHARES? 5

TOTAL COST = \$4401.576

NEXT REPORT

#2 BURNT BAKERIES		\$279.3634
#3 CRUMBLED COMMUNICATIONS		\$1662.6513
#4 DENTED DESKS, INC.		\$390.4039
#5 ELONGATED ENTERPRISES		\$137.3124
#6 FIZZLED FINANCE		\$59.1164
#7 INDEPENDENT VENTURES		?
TYCOON	CASH	STOCKHOLDINGS
JOE	5455.288	2793.634
BILL	5598.424	8313.2565

JOE WHAT WILL YOU INVEST IN?

(ENTER 1-7) ?7

AMOUNT INVESTED? 2000

PAYOUT IS 58%

BILL WHAT WILL YOU INVEST IN?

(ENTER 1-7) ?7

AMOUNT INVESTED? 3000

PAYOUT IS -12%

NEXT REPORT

*****#

TYCOON	CASH	STOCKHOLDINGS
JOE	6615.288	2542.814
BILL	5598.424	-504.59

(NOTE — It looks like BILL's in trouble)

Gambling Boxes

In this game, each of the two players starts out with \$10. The first player to gather more than \$500 wins the game.

On each round the players are offered three mystery boxes labeled X, Y, and Z that can multiply their fortunes (the computer randomly selects one of the players to go first each round). The maximum values for each is explained in the program instructions. Notice that, while X can have the greatest multiplier value, it can also have the greatest tax rate. Box Z can thus conceivably be worth more than box X. This, of course, is the gambling aspect of the game.

On each round any given box may be selected only once. For example, if Player #1 takes box X, Player #2 is limited to choosing box Y or box Z. See Fig. 2-3 for the flowchart.

Standard BASIC

```
5 PRINT:PRINT:PRINT:PRINT
7 PRINT",, GAMBLING BOXES":PRINT
10 LET A=10:LET B=10
12 LET X=17:LET Y=19:LET Z=23
15 PRINT"PLAYER #1";
17 INPUT A$
20 PRINT"PLAYER #2";
22 INPUT B$
25 PRINT"EACH X BOX CAN MULTIPLY YOUR FORTUNE BY"
27 PRINT"UP TO 10, WITH UP TO 75% TAXES. EACH"
30 PRINT"Y BOX CAN MULTIPLY BY UP TO 5 WITH TAXES"
32 PRINT"UP TO 50%. Z CAN MULTIPLY UP TO 2 WITH"
35 PRINT"NO MORE THAN 25% TAXES."
45 LET P=INT(RND(0)*2)+1
47 LET M=INT(RND(0)*100)+1
50 LET N=INT(RND(0)*50)+1
52 LET Q=INT(RND(0)*20)+1
55 LET M=M/10:LET N=N/10:LET Q=Q/10
57 LET R=INT(RND(0)*75)+1
60 LET S=INT(RND(0)*50)+1
62 LET T=INT(RND(0)*25)+1
64 IF P=1 THEN GOTO 100
66 LET C=0:GOSUB 200
68 GOSUB 320
70 IF A>500 THEN GOTO 150
75 IF B>500 THEN GOTO 160
80 PRINT"NEXT ROUND"
```

```
85 GOTO 45
100 LET D=0
105 GOSUB 320
110 GOSUB 200
115 GOTO 70
150 PRINT A$;" WINS!"
155 END
160 PRINT B$;" WINS!"
165 END
200 PRINT B$;“, WHAT BOX WILL YOU TAKE”;
205 INPUT D
210 IF D=C THEN GOTO 235
215 IF D=X THEN GOTO 245
220 IF D=Y THEN GOTO 265
225 IF D=Z THEN GOTO 275
230 GOTO 200
235 PRINT A$;“ ALREADY TOOK THAT BOX”
240 GOTO 200
245 LET E=B*M:LET F=E+(R/100)
247 LET H=R:LET B=E-F:LET L=B
250 PRINT“ VALUE IS $”;E
255 PRINT“ TAXES = ”;H;“ %”
257 PRINT“ YOU NOW HAVE $”;L
260 RETURN
265 LET E=B*N:LET F=E*(S/100)
267 LET B=E-F:LET H=S:LET L=B
270 GOTO 250
275 LET E=B*Q:LET F=E*(T/100)
277 LET B=E-F:LET H=T:LET L=B
280 GOTO 250
285 LET E=A*M:LET F=E*(R/100)
287 LET A=E-F:LET H=R:LET L=A
290 GOTO 250
295 LET E=A*N:LET F=E*(S/100)
297 LET A=E-F:LET H=S:LET L=A
300 GOTO 250
305 LET E=A*Q:LET F=E*(T/100)
307 LET A=E-F:LET H=T:LET L=A
310 GOTO 250
320 PRINT A$;“ WHAT BOX WILL YOU TAKE”;
325 INPUT C
330 IF C=D THEN GOTO 355
335 IF C=X THEN GOTO 285
340 IF C=Y THEN GOTO 295
345 IF C=Z THEN GOTO 305
350 GOTO 320
```




Fig. 2-3. Flowchart for Gambling Boxes.

355 PRINT B\$;" ALREADY TOOK THAT BOX"
360 GOTO 320

TRS-80 BASIC

```
5 CLS:P.:P."","GAMBLING BOXES":P.  
10 A=10:B=10:X=17:Y=19:Z=23  
15 IN."PLAYER #1";A$:IN."PLAYER #2";B$  
20 P."EACH X BOX CAN MULTIPLY YOUR FORTUNE BY"  
25 P."UP TO 10, WITH UP TO 75% TAXES. EACH"  
30 P."Y BOX CAN MULTIPLY BY UP TO 5 WITH TAXES"  
35 P."UP TO 50%. Z CAN MULTIPLY UP TO 2 WITH"  
40 P."NO MORE THAN 25% TAXES."  
45 P=RND(2):M=RND(100)/10:N=RND(50)/10  
50 Q=RND(20)/10:R=RND(75):S=RND(50):T=RND(25)  
55 IF P=1 G. 100  
60 C=0:GOS.200  
65 GOS.320  
70 IF A>500 G.150  
75 IF B>500 G.160  
80 P."NEXT ROUND":G.45  
100 D=0:GOS.320  
105 GOS.200  
110 G.70  
150 P.A$;" WINS!"  
155 END  
160 P.B$;" WINS!"  
165 END  
200 P.B$;" , WHAT BOX WILL YOU TAKE";  
205 IN.D  
210 IF D=C G.235  
215 IF D=X G.245  
220 IF D=Y G.265  
225 IF D=Z G.275  
230 G.200  
235 P.A$;" ALREADY TOOK THAT BOX"  
240 G.200  
245 E=B*M:F=E+(R/100):H=R:B=E-F:L=B  
250 P."VALUE IS $";E  
255 P."TAXES = ";H;"%"  
257 P."YOU NOW HAVE $";L  
260 RET.  
265 E=B*N:F=E*(S/100):H=S:B=E-F:L=B  
270 G.250  
275 E=B*Q:F=E*(T/100):H=T:B=E-F:L=B  
280 G.250  
285 E=A*M:F=E*(R/100):A=E-F:H=R:L=A
```

290 G.250
295 E=A*N:F=E*(S/100):A=E-F:H=S:L=A
300 G.250
305 E=A*Q:F=E*(T/100):A=E-F:H=T:L=A
310 G.250
320 P.A\$;“, WHAT BOX WILL YOU TAKE”;
325 IN.C
330 IF C=D G.355
335 IF C=X G.285
340 IF C=Y G.295
345 IF C=Z G.305
350 G.320
355 P.B\$;“ALREADY TOOK THAT BOX”
360 G.320

Summary of Variables Used

A\$ PLAYER #1
B\$ PLAYER #2
A A\$SCORE
B B\$SCORE
C A\$BOX CHOICE
D B\$BOX CHOICE
E BOX VALUE
F TAXES
H TAX PERCENTAGE
L CURRENT SCORE
M BOX 1 MULTIPLIER
N BOX 2 MULTIPLIER
P PLAYER SELECT
Q BOX 3 MULTIPLIER
R BOX 1 TAX
S BOX 2 TAX
T BOX 3 TAX
X BOX 1
Y BOX 2
Z BOX 3

Sample Run (Excerpt)

GAMBLING BOXES

PLAYER #1?HOMER

PLAYER #2?JETHRO

EACH X BOX CAN MULTIPLY YOUR FORTUNE BY
UP TO 10, WITH UP TO 75% TAXES. EACH
Y BOX CAN MULTIPLY BY UP TO 5 WITH TAXES

UP TO 50%. Z CAN MULTIPLY UP TO 2 WITH
NO MORE THAN 25% TAXES.

HOMER, WHAT BOX WILL YOU TAKE?X

VALUE IS \$73

TAXES = 38%

YOU NOW HAVE 45.26

JETHRO, WHAT BOX WILL YOU TAKE?X

HOMER ALREADY TOOK THAT BOX

JETHRO, WHAT BOX WILL YOU TAKE?Y

VALUE IS \$21

TAXES = 42%

YOU NOW HAVE \$12.18

NEXT ROUND

JETHRO, WHAT BOX WILL YOU TAKE?X

VALUE IS \$101.094

TAXES = 50%

YOU NOW HAVE \$50.547

HOMER, WHAT BOX WILL YOU TAKE?Z

VALUE IS \$54.312

TAXES = 11%

YOU NOW HAVE \$48.33768

NEXT ROUND

JETHRO, WHAT BOX WILL YOU TAKE?X

VALUE IS \$121.3128

TAXES = 65%

YOU NOW HAVE \$78.85332

HOMER, WHAT BOX WILL YOU TAKE?Y

VALUE IS \$145.01304

TAXES = 7%

YOU NOW HAVE \$134.8621272

NEXT ROUND

Stack Cut

In this game, the computer shuffles a deck of cards. The suits are ignored: only the numerical values are counted. Aces count as one, Jacks as 11, Queens as 12, and Kings as 13. There are also two jokers in the deck which each count as -35.

The players take turns cutting the stack of cards. A player may cut from one to ten cards; then the value of the top card is added to that player's score. When the deck runs out of cards, the computer starts over at the beginning of the deck. The cards all retain the same positions throughout the game, and can be played more than once, so the position of certain cards can be memorized. For an even more difficult version of the game eliminate step 330, which displays the card's positional number.

As the program is written, each player cuts the stack 35 times, then the game is ended and the winner is declared. You can vary the number of turns in a game by changing step 205. The more turns there are, the more times the players will go through the deck, turning *Stack Cut* into more of a game of memory and strategy than chance. The first few passes through the deck are inevitably played as pure chance. See Fig. 2-4 for the flowchart.

Standard BASIC

```
10 FOR X=1 TO 54
15 LET A(X)=0
20 PRINT
25 NEXT X
30 PRINT",","STACK CUT"
35 LET X=INT(RND(0)*54)+1
40 LET A(X)=-35
45 LET X=INT(RND(0)*54)+1
50 IF A(X)<0 THEN GOTO 45
55 LET A(X)=-35
60 PRINT"NAME OF PLAYER #1";
65 INPUT A$
70 GOSUB 450
72 GOSUB 450
74 LET T=0.5
76 LET A=0
78 LET B=0
80 LET F=1
82 LET C=INT(RND(0)*2)+1
85 PRINT"NAME OF PLAYER #2";
90 INPUT B$
95 GOSUB 450
```

```
100 LET X=1
105 FOR Y=1TO13:FOR Z=1TO4
110 IF A(X)=0 THEN GOTO 125
115 LET X=INT(RND(0)*54)+1
120 GOTO 110
125 LET A(X)=Y
130 PRINT:NEXTZ
135 NEXT Y
140 IF C=2 THEN GOTO 180
150 GOSUB 200
155 PRINTA$;
160 GOSUB 300
165 LET A=A+Q
170 LET C=2
180 GOSUB 200
182 PRINT B$;
184 GOSUB 300
186 LET B=B+Q
188 LET C=1
190 GOTO 150
200 LET T=T+0.5
202 LET S=INT(T)
205 IF S>35 THEN GOTO 235
207 PRINT "TURN #";S
210 PRINT
215 PRINT"SCORE TO DATE"
220 PRINT" ",A$,B$
222 PRINT" ",A,B
225 PRINT
227 PRINT
230 RETURN
235 PRINT"GAME IS OVER"
237 PRINT
240 FOR X=1TO 555
242 NEXT X
245 PRINT"FINAL SCORE"
250 PRINT" ",A$,B$
255 PRINT" ",A,B
260 IF A=B THEN GOTO 280
265 IF A>B THEN PRINT A$;
270 IF A<B THEN PRINT B$;
275 PRINT" WINS!"
277 END
280 PRINT"HOW ABOUT THAT?! ** A TIE!"
285 END
300 PRINT" , YOUR CUT";
```




Fig. 2-4. Flowchart for Stack Cut.

```
302 INPUT D
305 LET D=INT(D)
310 IF D>10 THEN GOTO 420
315 IF D<1 THEN GOTO 420
320 LET F=F+D
322 FOR X=1 TO 470
324 NEXT X
326 IF F>54 THEN LET F=F- 54
330 PRINT“CARD #”;E
335 PRINT“YOU DREW --- ”;
337 LET Q=A(F)
340 FOR X=1 TO 333
342 NEXT X
345 IF Q>10 THEN GOTO 400
350 IF Q<2 THEN GOTO 360
352 PRINT Q
355 RETURN
360 IF Q=- 35 THEN GOTO 375
365 PRINT“ACE”
370 RETURN
375 PRINT“JOKER!”
377 FOR X=1 TO 370
380 NEXT X
382 LET G=INT(RND(0)*30)+4
385 FOR X=1 TO G
387 LET Z=INT(RND(0)*1047)+1
390 PRINT AT Z, “ HA! ”;
392 FOR Y = 1 TO 75
394 NEXT Y
396 NEXT X
398 RETURN
400 IF Q=11 THEN PRINT “JACK”
405 IF Q=12 THEN PRINT “QUEEN”
410 IF Q=13 THEN PRINT “KING”
415 RETURN
420 PRINT“PLEASE DO NOT CHEAT, ”;
425 IF C=1 THEN PRINT A$
430 IF C=2 THEN PRINT B$
435 GOTO 300
450 FOR X=1 TO 13
455 LET Y=INT(RND(0)*54)+1
460 IF A(Y)=0 THEN GOTO 470
465 GOTO 455
470 LET A(Y)=X
475 NEXT X
480 RETURN
```

TRS-80 BASIC

```
10 CLS
15 F.X=1TO54:A(X)=0:N.X
20 P.:P."","STACK CUT":P.
25 X=RND(54)
30 A(X)=-35
35 X=RND(54)
40 IF A(X)<0 G.35
45 A(X)=-35
50 IN."NAME OF PLAYER #1";A$
55 GOS.450
60 GOS.450
65 T=0.5:F=1
70 A=0:B=0
75 C=RND(2)
80 IN."NAME OF PLAYER #2";B$
85 GOS.450
90 X=1
95 F.Y=1TO13:F.Z=1TO4
100 IF A(X)=0 G.115
105 X=RND(54)
110 G.100
115 A(X)+y:N.Z
120 N.Y
130 CLS
140 IF C=2 G.180
150 GOS.200
155 P.A$;
160 GOS.300
165 A=A+Q
170 C=2
180 GOS.200
185 P.B$;GOS.300
190 B=B+Q:C=1
195 G.150
200 T=T+0.5:S=INT(T)
205 IF S>35 G.235
210 P."TURN #";S:P.
215 P."SCORE TO DATE"
220 P.",A$,B$
225 P.",A,B:P.:P.
230 RET.
235 P."GAME IS OVER":P.
240 F.X=1TO555:N.X
245 P."FINAL SCORE"
250 P.",A$,B$
```

255 P. "",A,B
260 IF A=B G.280
265 IF A>B P.A\$;
270 IF A<B P.B\$;
275 P."WINS!":END
280 P."HOW ABOUT THAT?! ** A TIE!"
285 END
300 IN.", YOUR CUT";D
305 D=INT(D)
310 IF D>10 G.420
315 IF D<1 G.420
320 F=F+D:F.X=1TO 470:N.X
325 IF F>54 THEN F=F-54
330 P."CARD #";F
335 P."YOU DREW ---";
340 Q=A(F):F.X=1TO333:N.X
345 IF Q>10 G.400
350 IF Q<2 G.360
355 P.Q:RET.
360 IF Q=-35 G.375
365 P."ACE"
370 RET.
375 P."JOKER!"
380 F.X=1TO470:N.X:G=RND(30)+3
385 F.X=1TOG.Z=RND(1047)
390 P.ATZ, "HA! ";
395 F.Y=1TO75:N.Y:N.X:RET.
400 IF Q=11 P."JACK"
405 IF Q=12 P."QUEEN"
410 IF Q=13 P."KING"
415 RET.
420 P."PLEASE DO NOT CHEAT, ";
425 IF C=1 P.A\$
430 IF C=2 P.B\$
435 G.300
450 F.X=1TO13
455 Y=RND(54)
460 IF A(Y)=0 G.470
465 G.455
470 A(Y)=X
475 N.X
480 RET.

Sample Run (Excerpt)

STACKCUT
NAME OF PLAYER #1? RALPH

NAME OF PLAYER #2? POTSIE
TURN #1
SCORE TO DATE
RALPH POTSIE
0 0
RALPH, YOUR CUT? 5
CARD #6
YOU DREW --- QUEEN
TURN #1
SCORE TO DATE
RALPH POTSIE
12 0
POTSIE, YOUR CUT? 10
CARD #16
YOU DREW --- 3
TURN #2
SCORE TO DATE
RALPH POTSIE
12 3
RALPH, YOUR CUT? 3
CARD #19
YOU DREW --- 5
TURN #2
SCORE TO DATE
RALPH POTSIE
17 3
POTSIE, YOUR CUT? 54
PLEASE DO NOT CHEAT, POTSIE
YOUR CUT? 8
CARD #27
YOU DREW --- 6
TURN #3
SCORE TO DATE
RALPH POTSIE
17 9
RALPH, YOUR CUT? 5
CARD #32
YOU DREW --- JOKER!
HA! HA! HA!
HA! HA! HA!
HA! HA!

TURN #3
SCORE TO DATE
RALPH POTSIE
-18 9

POTSIE, YOUR CUT? 10
CARD #42
YOU DREW --- ACE
TURN #4
SCORE TO DATE
RALPH POTSIE
- 18 10


TURN #35
SCORE TO DATE
RALPH POTSIE
212 203
POTSIE , YOUR CUT? 4
CARD #6
YOU DREW --- QUEEN
GAME IS OVER
FINAL SCORE
RALPH POTSIE
212 215
POTSIE WINS!

Variables Used

A\$ PLAYER #1
B\$ PLAYER #2
A A\$'S SCORE
B B\$'S SCORE
C TURN SELECT
D CUT CHOICE
F CARD #
G "HA!" COUNT
Q TOP CARD VALUE
S TURN COUNT
T TURN COUNT
X TIMING & MISC.
Y TIMING & MISC.
Z "HA!" LOCATION

Chapter 3

Frustration



This game merits a chapter all to itself since it has four versions.

The object of *Frustration* is to move your marker through the hundred-space grid from the first space to the hundredth. Each space has a concealed value, usually from -10 to +10 points, but some are booby traps that send you back to the starting position (that's the frustrating part). There is also one bomb, and if you hit it the game ends immediately.

Since you want to reach the end position with as high a score as possible, you might want to deliberately land on a booby trap to pick up more points. But the computer also keeps track of how many moves you take. Develop your own par for reaching the end, in as few moves possible, with the highest score.

The two-player version repeats the main game program (which is a solitaire version), so that you can match your skill and luck against your opponent. Player number 2 moves from the hundredth space to the first, otherwise the rules are the same as the one-player version—with one addition. If you land on a space already occupied by your opponent, he must return to his starting position.

The third version of *Frustration* has the computer taking the part of your opponent. It is programmed not to cheat: in deciding its moves it only knows what's under a space if that space has already been played. But once it learns what's at a space, you can be sure it doesn't forget!

The fourth version is a sort of ultimate. Again a two-player game is played, but this time the computer plays both sides: you

just sit back and watch. If the other versions of the game get too frustrating, it can help to watch the computer have as much trouble with it as you do.

As frustrating as this game is, it appears to be addictive. I've tried it on several people, and once started they can't seem to leave it alone.

In all versions each move is from 1 to 6 spaces.

No sample run is given for the last two versions, because they'd look essentially like the two-player version, for which a sample run is given. See Fig. 3-1 for the flowchart.




Fig. 3-1. Frustration Flowchart.

Main Game Program

Standard BASIC

```
5 FOR X=1 TO 100
7 PRINT
10 LET A(X)=0
12 NEXT X
15 PRINT " ", "FRUSTRATION!": PRINT: PRINT
17 REM * PLANT POINTS *
20 FOR X=1 TO 50
22 LET Y=INT(RND(0)*98)+2
24 LET Z=INT(RND(0)*21)-10
26 LET A(Y)=Z:NEXT X
28 REM * PLANT BOOBY TRAPS *
30 FOR X=1 TO 13
32 LET Y=INT(RND(0)*90)-4
34 LET A(Y)=-50:N.X
36 REM * PLANT BOMB *
38 LET Y=INT(RND(0)*80)+11
40 LET A(Y)=-100:LET A(1)=100
42 LET P=1:LET N=0
44 LET Q=0:LET S=0
45 GOSUB 200
47 REM * THE PLAY *
50 LET S=S+1
52 PRINT:PRINT "MOVE #"; S, "YOUR SCORE IS"; Q
54 PRINT "YOUR MOVE ";
56 INPUT M
58 LET M=INT(M)
60 IF M<1 THEN GOTO 240
65 IF M>6 THEN GOTO 240
70 LET A(P)=N:LET R=P+M
72 LET N=A(R)
74 PRINT "YOU JUST FOUND ";
76 FOR X=1 TO 333
78 NEXT X
80 IF N=0 THEN GOTO 235
85 IF N<-10 THEN GOTO 245
90 PRINT N; " POINTS!"
95 REM * ADJUST SCORE & POSITION *
97 LET Q=Q+N
100 LET P=R
102 LET A(P)=100
105 FOR X=1 TO 555:NEXT X
107 REM * TEST FOR WIN *
```

```
110 IF P>99 THEN GOTO 120
115 GOTO 45
120 REM* END GAME *
125 GOSUB 200
130 PRINT“YOU MADE IT IN”;S;“ MOVES!”
135 FOR X=1 TO 333:NEXT X
140 PRINT“ AND WITH A SCORE OF ”;Q
145 IF Q<10 THEN PRINT “THAT STINKS!”
150 IF Q>50 THEN PRINT “FANTASTIC!”
155 END
200 REM* DRAW GRID *
202 LET J=0
205 FOR X=1 TO 10
206 PRINT“ ”,
208 FOR Y=1 TO 10
210 LET J=J+1
212 IF A(J)=100 THEN GOTO 230
215 PRINT“X ”;
220 NEXT Y
222 PRINT
224 NEXT X
226 PRINT:PRINT
228 RETURN
230 PRINT“O ”;
232 GOTO 220
235 PRINT “ABSOLUTELY NOTHING”
237 GOTO 100
240 PRINT“INVALID MOVE”
242 GOTO 54
245 IF N == 100 THEN GOTO 260
247 PRINT“A BOOBY TRAP!”
250 LET N=0
252 LET P=1
254 LET A(P)=100
256 GOTO 105
260 PRINT“THE BOMB”
265 FOR X=1TO470:NEXT X
270 FOR X=1TO 4:PRINT:NEXT X
275 PRINT“ ”,“**** BOOM!! ****”
280 FOR X=1TO4:PRINT:NEXT X
285 END
```

TRS-80 BASIC

- 5 CLS:P.:P.:F.X=1TO100:A(X)=0:N.X
- 15 P.“ ”,“FRUSTRATION!”:P.:P.
- 20 F.X=1TO50:Y=RND(98)+1:A(Y)=RND(21)-11:N.X

```

25 F.X=1TO13:Y=RND(90)+5:A(Y)==-50:N.X
30 Y=RND(80)+10:A(Y)==-100
40 A(1)=100:P=1:N=0:Q=0:S=0
45 GOS.200
50 S=S+1:P.:P. "MOVE #";S,"YOUR SCORE IS";Q
55 IN."YOUR MOVE";M:IF M<1 G.240
60 IF M>6 B.240
65 M=INT(M):A(P)=N:R=P+M:N=A(R)
70 P."YOU JUST FOUND";
75 F.X=1TO333:N.X:IF N=0 G.235
80 IF N<-10 G.245
90 P.N;"POINTS!"
95 Q=Q+N
100 P=R:A(P)=100
105 F.X=1TO555:N.X
110 IF P>99 G.120
115 G.45
120 GOS.200:P."YOU MADE IT IN";S;"MOVES!"
122 F.X=1TO333:N.X
125 P."AND WITH A SCORE OF";Q
130 IF Q<10 P."THAT STINKS!"
135 IF Q>50 P."FANTASTIC!"
140 END
200 J=0:F.X=1TO 10:P.""
205 F.Y=1TO10:J=J+1
210 IF A(J)=100 G.230
215 P."X";
220 N.Y:P.:N.X:P.:RET.
230 P."O":G.220
235 P."ABSOLUTELY NOTHING!":G.100
240 P."INVALID MOVE":G.55
245 IF N=-100 G.260
250 P."A BOOBY TRAP!"
255 N=0:P=1:A(P)=100:G.105
260 P."THE BOMB":F.X=1TO470:N.X
265 P.:P.:P.:P."","***** BOOM!! *****"
270 P.:P.:P.:P.
275 END

```

Summary of Variables Used

- J GRID POSITION
- M MOVE
- N VALUE OF OCCUPIED SPACE
- P POSITION
- Q SCORE
- R NEW POSITION

S MOVE #
X TIMING
Y VARIOUS

Sample Run (Excerpt)

FRUSTRATION
O XXXXXXXXX
X XXXXXXXXX

MOVE #1 YOUR SCORE IS 0

YOUR MOVE? 6

YOU JUST FOUND -10 POINTS!

X XXXXXOXXX
X XXXXXXXXX
X XXXXXXXXX

MOVE #2 YOUR SCORE IS -10

YOUR MOVE? 6

YOU JUST FOUND 2 POINTS!

X XXXXXXXXX
X O XXXXXXXXX
X XXXXXXXXX

MOVE #3 YOUR SCORE IS -5

YOUR MOVE? 6

YOU JUST FOUND A BOOBY TRAP

O XXXXXXXXX
X XXXXXXXXX

MOVE #4
YOUR MOVE?

YOUR SCORE IS -8

Two Man Frustration

Standard BASIC

ENTER THE MAIN GAME PROGRAM, THEN ADD
THE FOLLOWING STEPS:

```
16 PRINT"PLAYER #1";
18 INPUT A$;
27 PRINT"PLAYER #2";
29 INPUT B$;
35 GOSUB 500
52 PRINT:PRINT A$;"MOVE #";S,"YOUR SCORE IS";Q
82 IF N=200 THEN GOTO 290
110 IF P>99 THEN GOTO 400
115 GOTO 300
155 RETURN
213 IF A(J)=200 THEN GOTO 490
290 LET A(C)=B:LET A(100)=200
292 LET C=100:LET B=0
294 LET N=0
296 PRINT B$;
298 GOTO 100
300 GOSUB 200
305 PRINT:PRINT B$;" MOVE #";S,"YOUR SCORE IS";A
310 PRINT"YOUR MOVE";
312 INPUT M
315 IF M<1 THEN GOTO 460
320 IF M>6 THEN GOTO 460
325 LET A(C)=B:LET R=C-M
330 LET B=A(R)
333 PRINT" YOU JUST FOUND ";
335 FOR X=1 TO 333:NEXT X
340 IF B=0 THEN GOTO 440
345 IF B<-10 THEN GOTO 445
350 IF B=100 THEN GOTO 465
355 PRINT B;" POINTS!"
360 LET A=A+B
365 LET C=R:LET A(C)=200
370 FOR X=1 TO 555:NEXT X
375 IF C<2 THEN GOTO 405
380 GOTO 45
400 PRINT A$;
402 GOTO 410
405 PRINT B$;
410 PRINT" MADE IT IN ";S;" MOVES"
412 GOSUB 200
```

```
415 PRINT:PRINT "", "FINAL SCORE"
420 PRINT A$,Q
422 GOSUB 145
425 PRINT:Q=A
430 PRINT B$,Q
432 GOSUB 145
435 END
440 PRINT "ABSOLUTELY NOTHING"
442 GOTO 365
445 IF B==100 THEN GOTO 260
450 PRINT "A BOOBY TRAP!"
452 LET B=0:LET C=100
454 LET A(C)=200
457 GOTO 370
460 PRINT "INVALID MOVE"
462 GOTO 310
465 LET A(P)=N:LET A(1)=100
470 LET N=0:LET B=0
475 LET P=1
480 PRINT A$
485 GOTO 365
490 PRINT "8";
495 GOTO 220
500 LET A=0:LET B=0
505 LET C=100
510 LET A(100)=200
515 PRINT
520 RETURN
```

STEPS 120 to 125 are not used in this version and may be eliminated.

TRS-80 BASIC

ENTER THE MAIN GAME PROGRAM THEN ADD THE FOLLOWING STEPS:

```
22 IN."PLAYER #1";A$
27 IN."PLAYER #2";B$
42 A=0:B=0:C=100;A(100)=200:P.
50 S=S+1:P.:P.A$;" MOVE #";S,"YOUR SCORE IS ";Q
85 IF N=200 G.290
110 IF P>99 G.400
115 G.300
140 RET.
212 IF A(J)=200 G.285
285 P."8";:G.220
290 A(C)=B:A(100)=200:C=100:B=0:N=0
```

```

295 P.B$:G.100
300 GOS.200
305 P.:P.B$“MOVE #”;S,“YOUR SCORE IS”;A
310 IN.“YOUR MOVE”;M
315 IF M<1 G.460
320 IF M>6 G.460
325 A(C)=B:R=C-M:B=A(R)
330 P.“YOU JUST FOUND”;
335 F.X=1TO333:N.X
340 IF B=0 G.440
345 IF B<-10 G.445
350 IF B=100 G.465
355 P.G,“POINTS!”
360 A=A+B
365 C=R:A(C)=200
370 F.X=1TO555:N.X
375 IF C<2 G.405
380 G.45
400 P.A$,:G.410
405 P.B$;
410 P.“MADE IT IN”;S;“MOVES”:GOS.200
415 P.:P.“”,“FINAL SCORE”
420 P.A$,Q:GOS.130
425 P.:Q=A
430 P.B$,Q:GOS.130
435 END
440 P.“ABSOLUTELY NOTHING”:G.365
445 IF B=-100 G.260
450 P.“A BOOBY TRAP!”
455 B=0:C=100:A(C)=200:G.370
460 P.“INVALID MOVE”:G.310
465 A(P)=N:A(1)=100:N=0:P=1:B=0
470 P.A$:G.365

```

Steps 120 & 125 of the Main Game Program are bypassed in this version, and may be eliminated.

Summary of Variables Used

A\$	PLAYER #1
B\$	PLAYER #2
A	B\$'S SCORE
B	VALUE OF B\$'S SPACE
C	B\$'A POSITION
J	GRID POSITION
M	MOVE
N	VALUE OF A\$'S SPACE

P A\$'S POSITION
Q A\$'S SCORE
R NEWSPACE
S MOVE#
Y TIMING
Y VARIOUS

Two-Player Version Sample Run (Excerpt)

FRUSTRATION

PLAYER #1? HOMER

PLAYER #2? JETHRO

O XXXXXXXXX
XXXXXXX XXXX
XXXXXXX XXXX8

HOMER MOVE #1 YOUR SCORE IS 0

YOUR MOVE? 5

YOU JUST FOUND 3 POINTS!

XXXXX O XXXX
XXXXXXX XXXX8

JETHRO MOVE #1 YOUR SCORE IS 0

YOUR MOVE? 6

YOU JUST FOUND -5 POINTS!

XXXXX O XXXX
XXXXXXX XXXX

Frustration vs. The Computer

TRS-80 BASIC

ENTER MAIN PROGRAM, THEN ADD THE FOLLOWING:

```
35 A=0:B=0:C=100:A(100)=200
42 FOR X=101 TO 200:A(X)=0:N.X
85 IF N=200 G.290
110 IF P>99 G.405
115 G.300
    ELIMINATE STEPS 120 TO 140
212 IF A(J)=200 G.285
285 P."8";:G.220
290 A(C)=B:A(100)=100:B=0:N=0
295 P."ME":G.100
300 P."TLL MOVE ";
305 E=C-6:F=0:G=E+100:D=C-1:U=E+100:V=D+100
310 FOR X=UTOV
312 IF A(X)> F GOS. 460
315 N.X:G=G-100:H=C-G:P.H
320 F.X=1TO555:N.X
330 P.:P."IJUST FOUND ";
335 A(C)=B:B=A(G):A(G+100)=B
340 IF B=100 G.390
345 IF B=0 G.385
350 IF B<-10 G.450
355 P. B;"POINTS!"
360 A=A+B
365 C=G:A(C)=200
370 FOR X=1TO555:N.X
375 IF C<2 G.410
380 G.45
385 P."ABSOLUTELY NOTHING.":G.365
390 A(P)=N:A(1)=100:P=1:N=0:B=0:P."YOU":G.365
395 P."A BOOBY TRAP!":A(100)=200:B=0
400 C=100:G.370
405 P."YOU";:G.415
410 P."I";
415 P."MADE IT IN ";S;"MOVES"
420 GOS.200
425 P.:P."","FINAL SCORE":P.
430 P."YOU",Q
435 P."ME",A
440 END
450 IF B=-100 G.260
```

455 G.395
460 F=A(X):G=X:RET.

Standard BASIC

ENTER THE MAIN GAME PROGRAM THEN ADD
THE FOLLOWING STEPS:

43 GOSUB 470
82 IF N=200 THEN GOTO 290
110 IF P>99 THEN GOTO 405
115 GOTO 300

ELIMINATE STEPS 120 TO 155

213 IF A(J)=200 THEN GOTO 490
290 LET A(C)=B:LET A(100)=200
292 LET C=100:LET B=0:LET N=0
294 PRINT“ME”
296 GOTO 100
300 PRINT“TLL MOVE ”;
302 LET E=C-6:LET F=0
304 LET G=E+100:LET D=C-1
306 LET U=E+100:LET V=D+100
308 FOR X=U TO V
310 IF A(X)>F THEN GOSUB 445
315 NEXT X
317 LET G=G-100:LET H=C-G
320 PRINT H
322 FOR X=1 TO 555:NEXT X
325 PRINT:PRINT“I JUST FOUND ”;
330 LET A(C)=B:LET B=A(G)
335 LET X=G+100:LET A(X)=B
340 IF B=100 THEN GOTO 390
345 IF B=0 THEN GOTO 385
350 IF B<-10 THEN GOTO 450
355 PRINT B,“POINTS!”
360 LET A=A+B
365 LET C=G:LET A(C)=200
370 FOR X=1 TO 555:NEXT X
375 IF C<° THEN GOTO 410
380 GOTO 45
385 PRINT“ABSOLUTELY NOTHING.”
387 GOTO 365
390 LET A(P)=N:LET A(1)=100
392 LET P=1:LET N=0:LET B=0
394 PRINT “YOU”: GOTO 365
395 PRINT“A BOOBY TRAP!”
400 LET A(100)=200:LET B=0:LET C=100
402 GOTO 370

```
405 PRINT"YOU";
407 GOTO 415
410 PRINT"I";
415 PRINT"MADE IT IN";S;"MOVES"
420 GOSUB 200
425 PRINT:PRINT"","FINAL SCORE"
430 PRINT:PRINT"YOU",Q
435 PRINT"ME",A
440 END
450 IF B=-100 THEN GOTO 260
455 GOTO 395
460 LET F=A(X):LET G=X
465 RETURN
470 LET A=0:LET B=0
475 LET C=100:LET A(100)=200
480 FOR X=101 TO 200
482 LET A(X)=0
484 NEXT X
486 RETURN
490 PRINT"8";
495 GOTO 220
```

The Computer vs. Itself

Standard BASIC

```
5 PRINT:PRINT "", "FRUSTRATION!"
7 REM* SET UP BOARD *
10 FOR X=1 TO 300
12 LET A(X)=0
14 NEXT X
16 FOR X=1 TO 50
18 LET Y=INT(RND(0)*98)+2
20 LET A(Y)=INT(RND(0)*21)-10
22 NEXT X
24 FOR X=1 TO 13
26 LET Y=INT(RND(0)*90)+6
28 LET A(Y)=-50
30 NEXT X
32 LET A(1)=100:LET P=1
34 LET N=0:LET Q=0:LET S=0
36 LET A=0:LET B=0
38 LET C=100:LET A(C)=200
40 REM * THE PLAY*
42 GOSUB 265
44 LET S=S+1
46 PRINT"MOVE # ";S,"PLAYER 1 MOVES";
50 LET E=P+6:LET F=0
52 LET G=E+100:LET D=P+1
54 LET U=D+100:LET V=G
56 FOR X=U TO V
58 IF A(X)>F THEN GOSUB 200
60 NEXT X
62 LET G=G-100:LET M=G-P
64 PRINT M
66 FOR X=1 TO 555:NEXT X
70 PRINT"I JUST FOUND ";
75 LET A(P)=N:LET N=A(G)
80 LET X=G+100:LET A(G)=N
85 IF N=200 THEN GOTO 205
90 IF N=-50 THEN GOTO 225
95 PRINT N;" POINTS!"
100 LET Q=Q+N
105 LET P=G:LET A(P)=100
110 FOR X=1 TO 470:NEXT X
115 IF P>99 THEN GOTO 235
120 GOSUB 265
```

```
122 PRINT“PLAYER 2 MOVES”;
124 LET E=C-6:LET F=0
126 LET G=E+200:LET D=C-1
128 LET U=E+200:LET V=D+200
130 FOR X=U TO V
135 IF A(X)>F THEN GOSUB 200
140 NEXT X
142 LET G=G-200:LET M=C-G
144 PRINT M
146 FOR X=1 TO 555:NEXT X
148 PRINT “I JUST FOUND”;
150 LET A(C)=B
155 LET B=A(G)
160 IF B=100 THEN GOTO 215
165 IF B=-50 THEN GOTO 230
170 PRINT B;“ POINTS!”
175 LET A=A+B
180 LET C=G
182 LET A(C)=200
185 FOR X=1 TO 470:NEXT X
190 IF C<2 THEN GOTO 240
195 GOTO 40
200 LET F=A(X):LET G=X
202 RETURN
205 PRINT“PLAYER 2! HA! HA!”
207 LET A(C)=B:LET B=0
210 LET C=100:LET N=0
212 LET A(C)=200:GOTO 105
215 PRINT“PLAYER 1! YUK! YUK!”
217 LET A(P)=N:LET N=0
220 LET B=0:LET P=1:LET A(P)=100
222 GOTO 180
225 LET P=1:LET A(P)=100:LET N=0
227 PRINT“A BOOBY TRAP!”:GOTO 110
230 LET C=100:LET A(C)=200:LET B=0
232 PRINT“A BOOBY TRAP!”:GOTO 185
235 PRINT“PLAYER 1”;
237 GOTO 245
240 PRINT“PLAYER 2”;
245 PRINT“MADE IT IN”;S;“MOVES”
247 GOSUB 265
250 PRINT“FINAL SCORE”,“PLAYER 1”,“PLAYER 2”
255 PRINT“,Q,A
260 END
265 LET J=0
270 FOR X=1 TO 10
```

```
275 PRINT" ",  
280 FOR Y=1 TO 10  
285 LET J=J+1  
290 IF A(J)=100 THEN PRINT"O";  
295 IF A(J)=200 THEN PRINT"8";  
300 IF A(J)< 100 THEN PRINT"X";  
305 NEXT Y:PRINT  
310 NEXT X:PRINT  
315 RETURN
```


TRS-80 BASIC

```
5 P.:P." ","FRUSTRATION!"  
10 F.X=TO300:A(X)=0:N.X  
15 F.X.=1TO50:Y=RND(98)+1:A(Y)=RND(21)-11:N.X  
20 F.X=1TO13:Y=RND(90)+5:A(Y)=-50:N.X  
30 A(1)=100:P=1:N=0:Q=0:S=0:A=0:B=0:C=100:A(100)=200  
40 GOS.265:S=S+1:P.“MOVE #”;S.“PLAYER 1 MOVES”;  
50 E=P+6:F=0:G=E+100:D=P+1:U=D+100:V=E+100:F.X=  
UTOV  
55 IF A(X)>F GOS.200  
60 N.X:G=G-100:M=G-P:P.M  
65 F.X=1TO555:N.X  
70 P.“I JUST FOUND”;  
75 A(P)=N:N=A(G):A(G+100)=N  
80 IF N=200 G.205  
85 IF N=-50 G.225  
90 P.N;“ POINTS!”  
100 Q=Q+N  
105 P=G:A(P)=100  
110 F.X=1TO470:N.X  
115 IF P>99 G.235  
120 GOS.265:P.“PLAYER 2 MOVES”;  
125 E=C-6:F=0:G=E+200:D=C-1:U=E+200:V=D+200:F.X=  
UTOV  
130 IF A(X)>F GOS.200  
140 N.X:G=G-200:M=C-G:P.M  
145 F.X=1TO555:N.X:P.“I JUST FOUND ”;  
150 A(C)=B:B=A(G)  
160 IF B=100 GOTO 215  
165 IF B=-50 G.230  
170 P.B;“ POINTS!”:A=A+B  
180 C=G:A(C)=200  
185 F.X=1TO470:N.X  
190 IF C< 2 G.240  
195 G.40  
200 F=A(X):G=X:RET.  
205 P.“PLAYER 2! HA! HA!”:A(C)=B:B=0:N=0
```

210 C=100:A(C)=200:G.105
215 P.“PLAYER 1! YUK! YUK!”:A(P)=N:N=0:B=0
220 P=1:A(P)=100:G.180
225 P=:A(P)=100:N=0:P.“A BOOBY TRAP!”:G.110
230 C=100:A(C)=200:B=0:P.“ABOOBY TRAP!”:G.185
235 P.“PLAYER 1”::G.245
240 P.“PLAYER 2”;
245 P.“MADE IT IN”;S;“MOVES”:GOS.265
250 P.“FINAL SCORE”,“PLAYER 1”,“PLAYER 2”
255 P.””,Q,A:END
265 J=0:F.X=1TO10:P.“,:F.Y=1TO10:J=J+1
270 IF A(J)=100 P.“0”;
275 IF A(J)=200 P.“8”;
280 IF A(J)< 100P.“X”;
285 N.Y:P.:N.X:P.:RET.

Summary of Variables Used

A	PLAYER #2'S SCORE (i.e., THE COMPUTER)
B	VALUE OF 2's SPACE
C	2's POSITION
D	MOVE CHOICE
E	MOVE CHOICE
F	MOVE CHOICE
G	MOVE CHOICE
J	GRID POSITION
M	MOVE
N	VALUE OF PLAYER #1'S SPACE
P	1'S POSITION
Q	1'S SCORE
R	NEW POSITION
S	MOVE #
X	TIMING
Y	VARIOUS



Chapter 4

Non-Games

The programs in this chapter aren't really games at all. One is a random probability demonstration, two are gag programs, and one is a random music composer (it just writes the music, it doesn't play it.)

Coin Flipper

A glance at the listing of this program should tell you that it's a simple one. There are only 25 steps.

Step 10 simply clears the screen and prints the title. Step 15 asks how many times you want the computer to flip the coin: this number is called X and should be a positive integer. After all, how can a coin be flipped -5.7 times? If a negative number or zero is entered, the program will bomb.

If you like, you can block such an invalid entry by adding the following steps:

```
17  X=INT(X)  
18  IF X<1 GOTO 15
```

Step 20 tests the value of X. If X = 1 the program jumps, to step 115. C is randomly set to equal 1 (Heads) or 2 (Tails). The result is printed and the program cycles back to step 15 to start over.

If step 20 finds X to be greater than 1, the program continues through step 25 which sets variables H (number of Heads) and T (Number of Tails) to zero. These variables are increased by one count each time the appropriate face comes up. Y is counted up to the value of X for the loop program (steps 30 to 55). On each pass through the loop C is randomly set to either 1 or 2. A 1 is counted as Heads (steps 45 and 50), while a 2 is counted as Tails (100 through 110).

After X loops, the total number of Heads and Tails are printed along with their percentages of the total of flips. If X is a reasonably large number, each should be close to 50 %.

The percentage of Heads (A) and of Tails (B) are added (D). If the total does not equal 100%, an error message is printed and the computer stops. This can happen in certain cases because of internal rounding off of extreme decimal places within the computer. If D equals 100%, the program cycles back to step 15 to start over.

If you don't want to bother with the 100% check (the difference should never be more than a very minute fraction anyway), you can eliminate steps 85, 90 and 95, and replace them with

```
90 GOTO 15
```

When you want to terminate the program, just hit the BREAK key.

See Fig. 4-1 for the flowchart.




Fig. 4-1. Coin Flipper flowchart.

Standard BASIC

```

10  CLS:PRINT" ", "COIN FLIPPER
15  PRINT"HOW MANY TIMES SHALL I FLIP THAT OLD COIN";
20  INPUT X
25  IF X=1 GOTO 115
30  LET H = 0: LET T = 0
33  PRINT:PRINT:FOR Y=1 TO X
40  LET C=INT(RND(1)*2)+1
45  IF C=2 GOTO 100
50  LET H=H+1:PRINT "H ";
55  NEXT Y:PRINT:PRINT
60  PRINT " ", "TOTALS"
65  LET A=(H/X)*100:LET B=(T/X)*100:LET D=A+B
70  PRINT "HEADS",H,A;"%"
75  PRINT "TAILS",T,B;"%"
80  PRINT " ", "D;"%"
85  IF D=100 GOTO 15
90  PRINT "I THINK I GOOFED"
95  END

```

```

100 PRINT "T";
105 LET T=T+1
110 GOTO 55
115 LET C=INT(RND(1)*2)+1
120 IF C=1 PRINT "HEADS"
125 IF C=2 PRINT "TAILS"
130 GOTO 15

```

TRS-80 BASIC

```

10 CLS:P.:“,COIN FLIPPER”
15 IN.“HOW MANY TIMES SHALL I FLIP THAT OLD COIN”;X
20 IF X=1 G.115
25 H=0:T=0:P.:P.Y=1TOX:C=RND(2)
30 IF C=2 G.100
40 H=H+1:P.“H”;
50 N.Y:P.:P.“”,“TOTALS”
60 A=(H/X)*100:B=(T/X)*100:D=A+B
70 P.“HEADS”,H,A;“%”:P.“TAILS”,T,B;“%
80 P.“”,“”,D;“%
85 IF D=100 G.15
90 P.“I THINK I GOOFED.”:END
100 T=T+1:P.“T”;
110 G.50
115 C=RND(2):IF C=1P.“HEADS”
120 IF C=2P.“TAILS”
130 G.15

```

Summary Of Variables Used

- A % of Heads
- B % of Tails
- C Result of current flip
- D A+B (should be 100)
- H Number of Heads
- T Number of Tails
- X Number of flips
- Y Loop count

Sample Run

```

COIN FLIPPER
HOW MANY TIMES SHALL I FLIP THAT OLD COIN? 1
HEADS
HOW MANY TIMES SHALL I FLIP THAT OLD COIN? 25
H T T H T H H T T T H T H T T H
T H T T T H T H H
TOTALS
HEADS    11    44%
TAILS    14    56%
100%

```

Favorite Song Surprise Poem

These two programs are simply computerized gags. Try entering them without thinking about what they'll do. Then you can spring them on your friends.

Since there is no randomness in these programs, it would spoil the joke to include sample runs here.

Once you've run these programs a couple times you should be able to work up similar gag programs of your own.

Favorite Song Standard BASIC

```
5 LET Y=1:LET N=0:PRINT:PRINT:PRINT:PRINT
10 PRINT:PRINT:"WOULD YOU LIKE ME TO DO MY FAVORITE
   SONG";
12 INPUT X
15 IF X=Y THEN GOTO 100
20 LET A=INT(RND(0)*300)+1
22 FOR B=1 TO A
24 LET C=INT(RND(0)*300)+1
26 FOR D=1 TO C:NEXT D
28 PRINT" FINK!";
30 LET E=INT(RND(0)*20)+1
35 FOR F=1 TO E:PRINT" ";
40 NEXT F:NEXT B
45 END
70 IF X=1 THEN PRINT"ONE BOTTLE ";
75 IF X>1 THEN PRINT X;" BOTTLES";
80 RETURN
85 PRINT"OF BEER ";
90 RETURN
100 FOR A=1 TO 30:PRINT:NEXT A
102 PRINT" ","OOOHHH...":PRINT:PRINT:PRINT
105 PRINT:FOR A=1 TO 400:NEXT A
107 LET X=99
110 FOR A=1 TO 20:PRINT:NEXT A
112 FOR A=1TO333:NEXT A
114 GOSUB 70
116 GOSUB 85
118 GOSUB 190
120 PRINT:FOR A=1 TO 270:NEXT A
122 GOSUB 70
124 GOSUB 85
```

```

126 PRINT
128 PRINT
130 FOR A=1 TO 300:NEXT A
132 PRINT"IF ONE OF THOSE BOTTLES SHOULD HAPPEN TO
FALL"
134 PRINT:FOR A=1 TO 250:NEXT A
136 LET X=X-1
138 IF X=0 THEN GOTO 200
140 GOSUB 70
142 GOSUB 85
144 GOSUB 190
146 FOR A=1 TO 777:NEXT A
150 PRINT:PRINT:PRINT
155 PRINT:PRINT:PRINT
160 GOTO 112
190 PRINT"ON THE WALL"
195 RETURN
200 PRINT"NO MORE BOTTLES";
205 GOSUB 85
210 GOSUB 190
215 FOR A=1 TO 999:NEXT A
220 PRINT:PRINT:PRINT
225 PRINT:PRINT:PRINT
230 PRINT"WANT ME TO DO IT AGAIN";
235 GOTO 12

```

TRS-80 BASIC

```

55 CLS:Y=1:N=0:P.:P.:P.
10 IN."WOULD YOU LIKE ME TO DO MY FAVORITE SONG";X
15 IF X=Y.G.100
20 C=RND(300):F.D=1TOC:A=RND(300):F.B=1TOA
25 N.B:P."FINK! ";:E=RND(20):F.B=1TOE:P." ";
30 N.B:N.D:END
70 IF X=1P."ONE BOTTLE ";
75 IF X>1P.X;"BOTTLES ";
80 RET.
85 P."OF BEER";:RET.
100 F.A=1TO30:P.:N.A:P." ","OOOHHH...":P.:P.:P.
105 F.A=1TO400:N.A:X=99
110 CLS:P.:P.:P.
115 F.A=1TO333:N.A:GOS.70:GOS.85:GOS.190:P.
120 F.A=1TO270:N.A:GOS.70:GOS.85:P.:P.
125 F.A=1TO300:N.A:P."IF ONE OF THOSE BOTTLES ";
130 P."SHOULD HAPPEN TO FALL":P.:F.A=1TO250
135 N.A:X=X-1:IF X=0G.200
140 GOS.70:GOS.85:GOS.190:F.A=1TO777:N.A

```

145 P.:P.:P.:P.:G.115
190 P."ON THE WALL":RET.
200 P."NO MORE BOTTLES":GOS.85:GOS.190
205 F.A=1TO999:N.A:P.:P.:P.:P.
210 IN."WANT ME TO DO IT AGAIN";X
215 G.15

Surprise Poem Standard BASIC

```
5 FOR X=1 TO 30
10 PRINT:NEXT X
12 GOSUB 315
14 GOSUB 225
16 GOSUB 255
17 GOSUB 320
18 GOSUB 280
19 GOSUB 220
20 GOSUB 320
21 GOSUB 295
22 GOSUB 320
24 GOSUB 285
26 GOSUB 320
28 GOSUB 270
30 GOSUB 320
32 GOSUB 260
34 GOSUB 315
36 PRINT:PRINT
38 GOSUB 320
40 GOSUB 225
42 GOSUB 320
44 GOSUB 245
50 GOSUB 320
52 GOSUB 225
54 PRINT"N":GOSUB 320
56 GOSUB 200
58 GOSUB 320
60 GOSUB 230
62 GOSUB 320
64 GOSUB 205
66 GOSUB 320
68 GOSUB 215
70 GOSUB 315
72 PRINT:PRINT
74 GOSUB 320
76 PRINT" ",;GOSUB 225
78 GOSUB 320
80 GOSUB 250
82 GOSUB 320
```

```
84 GOSUB 300
86 GOSUB 320
87 GOSUB 325
88 GOSUB 320
90 GOSUB 290
92 GOSUB 265
94 PRINT":GOSUB 320
96 PRINT:GOSUB 320
98 GOSUB 320
100 PRINT":GOSUB 310
102 GOSUB 320
104 GOSUB 210
108 GOSUB 240
112 GOSUB 235
114 GOSUB 320
116 GOSUB 305
120 GOSUB 265
122 GOSUB 315
124 PRINT:PRINT
126 GOSUB 320
128 GOSUB 225
130 GOSUB 255
131 GOSUB 320
132 GOSUB 280
133 GOSUB 220
134 GOSUB 320
135 GOSUB 295
136 GOSUB 320
138 GOSUB 285
142 GOSUB 270
144 GOSUB 320
146 GOSUB 260
148 GOSUB 315
150 PRINT:PRINT
152 GOSUB 320
154 PRINT":GOSUB 300
156 GOSUB 275
158 GOSUB 320
160 PRINT:PRINT",",
162 GOSUB 320
164 GOSUB 300
166 GOSUB 275
168 PRINT":PRINT
170 GOSUB 315
172 GOSUB 315
175 END
```

```
200 PRINT" A";
202 RETURN
205 PRINT" BEGE";
207 RETURN
210 PRINT" BOW-";
212 RETURN
215 PRINT" CAN";:RETURN
220 PRINT" EYE";:RETURN
225 PRINT" I";:RETURN
230 PRINT" GAR";:RETURN
235 PRINT" GED";:RETURN
240 PRINT" LEG";:RETURN
245 PRINT" LIVE";:RETURN
250 PRINT" LOVE";:RETURN
255 PRINT" M";:RETURN
260 PRINT" MAN";:RETURN
265 PRINT" MIN";:RETURN
270 PRINT" OR";:RETURN
275 PRINT" OT";:RETURN
280 PRINT" POP";:RETURN
285 PRINT" SAIL";:RETURN
290 PRINT" SWIM";:RETURN
295 PRINT" THE";:RETURN
300 PRINT" TO";:RETURN
305 PRINT" WT";:RETURN
310 PRINT" WITH";:RETURN
315 FOR X=1 TO 222
317 NEXT X
320 FOR X=1 TO 55:NEXT X
322 RETURN
325 PRINT" GO";:RETURN
```

TRS-80 BASIC

```
5 CLS:P.:P.:GOS.315
10 GOS.225:GOS.255:GOS.320:GOS.280:GOS.220:GOS.320:
   GOS.295:GOS.320
15 GOS.285:GOS.270:GOS.320:GOS.260
20 GOS.315:P.:P.:GOS.320:GOS.225:GOS.320
25 GOS.245:GOS.320:GOS.225
30 P."N";:GOS.320:GOS.200:GOS.320:GOS.230
35 GOS.205:GOS.320:GOS.215:GOS.315:P.:P.:GOS.320
40 P." ";:GOS.225:GOS.320:GOS.250:GOS.320
45 GOS.300:GOS.320:GOS.325:GOS.320:GOS.290:GOS.265
50 P."";:GOS.320:P.:GOS.320:GOS.320
55 P." ";:GOS.310:GOS.320:GOS.210
60 GOS.240:GOS.235:GOS.320:GOS.305
```

65 GOS.265:GOS.315:P.:P.:GOS.320
70 GOS.225:GOS.255:GOS.320:GOS.280:GOS.220:
GOS.320:GOS.295:GOS.320
75 GOS.285:GOS.270:GOS.320:GOS.260
80 GOS.315:P.:P.:GOS.320:P.“”,:GOS.300
85 GOS.275:GOS.320:P.:P.“”,“”,
90 GOS.320:GOS.300:GOS.275:P.“!”:P.:P.
95 GOS.315:GOS.315
100 END
200 P.“A”;:RET.
205 P.“BEGE”;:RET.
210 P.“BOW-”;:RET.
215 P.“CAN”;:RET.
220 P.“EYE”;:RET
225 P.“I”;:RET.
230 P.“GAR”;:RET.
235 P.“GED”;:RET.
240 P.“LEG”;:RET.
245 P.“LIVE”;:RET.
250 P.“LOVE”;:RET.
255 P.“M”;:RET.
260 P.“MAN”;:RET.
265 P.“MIN”;:RET.
270 P.“OR”;:RET.
275 P.“OT”;:RET.
280 P.“POP”;:RET.
285 P.“SAIL”;:RET.
290 P.“SWIM”;:RET.
295 P.“THE”;:RET.
300 P.“TO”;:RET.
305 P.“WT”;:RET.
310 P.“WITH”;:RET.
315 F.X=1TO222:N.X
320 F.X=1TO55:N.X:RET.
325 P.“GO”;:RET.

Tunesmith

This program is a very simple random music composer. It prints out notes by their letter names, gives the note lengths ($\frac{1}{2}$, $\frac{1}{4}$, $\frac{1}{8}$, etc.), and indicates the overall dynamic level of the phrase. You may manually convert this information into standard musical notation to play it on any instrument you choose.

While the notes are randomly selected, the selection process is weighted (steps 10 to 45) so that the phrase will tend to stay in the key of C.

In all probability many of the phrases produced by this simple program won't sound very musical, but many are interesting, and some sound quite good.

While this program is too simplistic to be of much use as a real compositional aid, it might suggest themes to a composer with a reluctant muse. *Tunesmith* (like all of these programs) is mostly intended for fun.

See Fig. 4-2 for the sample run in standard musical notation and Fig. 4-3 for the flowchart.


Standard BASIC

```
10 DATA 1,0,1,2,13,8,6,1,3,4  
15 DATA 1,0,5,6,7,1,8,10,13,6  
20 DATA 1,13,8,12,10,11,1,9,13,3  
25 DATA 1,1,0,13,12,8,6,5,1,0  
30 DATA 5,13,1,3,5,13,10,8,6,1  
35 FOR X=1 TO 50  
37 READ Q  
40 PRINT
```



Fig. 4-2. Tunesmith sample run in standard musical notation.

Fig. 4-3. Tunesmith flowchart.



```

42 LET A(X)=Q
45 NEXT X
50 REM*DYNAMICS KEY *
55 LET F=INT(RND(0)*7)+1
60 IF F=1 THEN PRINT“PP”
65 IF F=2 THEN PRINT“P”
70 IF F=3 THEN PRINT“MP”
75 IF F=4 THEN PRINT“MF”
80 IF F=5 THEN PRINT“MF”
85 IF F=6 THEN PRINT“F”
90 IF F=7 THEN PRINT“FF”
95 PRINT
100 FOR X=1 TO 8
105 LET M=INT(RND(0)*50)+1
110 LET Q=A(M)
112 IF Q=0 THEN PRINT“REST”;
115 IF Q=1 THEN PRINT“C”;
  
```

```

120 IF Q=2 THEN PRINT"C#   ";
125 IF Q=3 THEN PRINT"D   ";
130 IF Q=4 THEN PRINT"D#   ";
135 IF Q=5 THEN PRINT"E   ";
140 IF Q=6 THEN PRINT"F   ";
145 IF Q=7 THEN PRINT"F#   ";
150 IF Q=8 THEN PRINT"G   ";
155 IF Q=9 THEN PRINT"G#   ";
160 IF Q=10 THEN PRINT"A   ";
165 IF Q=11 THEN PRINT"A#   ";
170 IF Q=12 THEN PRINT"B   ";
175 IF Q=13 THEN PRINT"C   ";
180 IF Q=0 THEN GOTO 195
185 LET B=INT(RND(0)*10)+1
190 IF B>7 THEN PRINT"-";
195 NEXT X
200 PRINT:PRINT
202 FOR X=1TO8
205 LET T=INT(RND(0)*9)+1
207 IF T=1 THEN PRINT"1/16 ";
210 IF T=2 THEN PRINT"1/8 ";
215 IF T=3 THEN PRINT"1/4 ";
220 IF T=4 THEN PRINT"1/4 ";
225 IF T=5 THEN PRINT"1/8 ";
230 IF T=6 THEN PRINT"1/2 ";
235 IF T=7 THEN PRINT"1/4 ";
240 IF T=8 THEN PRINT"1/2 ";
245 IF T=9 THEN PRINT"1   ";
250 NEXT X
255 PRINT:PRINT
260 PRINT"TYPE 1 FOR NEXT LINE OR 2 TO END"
265 INPUT H
270 IF H=1 THEN GOTO 100
275 END

```

TRS-80 BASIC

```

10 DATA 1,0,1,2,13,8,6,1,3,4
15 DATA 1,0,5,6,7,1,8,10,13,6
20 DATA 1,13,8,12,10,11,1,9,13,3
25 DATA 1,1,0,13,12,8,6,5,1,0
30 DATA 5,13,1,3,5,13,10,8,6,1
35 F.X=1TO50:READ Q
40 A(X)=Q:N.X
50 CLS:P.:P.
55 F=RND(7):IF F=1 P."PP"
60 IF F=2 P."P"
65 IF F=3 P."MP"

```

```

70 IF F=4 P.“MF”
75 IF F=5 P.“MF”
80 IF F=P.“F”
85 IF F=7 P.“FF”
90 P.
100 F.X=1TO8:M=RND(50)
110 Q=A(M);IF Q=0 P.“REST”;
115 IF Q=1 P.“C”;
120 IF Q=2 P.“C#”;
125 IF Q=3 P.“D”;
130 IF Q=4 P.“D#”;
135 IF Q=5 P.“E”;
140 IF Q=6 P.“F”;
145 IF Q=7 P.“F#”;
150 IF Q=8 P.“G”;
155 IF Q=9 P.“G#”;
160 IF Q=10 P.“A”;
165 IF Q=11 P.“A#”;
170 IF Q=12 P.“B”;
175 IF Q=13 P.“C”;
180 IF Q=0 G. 195
185 B=RND(10)
190 IF B>7 P.“—”;
195 NEXT X
200 P.:P.:F.X=1TO8
205 T=RND(9):IF T=1 P.“1/16”;
210 IF T=2 P.“½”;
215 IF T=3 P.“¼”;
220 IF T=4 P.“¾”;
225 IF T=5 P.“⅓”;
230 IF T=6 P.“⅔”;
235 IF T=7 P.“⅕”;
240 IF T=8 P.“⅖”;
245 IF T=9 P.“1”;
250 N.X:P.:P.
255 IN.“TYPE 1 FOR NEXT LINE OR 2 TO END”;H
260 IF H=1 G. 100
265 END

```

Sample Run

MF


CA—FF C—FCC
 $\frac{1}{4}$ $\frac{1}{4}$ $\frac{1}{4}$ $\frac{1}{2}$ 1/16 1/16 $\frac{1}{8}$ $\frac{1}{4}$

TYPE 1 FOR NEXT LINE OR 2 TO END ?1

CC—GF# C REST E E

$\frac{1}{2}$ $\frac{1}{4}$ $\frac{1}{8}$ $\frac{1}{4}$ $\frac{1}{2}$ 1 $\frac{1}{4}$ 1/16

TYPE 1 FOR NEXT LINE OR 2 TO END ?2



Chapter 5

Helpful Programs

This chapter includes four programs which are not exactly games. They are helpful programs, designed to be of use to you in many different ways.

BINARY/DECIMAL

You probably know that computers use a different system of counting than we do. Our system is called the *decimal* system because it has ten digits(1,2,3,4,5,6,7,8,9, and 0). The computer, on the other hand, uses the *binary* system, which has only two digits (1 and 0).

The following program allows you to convert from one system to the other, in either direction. It should be noted that 4095 is the largest decimal number this program can convert. See Fig. 5-1 for the flowchart. Here are some refreshers on how the binary system works.

ADDITION

$$\begin{array}{r} 0 \quad 0 \quad 1 \quad 1 \\ +0 \quad +1 \quad +0 \quad +1 \\ \hline 0 \quad 1 \quad 1 \quad 10 \end{array}$$

BINARY DECIMAL EQUIVALENT

1	1
10	2
11	3
100	4
101	5
110	6
111	7

1000	8
1001	9
1010	10
1011	11
1100	12
1101	13
1110	14
1111	15
10000	16
•	•
•	•
•	•




Fig. 5-1. Binary/Decimal Flowchart.

Standard BASIC

```
10 LET B=1
15 LET D=2
20 PRINT“ENTER B TO CONVERT FROM BINARY TO DECIMAL,”
25 PRINT“OR D TO CONVERT FROM DECIMAL TO BINARY”;
30 INPUT A
40 IF A=2 THEN GOTO 200
50 PRINT“ENTER YOUR BINARY NUMBER”;
55 INPUT N
60 IF<1 THEN GOTO 170
65 LET N=INT(N)
70 LET C=0.5
75 LET E=0
80 LET C=C*2
90 IF N=0 THEN GOTO 150
100 LET N=N/10
110 LET M=N-INT(N)
115 LET N=INT(N)
120 IF M=0.1 THEN LET E=E+C
130 IF M>0.1 THEN GOTO 190
140 GOTO 80
150 PRINT E
160 GOTO 10
170 PRINT“THE NUMBER MUST BE POSITIVE AND GREATER
THAN 1”
180 GOTO 10
190 PRINT “THIS NUMBER IS NOT IN BINARY FORM”
195 GOTO 10
200 PRINT“ENTER YOUR DECIMAL NUMBER”;
205 INPUT N
210 IF N<1 GOTO 170
215 LET N=INT(N)
220 IF N>4095 THEN GOTO 330
230 LET E=0
235 LET C=2048
240 LET M=INT(N/C)
250 IF M=1 THEN GOSUB 300
260 IF N=0 THEN GOTO 150
270 LET C=C/2
275 IF C<1 THEN GOTO 150
280 LET E=E*10
290 GOTO 240
300 LET E=E+1
310 LET N=N-M
```

```
320 RETURN
330 PRINT"THE PROGRAM CAN ONLY HANDLE NUMBERS UP
      TO 4095"
340 GOTO 10
```

TRS-80 BASIC

```
10 B=1:D=2
20 P."ENTER B TO CONVERT FROM BINARY TO DECIMAL, OR
      D"
30 P."TO CONVERT FROM DECIMAL TO BINARY";A
40 IF A=2 G. 200
50 IN."ENTER YOUR BINARY NUMBER";N
60 IF N<1 G. 170
65 N=INT(N)
70 C=0.5:E=0
80 C=C*2
90 IF N=0 G. 150
100 N=N/10
110 M=N-INT(N);N=INT(N)
120 IF M=.1 THEN E=E+C
130 IF M>.1 G. 190
140 G. 80
150 P.E
160 G. 10
170 P."THE NUMBER MUST BE POSITIVE AND GREATER THAN
      1"
180 G. 10
190 P."THIS NUMBER IS NOT IN BINARY FORM":G. 10
200 IN."ENTER YOUR DECIMAL NUMBER";N
210 If N<1 G. 170
215 N=INT(N)
220 IF N>4095 G. 330
230 E=0:C=2048
240 M=INT(N/C)
250 IF M=1 GOS. 300
260 IF N=0 G. 150
270 C=C/2
280 IF C<1 G. 150
285 E=E*10
290 G. 240
300 E=E+1
310 N=N-M
320 RET.
330 P."THIS PROGRAM CAN ONLY HANDLE NUMBERS UP TO
      4095"
340 G. 10
```

Summary of Variables Used

- A MODE CHOICE
- B BINARY
- C PLACE VALUE
- D DECIMAL
- E CONVERTED NUMBER
- M INTERMEDIATE VARIABLE
- N NUMBER TO BE CONVERTED

Sample Run

ENTER B TO CONVERT FROM BINARY TO DECIMAL, OR D
TO CONVERT FROM DECIMAL TO BINARY? D

ENTER YOUR DECIMAL NUMBER? 123

1111011

ENTER B TO CONVERT FROM BINARY TO DECIMAL, OR D
TO CONVERT FROM DECIMAL TO BINARY? B

ENTER YOUR BINARY NUMBER? 123

THIS NUMBER IS NOT IN BINARY FORM

ENTER B TO CONVERT FROM BINARY TO DECIMAL, OR D
TO CONVERT FROM DECIMAL TO BINARY? B

ENTER YOUR BINARY NUMBER? 1010101

85

ENTER B TO CONVERT FROM BINARY TO DECIMAL, OR D
TO CONVERT FROM DECIMAL TO BINARY? B

ENTER YOUR BINARY NUMBER? 111

7

ENTER B TO CONVERT FROM BINARY TO DECIMAL, OR D
TO CONVERT FROM DECIMAL TO BINARY? D

ENTER YOUR DECIMAL NUMBER? 17

10001

Day of the Week

Sometimes it's interesting to know on what day of the week some specific date fell. You could thumb your way through an old calendar, but this program will quickly calculate the day of the week for you.

The dates used in the sample run are July 31, 1980 (7/31/1980), April 17, 1954 (4/17/1954), and August 22, 1734 (8/22/1734). Of course you can also use dates which extend into the future.

Note that this program has no provisions for validating your entries. It will accept 34/952/21212 as a date, but of course the




Fig. 5-2. Flowchart for Day of the Week.

result will be utterly meaningless. So double-check your entries to avoid getting a wrong answer.

See Fig. 5-2 for the flowchart.

Standard BASIC

```
10 PRINT "", "DAY OF THE WEEK"
15 PRINT
20 PRINT "ENTER THE MONTH NUMBER — 1-12";
25 INPUT M
30 LET M=M-2
40 IF M=0 THEN LET M=12
50 IF M=-1 THEN LET M=11
60 PRINT "ENTER THE DAY OF THE MONTH";
65 INPUT D
70 PRINT "ENTER THE YEAR";
75 INPUT Y
80 IF M > 10 THEN LET Y=Y-1
85 LET X=INT(Y/100)
90 LET Y=(Y-X)*100
95 LET A=INT(D+(2.6*M-0.2))
100 LET A=A+INT(Y/4+Y)
110 LET A=A+INT((X/4)-2*X)
120 IF A<0 THEN GOTO 240
130 LET A=A/7
135 LET A=A-INT(A)*7
140 LET A=INT(A)
150 PRINT "THAT DATE FELL ON A ";
160 IF A=0 THEN PRINT "SUNDAY"
170 IF A=1 THEN PRINT "MONDAY"
180 IF A=2 THEN PRINT "TUESDAY"
190 IF A=3 THEN PRINT "WEDNESDAY"
200 IF A=4 THEN PRINT "THURSDAY"
210 IF A=5 THEN PRINT "FRIDAY"
220 IF A=6 THEN PRINT "SATURDAY"
230 END
240 LET A=A+7
250 GOTO 120
```

TRS-80 BASIC

```
10 P. "", "DAY OF THE WEEK":P.
20 IN."ENTER THE MONTH NUMBER — 1-12";M
30 M=M-2
40 IF M=0 THEN M=12
50 IF M=-1 THEN M=11
```

```
60 IN.“ENTER THE DAY OF THE MONTH”;D
70 IN.“ENTER THE YEAR”;Y
80 IF M>10 THEN Y=Y-1
90 X=INT(Y/100); Y=(Y-X)*100
100 A=INT(D+(2.6*M-.2))+INT(Y/4+Y)
110 A=A+INT((X/4)-2*X)
120 IF A<0 G.240
130 A=A/7:A=A-INT(A)*7
140 A=INT(A)
150 P.“THAT DATE FELL ON A ”;
160 IF A = 0 THEN P.“SUNDAY”
170 IF A=1 THEN P.“MONDAY”
180 IF A=2 THEN P.“TUESDAY”
190 IF A=3 THEN P.“WEDNESDAY”
200 IF A=4 THEN P.“THURSDAY”
210 IF A=5 THEN P.“FRIDAY”
220 IF A=6 THEN P.“SATURDAY”
230 END
240 A=A+7:G.120
```

Summary of Variables Used

A	Day of the week	X	Century
D	Date	Y	Year
M	Month		

Sample Run

RUN

DAY OF THE WEEK

ENTER THE MONTH NUMBER — 1-12? 7

ENTER THE DAY OF THE MONTH? 31

ENTER THE YEAR? 1980

THAT DATE FELL ON A THURSDAY

READY

RUN

DAY OF THE WEEK

ENTER THE MONTH NUMBER — 1-12? 4

ENTER THE DAY OF THE MONTH? 17

ENTER THE YEAR? 1954

THAT DATE FELL ON A SATURDAY

READY

RUN

DAY OF THE WEEK

ENTER THE MONTH NUMBER — 1-12? 8

ENTER THE DAY OF THE MONTH? 22

ENTER THE YEAR? 1734

THAT DATE FELL ON A SUNDAY

READY

Ohm's Law

This one is a bit off the main path as far as this book goes, but it's interesting to try entering various variables to see how voltage, current, resistance and power interact in an electrical circuit. Just pull numbers out of the air and then study the results.

Most experimenters are at least vaguely familiar with Ohm's Law (Voltage equals current times resistance), but by running a number of combinations through this program you can really get a solid feel for it.

See Fig. 5-3 for the flowchart.

Standard BASIC

```
5 REM ** OHM'S LAW **
10 LET O=1
15 LET V=2
20 LET C=3
25 PRINT"DO YOU NEED TO FIND OHMS, VOLTS, OR CURRENT";
30 INPUT X
35 IF X=O THEN GOTO 70
40 IF X=V THEN GOTO 130
50 IF X=C THEN GOTO 160
60 GOTO 20
70 PRINT"VOLTS";
75 INPUT E
80 GOSUB 220
85 LET R=E/I
90 LET P=E*I
95 PRINT"RESISTANCE IS ";R;" OHMS (OR ";
100 LET R=R/1000
105 PRINT R;" KILOHMS"
110 PRINT "POWER IS ";P;" WATTS"
120 GOTO 10
130 GOSUB 220
135 GOSUB 260
140 LET E=I*R
145 LET P=I*I*R
150 PRINT"VOLTAGE IS ";E;" VOLTS"
155 GOTO 110
160 PRINT"VOLTS";
165 INPUT E
170 GOSUB 260
175 LET I=E/R
```

```
180 PRINT“CURRENT IS”;I;“ AMPS (OR”;  
185 LET I=I*1000  
190 PRINT I;“ MILLIAMPS”  
200 GOTO 110  
220 LET A=1  
225 LET M=1000  
230 PRINT“IS CURRENT IN AMPS OR MILLIAMPS”;  
235 INPUT X
```




Fig. 5-3. Flowchart for Ohm's Law.

```
240 PRINT "CURRENT";
245 INPUT I
250 LET I=I/X
255 RETURN
260 LET R=1
270 LET K=1000
280 PRINT "IS RESISTANCE IN OHMS OR KILOHMS";
290 INPUT X
300 PRINT "RESISTANCE";
310 INPUT R
320 LET R=R*X
330 RETURN
```

TRS-80 BASIC

```
10 O=1:V=2:C=3
20 IN."DO YOU NEED TO FIND OHMS, VOLTS, OR CURRENT";X
30 IF X=O G.70
40 IF X=V G.130
50 IF X=C G.160
60 G.20
70 IN."VOLTS";E
80 GOS.220
90 R=I/E:P=E*I
100 "RESISTANCE IS ";R;"OHMS (OR ";
105 R=R/1000:P.R;" KILOHMS)"
110 P."POWER IS ";P;" WATTS"
120 G. 10
130 GOS. 220
135 GOS. 260
140 E=I*R:P=I*I*R
145 P."VOLTAGE IS ";E;" VOLTS"
150 G. 110
160 IN."VOLTS";E
165 GOS. 260
170 P."CURRENT IS ";I;" AMPS (OR ";
180 I=I*1000
190 P.I;" MILLIAMPS)"
200 G. 110
220 A=1:M=1000
230 IN."IS CURRENT IN AMPS OR MILLIAMPS";X
240 IN."CURRENT";I
250 I=I/X
255 RET.
260 R=1:K=1000
270 IN."IS RESISTANCE IN OHMS OR KILOHMS";X
```

280 IN."RESISTANCE";R
290 R=R*X
300 RET.

Summary of Variables Used

A AMPS
C CURRENT
E VOLTAGE VALUE
I CURRENT VALUE
K KILOHMS
M MILLIAMPS
O OHMS
P POWER VALUE
R RESISTANCE VALUE
V VOLTAGE
X INPUT VARIABLE

Sample Run

DO YOU NEED TO FIND OHMS, VOLTS, OR CURRENT? OHMS
VOLTS?9
IS CURRENT IN AMPS OR MILLIAMPS? AMPS
CURRENT? 1.5
RESISTANCE IS 6 OHMS (OR 0.0006 KILOHMS)
POWER IS 13.5 WATTS
DO YOU NEED TO FIND OHMS, VOLTS, OR CURRENT? OHMS
VOLTS? 130
IS CURRENT IN AMPS OR MILLIAMPS? MILLIAMPS
CURRENT? 50
RESISTANCE IS 2700 OHMS (OR 2.7 KILOHMS)
POWER IS 6.75 WATTS
DO YOU NEED TO FIND OHMS, VOLTS, OR CURRENT? CURRENT
VOLTS? 87
IS RESISTANCE IN OHMS OR KILOHMS? KILOHMS
RESISTANCE? 4.7
CURRENT IS 0.0185106 AMPS (OR 18.5106 MILLIAMPS)
POWER IS 1.6104255 WATTS
DO YOU NEED TO FIND OHMS, VOLTS, OR CURRENT? VOLTS
IS CURRENT IN AMPS OR MILLIAMPS? AMPS
CURRENT? 15
IS RESISTANCE IN OHMS OR KILOHMS? OHMS
RESISTANCE? 180
VOLTAGE IS 2700 VOLTS
POWER IS 40500 WATTS

Fahrenheit/Celsius

Here is another program that could be used for immediately practical purposes, i.e., when you need to convert a specific temperature from Fahrenheit to Celsius, or vice versa. But I think it can be better used with a string of randomly selected values so you can really get a solid feel of the relationship of Fahrenheit to Celsius. The U.S. seems to be slowing in the process of converting to metric, but it's still a good idea to get comfortable with the metric standards.

You can devise similar programs for converting inches/feet/miles to centimeters/meters/kilometers, pints/quarts/gallons to liters, ounces/pounds/tons to grams and kilograms, and so forth.

See Fig. 5-4 for the flowchart.

Standard BASIC

```
10 LET A=1
15 LET B=2
20 IN."A—FAHRENHEIT TO CELCIUS OR B— CELSIUS TO"
25 PRINT "FAHRENHEIT";
30 INPUT M
40 IF M=2 THEN GOTO 100
50 PRINT"TEMPERATURE IN FAHRENHEIT";
55 INPUT T
60 LET X=(5/9)*(T-32)
70 PRINT "TEMPERATURE IN CELSIUS IS ";X;" DEGREES"
80 GOTO 10
100 PRINT"TEMPERATURE IN CELSIUS";
110 INPUT T
120 LET X=(9/5)*T+32
130 PRINT"TEMPERATURE IN FAHRENHEIT IS ";X;" DEGREES"
140 GOTO 10
```

TRS-80 BASIC

```
10 A=1:B=2
20 P."A—FAHRENHEIT TO CELSIUS OR B— CELSIUS TO"
30 IN."FAHRENHEIT";M
40 IF M=2 G. 100
50 IN."TEMPERATURE IN FAHRENHEIT";T
60 X=(5/9)*(T-32)
70 P."TEMPERATURE IN CELSIUS IS ";X;" DEGREES"
80 G. 10
```

```
100 IN."TEMPERATURE IN CELSIUS";T  
110 X=(9/5)*T+32  
120 P."TEMPERATURE IN FAHRENHEIT IS ";X;" DEGREES"  
130 G. 10
```

Summary Of Variables Used

- A FAHRENHEIT TO CELSIUS
- B CELSIUS TO FAHRENHEIT
- M MODE SELECTION
- T INPUT TEMPERATURE
- X CONVERTED TEMPERATURE





Fig. 5-4. Flowchart for Fahrenheit/Celsius.

Sample Run

```
A — FAHRENHEIT TO CELSIUS OR B — CELSIUS TO
FAHRENHEIT? A
TEMPERATURE IN FAHRENHEIT? 72
TEMPERATURE IN CELSIUS IS 22.22222 DEGREES
A — FAHRENHEIT TO CELSIUS OR B — CELSIUS TO
FAHRENHEIT? B
TEMPERATURE IN CELSIUS? 0
TEMPERATURE IN FAHRENHEIT IS 32 DEGREES
A — FAHRENHEIT TO CELSIUS OR B — CELSIUS TO
FAHRENHEIT? B
TEMPERATURE IN CELSIUS? 32
TEMPERATURE IN FAHRENHEIT IS 89.6 DEGREES
A — FAHRENHEIT TO CELSIUS OR B CELSIUS TO FAHRENHEIT? B
TEMPERATURE IN CELSIUS? -15
TEMPERATURE IN FAHRENHEIT IS -26.11111111 DEGREES
A — FAHRENHEIT TO CELSIUS OR B CELSIUS TO FAHRENHEIT? B
TEMPERATURE IN CELSIUS? -15
TEMPERATURE IN FAHRENHEIT IS 5 DEGREES
```

Appendix: Notes on Programming in BASIC



BASIC is one of the easiest computer languages to learn because the commands are in English, or they're straightforward abbreviations. It's also fairly standardized, so a program written on one computer can usually be easily translated for use on another manufacturer's unit. There are slight variations from brand to brand, but they are usually quite minor.

In this Appendix, commands used in the programs contained in this book will be explained. Alternative forms used on some of the more popular models now on the market will also be given. By comparing the following list of commands with the commands given on the manual for your computer, you should have no difficulty adapting the programs for use on your own machine. In most cases the commands will be identical, so you can enter them directly.

There are a number of other commands available in most versions of BASIC, but I have listed here the most commonly used. The programs in this book are limited to these common commands so they can be run on almost any computer.

In some cases you will be able to substitute a number of program steps with a single special-purpose command. This option is available on some advanced computers. You should always feel quite free to modify the programs any way you like and to take advantage of any special functions of your own computer.

Note that the games in this book don't make any use of graphics. This is because graphic capabilities are quite unstandardized from computer to computer. To avoid confusion and translation difficulties, I decided to forego graphics in these programs. Naturally, you can add any graphics you please.

BASIC COMMANDS

+	Addition. This symbol is used exactly the way it should be in algebra. For example, the statement LET C=A+B will add A and B and identify the sum as C.
-	Subtraction. Example; LET C=A-B. See +
*	Multiplication. This symbol must be used when multiplication is desired. C=AB is acceptable in algebra, but not in BASIC. If you omit the symbol, the program will bomb

out. An asterisk is used instead of the more familiar X to avoid confusion with the variable. Example; C=A*B
Division. Example; C=A/B
Equals. See LET and IF . . . THEN . . .
Greater than. See IF . . . THEN . . .
Less than. See IF . . . THEN . . .
Parentheses are used primarily the way they are in algebra, to alter the order of mathematical operations. For example $2*3+4=10$, but $2*(3+4)=14$. Note That $2(3+4)$ is not acceptable to a computer. It must be written as $2*(3+4)$.

: A **colon** separates two commands on a single line. For example, LET A=5:LET B=7. Not all computers have this capability.
; A **semi-colon** allows you to combine text and/or variable values on a single line with a PRINT command. For example, if X = 5, the command PRINT "THE VALUE OF X IS"; X would produce this result—

THE VALUE OF X IS 5

See also PRINT

, A **comma** allows you to combine text and/or variables on a single line with a PRINT command, but they will separated by a space. For example, if X=5, the command PRINT "THE VALUE OF X IS", X would produce this result—

THE VALUE OF X IS 5

A comma can also be used on most computers to separate variables. For example, INPUT A, B. The input data must also be set off by commas.

See also INPUT

See also PRINT

A(X) **Array location number X.** X can be a variable, or a fixed numeral.

The array can give you a large supply of variables in addition to the 26 letters of the alphabet. The size and number of arrays varies from computer to computer. Only one array is used in the programs in this book.

A string variable. This allows the computer to recall up to sixteen (typically) characters. For example, if A\$= HELLO, PRINT A\$,A\$,A\$ would produce this result:

A\$

HELLO HELLO HELLO
While many computers offer a number of string variables, only A\$ and B\$ are used in these programs. Note that a space counts as a character in a string variable. That is, "JOHN DOE" is considered to have eight characters.

ABS(X)

This produces the **absolute value of X** (or any variable. The absolute value is always positive. If X is positive then ABS(X) will equal X. If X is negative, the sign is reversed. For example, ABS(5)=5, ABS(-7)=7.

CLE

Clear program. See NEW

CLS

Clear screen. This erases everything displayed on the video terminal, and resets the pointer to the top of the screen. This command sometimes varies on different computers. On the Apple II it is HOME, and on the PET it is PRINT "  "

A number of computers don't have a clear screen command that can be incorporated into the program. For these computers you can use the following sequence:

- FOR X=1 TO Y
- PRINT
- NEXT X

Y is greater than the number of lines the video display will hold at the time. This method (which is used for the Standard BASIC versions throughout this book) will start the text scrolling up from the bottom of the screen, rather than starting at the top and working down. Each new line will push the previous lines up one.

 E+

Exponential scientific notation. Often you need to work with numbers that are too large or too small to be dealt with directly. In these cases an exponent is used. An exponent tells you how many times to multiply (or divide if negative) the number by 10. For example, 1.568E+07= 15,680,000. Standard algebraic notation would be 1.568×10^7

END

This command tells the computer that the program is finished and it should stop. In

some cases this command is optional, but it is always necessary when subroutines follow the main body of the program.

On some computers the command is **STOP**.

FOR X=A TO B: NEXT X

This is a **loop counter**. For example, FOR X = 1 TO 100: NEXT X will make the computer count from 1 to 100. In this form the command produces a timing delay. Another use is to insert a number of additional program steps between FOR X = A TO B and NEXT X for repeated execution. The abbreviated form for the TRS-80 is F.X=ATOB:N.X

FOR X=A TO B STEP C

The **step statement** is used when you need to count by some interval other than one. A negative number can be used if A is larger than B. Example; FOR X=1TO11 STEP 2: PRINT NEXT X would result in the following:

1
3
5
7
9
11

GOSUB

Go to subroutine. This command must be followed by the line number where the subroutine begins. Subroutines are used when a number of steps are required at several points throughout a program. By using a subroutine rather than entering the steps separately each time, much memory space can be saved. A subroutine must end with the command RETURN to send the computer's pointer back to where it left off in the main program. The TRS-80 abbreviations are GOS. and RET.

GOTO

This command can send the program ahead or back to any program step. It must be followed by the appropriate line number. GOTO is most often used with an IF . . . THEN . . . test. The TRS-80 abbreviation is G.

HOME

Clears the display screen on APPLE II computers. See CLS

IF . . . THEN . . .

This statement is used to test and compare data. For example, IF X=Y THEN GOTO 200.

If X does equal Y then the program will jump to line 200, ignoring any intermediate steps. If X does not equal Y the program will continue the next consecutive step. The term GOTO can be skipped on some computers. Others will allow you to leave out the word THEN.

> or < can be used in place of =

Another frequent form is IF X=Y THEN A=B. This works in the same manner as above.

INPUT

This stops the program and signals the operator (usually with a question mark or some prompting symbol) to enter data. Generally a preceding PRINT command will specify what data is required. The INPUT command must be followed by a variable character that will take on the value of the inputted data. String variables can be used (see A\$). It can be abbreviated as IN. on the TRS-80. Some computers (including the TRS-80) will allow an implied PRINT command between the term INPUT and the variable. For example, INPUT "ENTER YOUR DATA";X will appear on the display as:

ENTER YOUR DATA?_____

The same result can be achieved with two commands, i.e. PRINT"ENTER YOUR DATA";:INPUT X See also,

INT(X)

This command chops off any fractional portion of the variable's value, leaving only the integer. For example, INT(3.14) = 3

LET

This command is used to assign values to variables. For examples, LET A=5, or LET X=A + B. The order cannot be reversed. A + B = X and 5=A are not allowed.

Many computers allow the LET to be left off and merely implied. The commands would then be simply A=5 or X= A + B


NEW

This command is never used within a program. It is used to clean an old program out of the computer's memory, and make it ready for a new program.

Some computers use CLE or SCR for this command.

PRINT	When used by itself the PRINT command prints a blank line. Or, if the preceding PRINT command ended with a comma or semi-colon, this command will start a new line.
PRINT X	See also ; and , When a PRINT command is followed by a variable, the numerical value of that variable will be printed. For example, LET X=5:PRINT X will result in 5
	This command can also print the result of a mathematical operation. For example, PRINT 5+3*(7-4) will result in 14
PRINT "xxxxx"	If a PRINT command is followed by any characters enclosed in quotes, whatever is in quotes will be printed. For example, PRINT "THIS IS AN EXAMPLE" will result in this display— THIS IS AN EXAMPLE
PRINT "  "	This is the clear screen command for the PET computer. See CLS
REM	Remark. This command is ignored by the computer. It allows you to place explanatory notes at convenient places in a program
RETURN	This command must be used to end a subroutine. On the TRS-80 it can be abbreviated as RET. See GOSUB
RND(X)	This command is used to generate a random number for adding an element of chance to games or probability studies. LET X=RND(0) produces a number between 0 and 1, for example, 0.5371. Some computers (such as the TRS-80) will allow you to directly produce larger random numbers. For example, X=RND(10) would produce an integer between one and ten, inclusive. Most computers, however, will only recognize RND(0). Higher numbers can be generated with a few additional commands. The random number can be multiplied by the largest number you want. LET X=RND(0)*10 will generate numbers from 0.0001 to 9.9999. By using the integer command (INT), you end up with random numbers between

	0 and 9, inclusive. To avoid the possibility of a zero, just add one to whatever the random number might be. In other words, $X=RND(10)$ is equivalent to $LET X=INT(RND(0)*10)+1$. A few computers (such as the PET) require $RND(1)$, rather than $RND(0)$.
RUN	This command is never used within a program. It is used to put the program into operation. There is usually a BREAK key to let you stop the computer in mid program
SCR	SCRUB program, or SCRATCH program. See NEW
SGN(X)	Reverse sign of variable in brackets. For example, $LET X=-5:SGN(X):PRINT X$ ⁵ $LET Y=7:SGN(Y):PRINT Y$ ⁻⁷
SQR(X)	Square X. That is, multiply the variable times itself. $LET X=X*X$ would give the same result.



Index

B				P	
Balancing the Scales	99	summary of variables used	159	Passing Points	141
sample run	103	TRS-80 BASIC	158	sample run	146
standard BASIC	99	GoFish	113	standard BASIC	141
summary of variables used	102	sample run	120	summary of variables used	145
TRS-80 BASIC	101	standard BASIC	113	TRS-80 BASIC	144
Binary/Decimal	202	summary of variables used	121		
sample run	206	TRS-80 BASIC	118		
standard BASIC	204	Guess the Variables	16	Q	
summary of variables used	206	sample run	21	Quiz Whiz	122
TRS-80 BASIC	205	standard BASIC	17	sample run	126
		summary of variables used	22	standard BASIC	122
		TRS-80 BASIC	19	TRS-80 BASIC	124
				variables used	126
C					
Coin Flipper	189	H			
sample run	191	Hangman	23	\$	
standard BASIC	190	sample run	33	Stack Cut	161
summary of variables used	191	standard BASIC	23	sample run	166
TRS-80 BASIC	191	summary of variables used	32	standard BASIC	161
Crap	8	TRS-80 BASIC	29	TRS-80 BASIC	165
sample run	11	High Bid	90	variables used	168
standard BASIC	8	sample run	93	Surprise Poem	192
summary of variables use	11	standard BASIC	90	standard BASIC	194
TRS-80 BASIC	10	summary of variables used	94	TRS-80 BASIC	196
Crops	41	TRS-80 BASIC	92		
sample run	48	II, standard BASIC	95	T	
standard BASIC	41	TRS-80 BASIC	97	The Computer vs. Itself	184
summary of variables used	47	High/Low	12	standard BASIC	184
TRS-80 BASIC	45	sample run	14	summary of variables used	187
		standard BASIC	12	TRS-80 BASIC	186
		summary of variables used	15	There's Gold in	
D				Them There Skyscrapers	49
Day of the Week	207	TRS-80 BASIC	14	sample run	54
sample run	209			standard BASIC	50
standard BASIC	208			summary of variables used	54
summary of variables used	209			TRS-80 BASIC	53
TRS-80 BASIC	208	L			
		Lost in the Jungle	56	Tic Tac Toe	127
		sample run	70	margin sample run	134
F		standard BASIC	56	margin summary of	
Fahrenheit/Celsius	214	summary of variables used	69	variables used	133
sample run	216	TRS-80 BASIC	66	standard BASIC	127
standard BASIC	214			TRS-80 BASIC	131
summary of variables used	215				
TRS-80 BASIC	214	M		To and Fro	136
Favorite Song	192	Main Game Program	171	sample run	140
standard BASIC	192	sample run	174	standard BASIC	136
TRS-80 BASIC	193	standard BASIC	171	summary of variables used	140
Freebish	34	summary of variables used	173	TRS-80 BASIC	138
rules	40	TRS-80 BASIC	172	Tunesmith	198
standard BASIC	34	Memory Test	104	sample run	201
TRS-80 BASIC	38	standard BASIC	104	standard BASIC	198
Frustration vs. the Computer	181	summary of variables used	108	TRS-80 BASIC	200
standard BASIC	182	TRS-80 BASIC	107	Two Man Frustration	176
TRS-80 BASIC	181	Money Mad	147	sample run	179
		sample run	153	standard BASIC	176
		standard BASIC	147	summary of variables used	178
G				TRS-80 BASIC	177
Galactic Search	72	TRS-80 BASIC	151	W	
sample run	88			What Comes Next	109
standard BASIC	72			sample run	112
summary of variables used	88	Ohm's Law	210	standard BASIC	109
TRS-80 BASIC	82	sample run	213	summary of variables used	112
Gambling Boxes	155	standard BASIC	210	TRS-80 BASIC	110
sample run	159	summary of variables used	213		
standard BASIC	155	TRS-80 BASIC	212		

34 MORE Tested, Ready-To-Run Game Programs In BASIC

by Delton T. Horn

Who says computers are *only* for serious business? There's a light side to your computer's personality, too . . . and here's the book that shows you just *how much fun* it can really be! Includes a complete collection of games, gags, brain teasers, and "non-games" . . . even programs that entertain *while they educate*! In easy-to-read-and-use format with lots of illustrations, flowcharts, and listings for each program in both standard BASIC and abbreviated BASIC using the Level I TRS-80 computer.

Find one-player games ranging from Tic-Tac-Toe and Lost in the Jungle to the challenging Guess The Variables, Quiz Whiz, or Memory Test. Two-player games like Money Mad or Stack Cut can keep you and a friend occupied for hours . . . and there are even games you can watch the computer play against itself! Looking for something that's practical, but still entertaining? Try Binary Decimal. It tests your skills at converting numbers from one system to another, and is a must for anyone attempting to write their own programs. Day of the Week, Ohm's Law, and Fahrenheit/Celsius are other fun-but-practical exercises included in this upbeat ideabook.

Discover Frustration, a game so complex and challenging that there's a whole section detailing the strategies involved. Plus, there are lots of "non-games" . . . Coin Flipper, Surprise Poem, Favorite Song, or Tunesmith. An appendix gives you all the data you need to understand the BASIC commands included in the various programs, and suggests alternative forms for computer models other than the TRS-80. Hints, too, on how to change and modify the programs to suit your own particular fancy.

If you'd like to put some fun-and-games into your computer's life, then this is a program guide you won't want to miss!

Delton T. Horn is a field engineer and technician, and author of TAB's *Electronic Music Synthesizers*.

OTHER POPULAR TAB BOOKS OF INTEREST

1001 Things To Do With Your Personal Computer
(No. 1160—\$9.95 paper; \$14.95 hard)

24 Tested, Ready-to-Run Game Programs in BASIC
(No. 1085—\$9.95 paper only)

57 Practical Programs & Games in BASIC (No.
1000—\$7.95 paper; \$13.95 hard)

The Most Popular Subroutines in BASIC (No.
1050—\$7.95 paper; \$13.95 hard)



TAB BOOKS Inc.

Blue Ridge Summit, Pa. 17214

Send for FREE TAB Catalog describing over 750 current titles in print.

FPT > > \$9.25

ISBN 0-8306-1228-9